新编计算机类本科规划教材

Visual Basic 程序设计教程习题及习题解答（第4版）

刘瑞新　等编著



電子工業出版社
PUBLISHING HOUSE OF ELECTRONICS INDUSTRY
http://www.phei.com.cn

新编计算机类本科规划教材

Visual Basic程序设计教程习题及习题解答

（第4版）

刘瑞新　等编著

電子工業出版社
Publishing House of Electronics Industry
北京・BEIJING

内 容 简 介

本书为普通高等教育“十一五”国家级规划教材《Visual Basic 程序设计教程（第 4 版）》的配套教材，对该教材中的习题做了详细解答。本书以强化学生实践能力为目的，详细讲解了等级考试中的选择题、填空题、思考题、编程题等各种题型的应试技巧和分析解答方法，既便于学生检测知识掌握程度，又符合各类 VB 考试题型。同时，提供的各种编程典型习题既方便教师掌握重点，也便于学生复习应试。本书的程序调试技术一章，对 VB 上机技巧和程序调试方法进行了详细讲解。由于 VB 控件及其属性内容繁多，为方便读者使用时查询，本书附录中列出了 VB 常用事件、属性和方法及其含义。

本书可作为高等学校、高职高专院校相关专业的教材，也可作为全国计算机等级考试二级 Visual Basic 语言的辅导教材，或者作为 Visual Basic 语言的“编程实例详解”单独使用。

图书在版编目（CIP）数据

Visual Basic 程序设计教程习题及习题解答 / 刘瑞新等编著. —4 版. —北京：电子工业出版社，2013.3
新编计算机类本科规划教材
ISBN 978-7-121-19781-9

I. ①V… II. ①刘… III. ①BASIC 语言—程序设计—高等学校—题解 IV. ①TP312-44

中国版本图书馆 CIP 数据核字（2013）第 045849 号

责任编辑：冉 哲
印 刷：
装 订：北京季蜂印刷有限公司
出版发行：电子工业出版社
北京市海淀区万寿路 173 信箱 邮编 100036
开 本：787×1 092 1/16 印张：13.75 字数：348 千字
印 次：2013 年 3 月第 1 次印刷
定 价：29.80 元

凡所购买电子工业出版社图书有缺损问题，请向购买书店调换。若书店售缺，请与本社发行部联系，联系及邮购电话：（010）88254888。

质量投诉请发邮件至 zlts@phei.com.cn，盗版侵权举报请发邮件至 dbqq@phei.com.cn。

服务热线：（010）88258888。

前　言

Visual Basic 是一种可视化的编程语言，人们利用这种可视化技术进行编程，将使编程工作变得轻松快捷，使个人摆脱面向过程语言的许多细节，而将主要精力集中在解决实际问题和设计友好界面上。因此，其在国内外各个领域中应用非常广泛，许多计算机专业和非计算机专业的人员常利用它来开发应用程序和软件。

普通高等教育“十一五”国家级规划教材《Visual Basic 程序设计教程（第 4 版）》一书从实例出发，通过大量有趣的实例介绍程序设计基础知识和方法，避免枯燥、空洞的理论，容易上手，使读者于不知不觉之中学会在 Windows 环境下编程。该书在例题讲解的处理上，先给出设计目标，然后介绍为实现本目标而采取的设计方法。采用这种处理方式，可使学生明确程序设计的思想和方法，做到有的放矢。另外，对于工科院校的学生，他们需要用 Visual Basic 解决实际问题，而这些问题大多都有数学模型，所以该书把重点放在解决实际问题上。该书以 Visual Basic 6.0 中文版为语言背景，通过大量实例，深入浅出地介绍 Visual Basic 程序开发环境，Visual Basic 程序设计基础知识，Visual Basic 可视化编程的概念与方法，顺序结构程序设计，选择结构程序设计，循环结构程序设计，数组，过程，变量与过程的作用域，用户定义类型与枚举类型，图形与图像，菜单、工具栏与对话框，键盘与鼠标事件过程，数据文件，数据库访问技术等内容。

通常，对于初学程序设计的学生，老师讲课时能听得懂，但自己动手时却不知如何下手。为此，我们编写了本书作为《Visual Basic 程序设计教程（第 4 版）》的配套教材，提供了该书的全部习题解答。

本书以强化学生实践能力为目的，详细讲解了 VB 选择题、填空题、思考题、编程题等各种题型的应试技巧和分析解答方法，既便于学生检测知识掌握程度，又符合各类 VB 考试题型。同时，提供的各种编程典型习题便于教师掌握重点，也便于学生复习应试。

在本书的程序调试技术一章中，对 VB 上机技巧和程序调试方法进行了详细讲解。由于 VB 控件及其属性内容繁多，为方便读者使用时查询，本书附录中列出了 VB 常用事件、属性和方法及其含义。

总之，本书习题解答方法多样，界面丰富多彩，对开拓思维具有启发作用。本书所有程序都可以在 Visual Basic 6.0 下正常运行。本书也非常适合作为 Visual Basic 的“编程实例详解”单独使用。

本书由刘瑞新编著，参加编写的还有蔡峰、张鸣、刘慧敏、王瑶、胡楠、张志强、张明增、贾俊亮、马志刚、冯全民、董福新、刘美想、张锐、张小兵、杨桦。随着社会发展和教学改革的深入，请读者将教改的成果经验和对本书的建议及时告知，以便精益求精。

本书可作为高等学校、高职高专院校相关专业的教材，也可作为全国计算机等级考试二级 Visual Basic 语言的辅导教材，或者作为 Visual Basic 语言的“编程实例详解”单独使用。

编　者

目　　录

第 1 章　Visual Basic 程序开发环境

一、选择题

1.1　Visual Basic 6.0 分为 3 种版本，不属于这 3 种版本的是（　　）。

A）学习版　　B）专业版　　C）企业版　　D）业余版

【解答】正确答案为 D。

分析：Visual Basic 包括 3 种版本，分别为学习版、专业版和企业版。学习版是 Visual Basic 最基本、最便宜的版本，包括创建 Windows 应用程序所需要的内部控件及数据网络、数据绑定控件等；专业版包括学习版中的全部内容，又增加了立体控件、动画按钮、通信控件、进度条、工具栏和 Internet 控件等开发应用程序所需要的全套工具，功能更强大；企业版包括专业版式中的全部内容，是 Visual Basic 的最强版本，为软件开发团队开发大型的网络环境应用软件体系提供了强有力的支持。

1.2　下列方法中不能退出 Visual Basic 的是（　　）。

A）按 Alt+Q 组合键　　B）按 Alt+F 组合键，然后按 Esc 键

C）按 F10 键，然后按 F 键，再按 X 键　　D）执行“文件”→“退出”菜单命令

【解答】正确答案为 B。

分析：Alt+Q 为退出 Visual Basic 的快捷键；Alt+F 为激活“文件”菜单的快捷键，打开菜单后，相应的快捷命令在各命令选项后都有提示，按 Esc 键将关闭菜单，不能退出 VB；F10 为激活菜单栏的快捷键，激活菜单栏后，按 F 键选择“文件”菜单，再按 X 键选择“退出”命令，可以退出 VB；打开“文件”菜单，执行“退出”命令，也是可以的。

1.3　以下叙述中错误的是（　　）。

A）Visual Basic 是事件驱动型可视化编程工具

B）Visual Basic 应用程序不具有明显的开始和结束语句

C）Visual Basic 工具箱中的所有控件都具有宽度（Width）和高度（Height）属性

D）Visual Basic 中控件的某些属性只能在运行时设置

【解答】正确答案为 C。

分析：VB 中的计时器控件（Timer 控件）没有宽度（Width）和高度（Height）属性，故选项 C 错误，其他都正确，所以答案为 C。

1.4　以下叙述中错误的是（　　）。

A）在工程资源管理器窗口中只能包含一个工程文件及属于该工程的其他文件

B）以.bas 为扩展名的文件是标准模块文件

C）窗体文件包含该窗体及其控件的属性

D）一个工程中可以含有多个标准模块文件

【解答】正确答案为 A。

分析：在工程资源管理器中可以包含多个工程，故选项 A 中的叙述是错误的，其他各选项都正确，所以答案为选项 A。

1.5　以下不属于 Visual Basic 系统的文件类型是（　　）。

A）frm　　B）bat　　C）vbg　　D）vbp

【解答】正确答案为 B。

分析：.bat 文件是 DOS 系统下的批处理文件。

1.6　以下叙述中错误的是（　　）。

A）打开一个工程文件时，系统自动装入与该工程有关的窗体、标准模块等文件

B）保存 Visual Basic 程序时，应分别保存窗体文件及工程文件

C）Visual Basic 应用程序只能以解释方式执行

D）事件可以由用户引发，也可以由系统引发

【解答】正确答案为 C。

分析：此题可运用排除法。其中选项 A、选项 B、选项 D 都正确，故答案为选项 C。

1.7　Visual Basic 集成的主窗口中不包括（　　）。

A）属性窗口　　B）标题栏　　C）菜单栏　　D）工具栏

【解答】正确答案为 A。

分析：Visual Basic 的主窗口又称为设计窗口，它由标题栏、菜单栏、工具栏等部分组成。标题栏是位于屏幕顶部的水平条，它显示应用程序的名字，启动 Visual Basic 后，标题栏中出现的信息是“工程 1-Microsoft Visual Basic［设计］”；菜单栏位于标题栏之下，提供了开发、调试应用程序的工具，包括“文件”菜单、“编辑”菜单、“视图”菜单、“工程”菜单等；工具栏位于菜单栏的下面，它以图标形式提供了部分常用命令的功能；属性窗口不在主窗口的范围之内。

1.8　下列操作可以打开立即窗口的是（　　）。

A）Ctrl+D　　B）Ctrl+F　　C）Ctrl+G　　D）Ctrl+E

【解答】正确答案为 C。

分析：在 Visual Basic 集成开发环境中，Ctrl+D 组合键用来打开“添加文件”对话框，所以选项 A 不正确；Ctrl+F 组合键用来打开属性窗口，所以选项 B 也不正确；Ctrl+G 组合键用来打开立即窗口，所以选项 C 正确；Ctrl+E 组合键用来打开菜单编辑器，所以选项 D 不正确。

1.9　下列说法错误的是（　　）。

A）方法是对象的一部分　　B）在调用方法时，对象名是不可缺少的

C）方法是一种特殊的过程和函数　　D）方法调用格式和对象属性使用格式相同

【解答】正确答案为 B。

分析：方法是对象的一部分，所以选项 A 的表述是正确的；其调用格式为：对象名.方法名称，与属性的使用格式相同，所以选项 D 是正确的；方法在调用时，可以省略对象名，但这只限于当前窗体（或控件），如果不是当前窗体或控件要加上对象名称，所以选项 B 是不正确的；方法是一种特殊的过程和函数，所以选项 C 是正确的。

1.10　下列说法错误的是（　　）。

A）窗体文件的扩展名为.frm　　B）一个窗体对应一个窗体文件

C）VB 中一个工程只包含一个窗体　　D）VB 中一个工程最多可以包含 255 个窗体

【解答】正确答案为 C。

分析：VB 中的一个工程可以包括一个或多个窗体，最多不超过 255 个窗体，因此选项 D 是正确的，选项 C 是不正确；窗体文件的扩展名是.frm，所以选项 A 是正确的；一个窗体对应于一个窗体文件，所以选项 B 也是正确的。

1.11　一个工程必须包含的文件的类型是（　　）。

A）.vbp,.frm ,.frx　　B）.vbp ,.cls ,.bas

C）.bas ,.ocx ,.res　　D）.frm ,.cls ,.bas

【解答】正确答案为 A。

分析：一个工程中可以包含 7 类文件，其中，工程文件、窗体文件和窗体的二进制数据文件是一个工程中不可缺少的文件。这 7 类文件的扩展名及含义分别是：

.vbp 是 Visual Basic Project 的缩写，表示工程文件；

.frm 是 Form 的缩写，表示窗体文件；

.frx 是窗体的二进制数据文件；

.cls 是 class 的缩写，表示类模块文件；

.bas 是 Basic 的缩写，表示标准模块文件；

.ocx 表示 ActiveX 控件的文件；

.res 是 resource 的缩写，表示资源文件。

1.12　新建一个窗体，其 BorderStyle 属性设置为 Fixed Single，但运行时却没有最大化和最小化按钮，可能的原因是（　　）。

A）BorderStyle 的值设为 Fixed.Single，此项设置值的作用是禁止最大化和最小化按钮

B）窗体的 MaxButton 和 MinButton 值设为 False

C）在正常情况下新建的窗体都没有最大化和最小化按钮

D）该窗体可用鼠标拖动框的方法改变窗体的大小

【解答】正确答案为 B。

分析：如果窗体的 BorderStyle 属性设置为 Fixed Single，则窗体为固定单边框，可以包含控制菜单框、标题栏、最大化和最小化按钮，但要注意的是，如果窗口的 MaxButton 和 MinButton 属性设置为 True，则运行时显示最大化和最小化按钮；如果设置为 False，则运行时不显示。

二、填空题

1.13　与传统的程序设计语言相比，Visual Basic 最突出的特点是____。

【解答】正确答案为：事件驱动编程机制。

1.14　如果不使用鼠标，用键盘打开菜单和执行菜单命令，第一步应按____键。

【解答】正确答案为：F10 或 Alt。

1.15　建立一个新的标准模块，应该选择____菜单下的“添加模块”命令。

【解答】正确答案为：“工程”。

分析：在 Visual Basic 中，建立一个新的标准模块应选择“工程”→“添加模块”菜单命令，弹出“添加模块”对话框，在“新建”选项下选择“模块”选项，然后单击“打开”按钮，打开标准模块代码窗口，在这个窗口中就可以输入标准模块代码。

三、思考题

1.16　叙述 Visual Basic 6.0 的安装过程。

【解答】Visual Basic 6.0 的安装过程如下。

（1）将 Visual Basic 6.0（简称 VB 6.0）的安装光盘放入光驱中，若没有取消“自动播放”功能，安装程序将会自动运行，否则应在“我的电脑”或“资源管理器”中运行安装光盘中的 Setup 程序，运行后显示出“Visual Basic 6.0 中文企业版安装向导”，如图 1-1 所示。

图 1-1　VB 6.0 中文企业版安装向导

（2）单击“下一步”按钮，则打开最终用户许可协议对话框，在该对话框中选择“接受协议”后，单击“下一步”按钮。此时安装程序会要求用户输入产品的 ID 号、用户的姓名和公司名称。

（3）输入产品 ID 号和用户信息后，单击“下一步”按钮，打开选择安装程序对话框，如图 1-2 所示。

图 1-2　选择安装程序对话框

（4）在图 1-2 中选择“安装 Visual Basic 6.0 中文企业版”后，单击“下一步”按钮，

在完成安装路径选择后，安装程序将打开选择安装类型对话框，如图 1-3 所示。

图 1-3　选择安装类型对话框

（5）在选择安装类型对话框中，安装程序为用户提供了两个选择：“典型安装”和“自定义安装”。若选择前者将安装最典型的组件，安装过程无须用户干预。若用户选择了后者，将打开自定义安装对话框，如图 1-4 所示。用户使用这种方法可以根据需要有选择地安装需要的组件。

（6）单击图 1-4 对话框中的“继续”按钮后，安装程序将复制文件到计算机硬盘中，复制结束后重新启动计算机，完成 VB 6.0 的安装。



图 1-4　自定义安装对话框

1.17　叙述 MSDN 的安装方法、作用及使用方法。

【解答】VB 6.0 安装完毕，计算机重新启动后，安装程序将自动打开安装 MSDN 对话框，如图 1-5 所示。安装程序询问用户是否需要安装 MSDN（Microsoft Developer Network）Library。MSDN Library 是开发人员的重要参考资料，包含容量约为 1GB 的编程技术信息，包括示例代码、文档、技术文章、Microsoft 开发人员知识库及开发程序时所需的其他资料。它是 Microsoft Visual Studio 6.0 套件之一，由两张光盘组成。注意：VB 6.0 的联机帮助文档只有在安装了 MSDN 后方可使用。

此时，可将 MSDN 第一张光盘放入光驱中，单击图 1-5 中的“下一步”按钮，安装程序打开 MSDN 自定义安装对话框，如图 1-6 所示。选择需要安装的组件后单击“继续”按钮，用户可根据屏幕提示完成 MSDN 的安装。如果取消图 1-5 中的“安装 MSDN”复选框将暂不安装 MSDN，待直接运行安装盘中的 Setup 程序来完成后续安装。

至此，MSDN 的安装结束。用户可以从“开始”菜单中启动它，也可以根据需要通过 VB 6.0 的“帮助”菜单启动 MSDN。

图 1-5　安装 MSDN 对话框

图 1-6　MSDN 自定义安装对话框

无论使用何种方法启动 MSDN 后，都将出现图 1-7 所示的窗口。



图 1-7　MSDN 窗口

该窗口中包含定位窗格（左）和主题窗格（右），在定位窗格中有“目录”、“索引”、“搜索”和“书签”4 个选项卡，选择其中一个选项卡后，即可在主题窗格中查看有关的信息。例如，选择了“搜索”选项卡后可以输入单词或短语，以便快速获得需要的帮助信息。

在主题窗格中有些带下画线的文字（超链接文字），单击这些文字可以获得进一步的解释和说明，或链接到其他主题和页面。

第 2 章　Visual Basic 程序设计基础

一、选择题

2.1　以下关于 VB 数据类型的说法，不恰当的是（　　）。

A）VB 6.0 提供的数据类型主要有字符串型和数值型，此外还有字节、货币、对象、日期、布尔和变体数据类型等

B）目前 Decimal 数据类型只能在变体类型中使用

C）用户不能定义自己的数据类型

D）布尔型数据只能取两种值，用两个字节存储

【解答】正确答案为 C。

分析：在 VB 中，提供的数据类型主要有字符串型和数值型，此外还有字节、货币、对象、日期、布尔和变体数据类型等，所以选项 A 是正确的；除了基本数据类型外，用户还可以自己定义数据类型，这个是 VB 所支持的，所以选项 C 不正确；因为目前 Decimal 数据类型只能在变体类型中使用，也就是说，不能把一个变量声明为 Decimal 类型，所以选项 B 是正确的；布尔型数据是一个逻辑值，用两个字节存储，它只能取两种值，即 True 或 False，所以选项 D 也是正确的。

2.2　以下各项，可以作为 VB 变量名的是（　　）。

A）Book　　B）2_Seek　　C）123.58　　D）Book-1

【解答】正确答案为 A。

分析：在 VB 中，可以用名字来表示内存的位置，以便访问内存中的数据。有关变量命名的规则如下：

名字只能由字母、数字和下画线组成；

名字的第一个字符必须是英文字母，最后一个字符可以是类型说明符；

名字的有效字符为 255 个；

不能用 VB 的保留字作为变量名，但可以把保留字嵌入变量名中；同时变量名也不能是末尾带有类型说明符的保留字，例如，变量 Print 和 Print$是非法的，而变量 Print_Number 是合法的。

根据变量命名规则，选项 B 开头第一个字符是数字，非法；选项 C 变量名开头第一个字符是数字，非法；选项 D 中包含非法字符“-”；所以只有选项 A 是正确的。

2.3　下列（　　）不能作为 VB 中的变量名。

A）ABCDEFG　　B）P000000　　C）89TWDDFF　　D）xyz

【解答】正确答案为 C。

分析：VB 变量名的命名规则是：变量名必须以字母开头，长度不得超过 255 个字符，变量名中不能包含小数点，在同一个范围内必须是唯一的。

在为变量命名时应见名知义，尽可能简单明了，不用 VB 的关键字、过程名和符号常量名作为变量名，尽量采用 VB 建议的变量名前缀或后缀的约定来命名，以便区分变量的类型。

该题中给出的 4 个符号中，只有 89TEDDFF 不是以字母开头的，它不能作为变量名。而其他 3 个变量都满足 VB 中变量名的命名规则，是合法的变量名。因此，正确答案为 C。

2.4　下列（　　）是 VB 中的合法变量名。

A）AB7　　B）7AB　　C）IF　　D）A[B]7

【解答】正确答案为 A。

分析：见上题分析。

2.5　表达式 2 *3^2 + 2 * 8 / 4 + 3^2 的值为（　　）。

A）64　　B）31　　C）49　　D）22

【解答】正确答案为 B。

分析：根据运算符的优先级，计算方法如下：

2 * 3^2 + 2 * 8 / 4 + 3^2

= 2 * 9 + 16 / 4 + 9

= 18 + 4 + 9 = 31

2.6　函数 Int(Rnd(0)*10)是在（　　）范围内的整数。

A）(0,1)　　B）(1,10)　　C）(0,9)　　D）(1,9)

【解答】正确答案为 C。

分析：函数 Rnd(0)是 0～1 之间的数，因此 Int(Rnd(0) * 10)的值是 0～9 之间的整数。

2.7　表达式 3 ^ 2 Mod 14＼2^3 的值是（　　）。

A）1　　B）0　　C）2　　D）3

【解答】正确答案为 B。

分析：在做本题之前要先了解各种运算符的优先级，只有了解了这些之后才能做对。幂运算符（^）优先级最高，其次是取负、乘、浮点除、整除、取模、加减、字符串连接等。其中，乘和浮点除是同级运算符，加和减是同级运算符。此外，如果表达式中有括号，则先计算括号内表达式的值。通过运算，本题的结果是 0。

2.8　在 VB 中，下列两个变量名相同的是（　　）。

A）Japan 和 Ja_pan　　B）English 和 ENGLish

C）English 和 Engl　　D）China 和 Chin

【解答】正确答案为 B。

分析：在 VB 中，组成变量名的英文字母不区分大小写，例如，SINGLE 和 Single 是一样的。在定义了一个变量之后，只要字符相同，不管它大小写是否相同，指的都有同一个变量。

2.9　数学式子 sin25°写成 VB 表达式是（　　）。

A）Sin25　　B）Sin(25)

C）Sin(25°)　　D）Sin(25*3.14/180)

【解答】正确答案为 D。

分析：由于 VB 中 Sin( )函数中的参数要求是弧度数，因此应将角度值转换为弧度值，正确答案为 D。

2.10　在 VB 中，要强制用户对所用的变量进行显式声明，可以在（　　）中设置。

A）“属性”对话框　　B）“程序代码”窗口

C）“选项”对话框　　D）对象浏览器

【解答】正确答案为 C。

分析：要强制用户显式声明变量，可以单击“工具”→“选项”菜单命令，打开“选项”对话框，在“编辑器”选项卡下，选中“要求变量声明”复选框即可。这样，在每次新建文件时，VB 将 Option Explicit（选择显式）自动添加到全局变量或模块级变量的声明部分，或者也可以在声明部分直接输入这条语句。这样就必须在使用变量前声明，否则将会发生出错信息。属性窗口是用来设置对象属性的。程序代码窗口是用来编辑事件过程的。

2.11　下列符号常量的声明中，不合法的是（　　）。

A）Const a As Single = 1.1　　B）Const a =″ OK″

C）Const a As Double = Sin(1)　　D）Const a As Integer =″ 12″

【解答】正确答案为 C。

分析：在用 Const 定义符号常量时，格式是：

Const 常量名 = 表达式［，常量名 = 表达式］……，

其中常量名可以用类型说明符，需要注意的是，“表达式”中不能使用字符串连接运算符、变量及用户定义的函数或者内部函数。

2.12　在代码编辑器中，续行符是换行书写同一个语句的符号，用以表示续行符的是（　　）。

A）一个空格加一个下画线“_”　　B）一个下画线“_”

C）一个连字符“-”　　D）一个空格加一个连字符“-”

【解答】正确答案为 A。

分析：在一般情况下，输入程序的语句要求一句一行，一行一句，但 VB 允许使用续行符把程序分在几行中书写，但所使用的下画线要注意与它前面的字符之间要至少有一个空格，这样书写有助于提高程序的条理性和可读性。

二、填空题

2.13　如果希望使用变量 x 来存放数据 765432.123456，应将变量 x 声明为____类型。

【解答】正确答案为：双精度型。

2.14　把 VB 算术表达式 a/(b + c/(d + e/Sqr(f)))改写成数学表达式为____。

【解答】正确答案为：$\dfrac{a}{b+\dfrac{c}{d+\dfrac{e}{\sqrt{f}}}}$。

2.15　如果 *x* 是一个正实数，对 *x* 的第 3 位小数四舍五入的表达式是____。

【解答】正确答案为：0.01 * Int(100 * (x + 0.005))。

2.16　函数 Str＄(256.36)的值是____。

【解答】正确答案为：256.36。

分析：Str＄()函数的格式是：Str＄（数值表达式），其作用是把“数值表达式”的值转

换为一个字符串，表达式的值不受转换过程的影响。对于本题，在立即窗口中可看到输出的结果。

2.17　表达式(7\2+1)*(8\2+2)的值为____。

【解答】正确答案为：24。

分析：在 VB 中符号“\”为整数除法运算符，运算结果为整数值。如果操作数带有小数点，首先四舍五入为整型数或长整型数，然后再进行整除运算。通过运算本题的结果为 24。

2.18　下列语句的输出结果是____。

```
Print Format $ (1258.6,"000,000.00")
```

【解答】正确答案为：001,258.60。

分析：Format $ ()为格式输出函数，格式为：

```
Format $ （数值表达式，"格式字符串"）
```

其功能是按“格式字符串”指定的格式输出“数值表达式”的值。如果省略了“格式字符串”，则把“数值表达式”的值转换为一个字符串原样输出；在本题中，1258.6 将按 000,000.00 的格式输出，即输出结果为：001,258.60。

三、思考题

2.19　VB 定义了哪几种数据类型？变量有哪几种数据类型？常量有哪几种数据类型？

【解答】VB 6.0 定义了 11 种数据类型，它们是整型、长整型、单精度浮点型、双精度浮点型、字符串型（变长字符串、定长字符串）、逻辑型、日期型、货币型、字节型、对象型、变体型（数值、字符）。

变量的数据类型有数值型、字符型、逻辑型、日期型、对象型、变体型。其中，数值型变量的数据类型有：整型、长整型、单精度浮点型、双精度浮点型、货币型和字节类型。字符型变量的数据类型有变长字符串和定长字符串。



常量的数据类型有：字符串常量、数值常量、逻辑常量、日期常量。

常量的数据类型
- 数值型
  - 整型
  - 长整型
  - 单精度浮点型
  - 双精度浮点型
  - 货币型
  - 字节型
- 字符串型
- 逻辑型
- 日期型

2.20　下列数据哪些是变量？哪些是常量？是什么类型的常量？

（1）name　　（2）"name"　　（3）False　　（4）ff

（5）"11/16/99"　　（6）cj　　（7）"120"　　（8）n

（9）#11/16/1999#　　（10）12.345

【解答】变量有（1），（4），（6），（8），常量有（2），（3），（5），（7），（9），（10）。其中：（2），（5），（7）为字符型常量，（3）为逻辑型常量，（9）为日期型常量，（10）为数值型常量。

2.21　VB 共有几种表达式？根据什么确定表达式的类型？

【解答】有 5 种类型的表达式：算术表达式、字符串表达式、关系表达式、逻辑表达式和日期表达式。VB 是根据表达式的运算符来确定表达式的类型的。

2.22　在 VB 中，对于没有赋值的变量，系统默认值是什么？

【解答】使用声明语句建立一个变量后，VB 自动将数值类型的变量赋初值 0，变长的字符串被初始化为一个零长度的字符串（""），而定长的字符串则用空格填充。Variant 变量被初始化为 Empty，逻辑型的变量则初始化为 False。

2.23　将下列数学表达式改写为等价的 VB 算术表达式。

（1）$\dfrac{1+\dfrac{y}{x}}{1-\dfrac{y}{x}}$　　（2）$x^2+\dfrac{3xy}{2-y}$

（3）$\sqrt{|ab-c^3|}$　　（4）$\sqrt{s(s-a)(s-b)(s-c)}$

【解答】VB 中的算术表达式与数学中的表达式写法有所区别，主要表现在：VB 中每个符号占 1 格，所有符号都必须一个一个地并排写在同一条横线上，不能在右上角或右下角写方次或下标；在数学表达式中省略的内容必须重新写上；所有括号都要使用小括号“( )”，而且括号必须配对；数学表达式中的有些符号需要改成 VB 中可以表示的符号。

（1）(1 + y / x) / (1 – y / x)

（2）x^2 + 3 * x * y / (2 – y)

（3）(Abs(a * b – c^3))^0.5 或 (Abs(a * b – c^3))^(1 / 2) 或 Sqr(Abs(a * b – c^3))

（4）(s * (s – a) * (s – b) * (s – c))^0.5 或 (s * (s – a) * (s – b) * (s – c))^(1 / 2)
或 Sqr(s * (s – a) * (s – b) * (s – c))

2.24　写出下列表达式的值。

（1）(2 + 8 * 3) / 2　　（2）3^2 + 8

（3）#11/22/99# – 10　　（4）"ZYX" & 123 & "ABC"

【解答】题中各表达式的值分别为：

（1）13　　（2）17　　（3）#11/12/99#　　（4）"ZYX123ABC"

2.25　设 A = 7, B = 3, C = 4，求下列表达式的值：

（1）A + 3 * C　　（2）A^2 / 6

（3）A / 2 * 3 / 2　　（4）A Mod 3 +B^3 / C \ 5

【解答】题中各表达式的值分别为：

（1）19　　（2）8.16666666666667

（3）5.25　　（4）2

2.26　写出下列表达式的值，并在立即窗口中验证。

（1）"Visual"+"Basic"　　（2）"xyz" & 1234 & "ABCD"

（3）"12345" < > "12345" & "ABC"　　（4）Not 2*5< >11

（5）4=4 And 5>2+2　　（6）Not 8<5 Or 9>3 And 7<9 Or 8=6

【解答】题中各表达式的值分别为：

（1）"VisualBasic"　　（2）"xyz1234ABCD"

（3）True　　（4）False

（5）True　　（6）True

2.27　根据所给条件写出对应的逻辑表达式。

（1）征兵的条件为：男（sex），年龄（age）在 18～20 岁之间，身高（height）在 1.65 米以上；或者女，年龄在 16～18 岁之间，身高在 1.6 米以上。

（2）工资调整的条件为：工龄（gongling）在 15 年以上，岗位（gangwei）是工人；或者工龄在 10 年以上，岗位是教师。

【解答】

（1）设性别 sex 值为 True 代表男，sex 值为 False 代表女，逻辑表达式如下：

(sex And age>=18 And age<=20 And height>=1.65) Or (Not sex And age>=16 And age<=18 And height>=1.60)

（2）设工龄 gongling 为数值型，岗位 gangwei 为字符型，逻辑表达式如下：

(gongling>=15 And gangwei="工人") Or (gongling>=10 And gangwei="教师")

2.28　写出下列函数的值，并在立即窗口中验证。

（1）Int(–3.14159)　　（2）Sqr(Sqr(64))

（3）Fix(–3.1415926)　　（4）Int(Abs(99–100)/2)

（5）Sgn(7*3+2)　　（6）Month("02,08,26")

【解答】函数的值为：

（1）– 4　　（2）2.828427

（3）– 3　　（4）0

（5）1　　（6）8

第 3 章　Visual Basic 可视化编程的概念与方法

一、选择题

3.1　在 VB 中，被称为对象的是（　　）。

A）窗体　　B）控件

C）控件和窗体　　D）窗体、控件和属性

【解答】正确答案为 C。

分析：在 VB 中，窗体和控件被称为 VB 中的对象，而属性是针对具体对象来说的，离开对象谈属性没有任何意义，所以选项 D 是不正确的；选项 A 和选项 B 都只谈了其中的一个方面，不全面；选项 C 是正确的。

3.2　关于 VB“方法”的概念错误的是（　　）。

A）方法是对象的一部分　　B）方法是预先定义好的操作

C）方法是对事件的响应　　D）方法用于完成某些特定的功能

【解答】正确答案为 C。

分析：对象是属性、方法和事件的集成，所以选项 A 说法正确；例如，Print 方法用来输出字符串、变量和表达式的值，也就是完成输出的这一功能，所以选项 D 的说法正确；在执行方法时，不必考虑实现输出的具体步骤，方法的步骤是系统预先规定好的，所以选项 B 说法正确。

3.3　VB 程序设计采用的编程机制是（　　）。

A）可视化　　B）面向对象　　C）事件驱动　　D）过程结构化

【解答】正确答案为 C。

分析：可视化是一种程序设计技术，它把烦琐、复杂的工作交由系统完成，从而减少程序设计人员编写代码的工作量；面向对象是一种程序设计方法，这种方法将数据和代码封装起来而成为对象；事件驱动是一种编程机制，它由事件而驱动程序调用通用过程来执行指定的操作；过程结构化则是传统的“面向过程”程序设计语言的编程思想。

3.4　确定窗体控件启动位置的属性是（　　）。

A）Width 和 Height　　B）Width 或 Height

C）StartUpPosition　　D）Top 和 Left

【解答】正确答案为 C。

分析：控制窗体启动位置的是 StartUpPosition，所以选项 C 是正确的；而 Width 和 Height 用来控制窗体的大小，所以 A 和 B 选项不正确；Top 和 Left 用来控制窗体的左、右边界距屏幕的左、右边界的相对距离，所以选项 D 不正确。

3.5　下列说法正确的是（　　）。

A）对象的可见性可设为 True 或 False

B）标题的属性值不可设为任何文本

C）属性窗口中属性只能按字母顺序排列

D）某些属性的值可以跳过不设置，自动设为空值

【解答】正确答案为 A。

分析：在 VB 中，对象具有 Visible 属性，它有两个值：True 和 False，用来决定对象是否可见，如果设置为 True 则可见，如果设置为 False 则不可见，所以选项 A 说法是正确的；标题，即 Caption 属性可设为任何文本，包括空字符串，所以选项 B 说法是不正确的；属性的值可以不设置，但是系统会设为默认值，而不是空值，所以选项 D 说法是不正确的；属性窗口除按字母排列外，还可以分类排列，所以选项 C 说法是不正确的。

二、填空题

3.6　VB 对象可以分为两类，分别为____和____。

【解答】正确答案为：预定义对象；用户定义对象。

分析：对象分为两类：预定义对象和用户定义对象。预定义对象是由系统设计好的，可以直接使用或对其进行操作；而用户定义对象中的对象可由程序员自己定义，建立自己的对象。

三、思考题

3.7　试述可视化编程中对象、属性、事件和方法的含义。

【解答】在可视化编程语言中，对象是代码和数据的集合，它可以是窗体和控件，也可以是菜单或数据库等。从可视化编程的角度来看，这些对象都具有属性（数据）和方法（行为方式）。简单地说，属性是用于描述对象的一组特征，方法是对象实施的一些动作，对象的动作常常要触发事件，而触发事件又可以修改属性。一个对象建立以后，其操作就通过与该对象有关的属性、事件和方法来描述。

3.8　简述 VB 可视化编程的一般步骤。

【解答】

（1）设计界面。先建立窗体，再利用控件在窗体上创建各种对象。

（2）设置属性。设置窗体或控件等对象的属性。

（3）编写代码。编写对象的事件代码。

第 4 章　顺序结构程序设计

一、选择题

4.1　输入代码时，VB 可以自动检测（　　）错误。

A）语法　　B）编译　　C）运行　　D）逻辑

【解答】正确答案为 A。

分析：VB 编辑环境可以自动识别用户的输入是否符合语法规则。

4.2　在 VB 中，要将一个窗体加载到内存进行预处理但不显示，应使用的语句是（　　）。

A）Load　　B）Show　　C）Hide　　D）Unload

【解答】正确答案为 A。

分析：Load 语句把一个窗体装入内存，执行 Load 语句后，可以引用窗体中的控件及各种属性，但此时窗体没有显示出来，“窗体名称”是窗体的 Name 属性。

Unload 语句与 Load 语句的功能相反，它清除内存中指定的窗体。

Show 方法用来显示一个窗体。如果省略“窗体名称”，则显示当前窗体。

Hide 方法使窗体隐藏，即不在屏幕上显示，但仍在内存中，因此，它与 Unload 语句的作用是不同的。

4.3　以下能在窗体 Form1 的标题栏中显示"Visual Basic 窗体"的语句是（　　）。

A）Form1.Name="Visual Basic 窗体"　　B）Form1.Title="Visual Basic 窗体"

C）Form1.Caption="Visual Basic 窗体"　　D）Form1.Text="Visual Basic 窗体"

【解答】正确答案为 C。

分析：此题考查窗体的属性，Caption 属性用来设置窗体标题栏中显示的文本内容，题目要求在窗体标题栏中显示"Visual Basic 窗体"，只有选项 C 为正确答案。

4.4　对下列程序段，说法正确的是（　　）。

Text1.Top=2000　：　Text1.Left=800

A）Text 对象的左边距窗体的左边的距离为 800twip，上边距窗体的上边的距离为 2000twip

B）Text1 对象的左边距屏幕的左边的距离为 800twip，上边距屏幕的上边的距离为 2000twip

C）Text1 对象的宽度为 2000twip，高度为 800twip

D）Text1 对象的高度为 800 点，宽度为 2000 点

【解答】正确答案为 A。

分析：如果对象为窗体，Left 指的是窗体的左边与屏幕的左边之间的相对距离，Top 指的是窗体的顶边与屏幕的顶边之间的相对距离；而当对象为控件时，Left 指的是控件的左边与窗体的左边的相对距离，Top 指的是控件的顶边与窗体的顶边的相对距离。另外，Top 和

Left 属性值的单位为 twip，是 1 点（point）的二十分之一，Height 和 Width 为指定对象的高度和宽度的属性，所以只有选项 A 正确。

4.5　在运行程序时，在文本框中输入新的内容，或者在程序代码中改变 Text 的属性值，相应会触发（　　）事件。

A）GotFocus　　B）Click　　C）Change　　D）DblClick

【解答】正确答案为 C。

分析：在本题的 4 个选项中，GotFocus 是设置焦点事件，所以选项 A 不合题意；Click 是单击事件，不合题意；DblClick 是双击事件，也不合题意；Change 是改变文本框内容事件，只要文本框中的内容改变就会触发，故选项 C 是本题的答案。

4.6　单击窗体上的关闭按钮时，触发的事件是（　　）。

A）Form_Initialize()　　B）Form_Load()　　C）Form_Unload()　　D）Form_Click()

【解答】正确答案为 C。

分析：Initialize 事件和 Load 事件是由系统自动触发的事件，Form_Click 是单击窗体触发的事件，Unload 事件是清除窗体（关闭窗体或执行 Unload 语句）时触发的事件。

4.7　可以实现从键盘输入一个作为双精度变量 a 的值的语句是（　　）。

A）a=InputBox()　　B）a=InputBox("请输入一个值")

C）a=Val(InputBox("请输入一个值"))　　D）a=Val(InputBox())

【解答】正确答案为 C。

分析：InputBox 函数的格式为：

〈变量〉 = InputBox(〈信息内容〉[,〈对话框标题〉][,〈默认内容〉])

其中，〈信息内容〉是一个字符串，用于提示用户输入，是在对话框中显示的信息，不可省略，故选项 A 和选项 D 不正确；Val（字符串）函数的功能是把自变量中的第一个字符串转换为数值，其中的数值是一个双精度的实数，所以正确答案为选项 C；选项 B 没有类型转换符，不正确。

4.8　在属性窗口中设置（　　）属性，可以把指定的图形放入当前对象中。

A）CurrentY　　B）Picture　　C）CurrentX　　D）Stretch

【解答】正确答案为 B。

分析：打开对象的属性列表，从中选择 Picture 属性栏，单击后面的“…”按钮，将弹出“加载图片”对话框，用户选择相应的路径和文件名，图形就显示在当前对象中了，可见选项 B 是正确的；而 CurrentY 和 CurrentX 用来设置光标当前位置；Stretch 决定图像是否可以伸缩。

4.9　用于将屏幕上的对象分组的控件是（　　）。

A）列表框　　B）组合框　　C）标签　　D）框架

【解答】正确答案为 D。

分析：本题是考查控件的作用。列表框控件显示一个项目列表，让用户从其中选择一项或多项；组合框是文本框和列表框的集合，也可以像列表框一样，让用户通过鼠标选择所需要的项目；标签是用于显示信息的；框架用于将屏幕上的对象分组。

4.10　要将名为 MyForm 的窗体显示出来，正确的使用方法是（　　）。

A）MyForm.Show　　B）Show.MyForm　　C）MyForm Load　　D）MyForm Show

【解答】正确答案为 A。

分析：VB 中方法和语句的书写格式不同。

方法的格式为：对象.方法

语句的格式为：语句　对象名

使用方法要先写对象名，要用“.”运算符；使用语句要先写语句，语句后要空一格，后跟对象名。

4.11　如果要将窗体中的某个命令按钮设置成无效状态，应设置命令按钮的（　　）属性。

A）Value　　B）Visible　　C）Enabled　　D）Default

【解答】正确答案为 C。

分析：在 VB 中，控制控件有效性的属性是 Enabled，设置为 True，则处于有效状态，如果设置成 False，则处于无效状态，所以选项 C 是正确的；Visible 属性是控制控件可见性的，有两种值：True 和 False，默认为 True；如果将对象的 Visible 值设为 False，窗体和控件在设计阶段仍可见，只有运行时才隐藏，所以选项 B 不正确；Default 是决定窗体的默认命令的命令按钮的属性，另外，一个窗体只能有一个默认命令按钮，所以选项 D 也不正确；Value 属性在不同的控件中有不同的作用。

4.12　能够获得一个文本框中被选取文本的内容的属性是（　　）。

A）Text　　B）Length　　C）SelText　　D）SelStart

【解答】正确答案为 C。

分析：获得文本框中的文本需要用到 SelStart 属性（文本选取开始位置）、SelLength 属性（文本选取长度）和 SelText 属性（选取文本的内容）。

4.13　用 InputBox 函数设计的对话框，其功能是（　　）。

A）只能接收用户输入的数据，但不会返回任何信息

B）能接收用户输入的数据，并能返回用户输入的信息

C）既能用于接收用户输入的信息，又能用于输出信息

D）专门用于输出信息

【解答】正确答案为 B。

分析：InputBox 函数能接收用户输入的数据，并能返回用户输入的信息，不能用于输出信息。InputBox 函数中的 Prompt 参数在对话框中显示提示信息，这种提示信息是由用户确定的，不是专门用于输出信息的。

二、填空题

4.14　当对象得到焦点时，会触发____事件；当对象失去焦点时，会触发____事件。

【解答】正确答案为：GotFocus；LostFocus。

分析：在 VB 中，当对象得到焦点时，它可以接收用户的输入。当得到焦点时，会触发 GotFocus 事件；当失去焦点时，会触发 LostFocus 事件。

4.15　下列语句的输出结果为____。

```
Print Format $ (5689.36," 000,000.000")
```

【解答】正确答案为 005,689.360。

分析：用 Format＄()函数可以将数值按“格式字符串”指定的格式输出，包括在输出字符串前面加＄，字符串后面补 0 及千位分隔符等。“格式字符串”是一个常量或变量，它由专门的格式说明符组成，由这些字符决定数据项的显示格式，并指定显示区域的长度。当格式字符串为常量时，必须放在双引号中。本题指定 5689.36 按给定"000,000.000"格式输出，不足的部分补 0，输出结果为 005,689.360。

4.16　新建一个工程，内有两个窗体，窗体 Form1 上有一个命令按钮 Command1，单击该按钮，Form1 窗体消失，显示 Form2 窗体，试补全程序。

```
Private Sub Command1_Click()
    ____
    Form2.____
End Sub
```

【解答】正确答案为：Me.Hide 或 Form1.Hide 或 Unload Me；Show。

分析：解答此题首先要了解以下方法。

Show 方法：将窗体加载到内存中并显示。

Hide 方法：将窗体加载到内存中并隐藏。

Load 语句：加载窗体到内存中。

Unload 语句：从内存卸载窗体，Unload Me 卸载本窗体。

所以本题中，可以用 Hide 方法，也可以用 Unload 语句使 Form1 窗体消失；用 Show 方法显示窗体。

4.17　设有如下程序段：

```
a＄="BeijingShanghai"　:　b＄=Mid(a＄,InStr(a＄,"g")+1)
```

执行上面的程序段后，变量 b＄的值为_____。

【解答】正确答案为：Shanghai。

分析：InStr 函数的格式为：

InStr(［start,］ string1, string2 ［,compare］)

其作用是返回指定字符串 string2 在字符串 string1 中最早出现的位置。

Mid 函数的格式为：

Mid［＄］(string,start［,length］)

其作用是返回字符串中从 start 位置开始，长度为 length 的子字符串。

因此，变量 b$的值为 Shanghai。

4.18　为了使一个窗体从屏幕上消失但仍在内存中，所使用的方法或语句为_____。

【解答】正确答案为：Hide 方法。

分析：Hide 方法用来将窗体从屏幕上删除，但该窗体仍然留在内存中。

4.19　在文本框中要使输入的所有字符显示为*号，应设置_____属性为"*"。

【解答】正确答案为：PasswordChar。

分析：PasswordChar 属性返回或设置一个值，该值指示所输入的字符或占位符在 TextBox 控件中是否要显示出来。在对话框中创建一个密码可使用此属性。虽然能够使用任何字符，

但是大多数基于 Windows 的应用程序都使用星号（*）。

三、编程题

4.20　设计工程，已知圆的半径 *r*，求圆面积 *S*。

【解答】设圆半径为 *r*，圆面积为 *S*。根据数学知识，已知圆半径 *r*，求圆面积 *S* 的公式为：$S = \pi r^2$。

设计步骤如下。

（1）建立应用程序用户界面，如图 4-1 所示。

（2）设置对象属性：

Label1 的 Caption 属性为“已知圆半径 r=”；

Text1 的 Text 属性为空；

Command1 的 Caption 属性为“圆面积为：”；

Label2 的 Caption 属性为空；

Label2 的 BorderStyle 属性为 1-Fixed Single。

各控件的属性设置如图 4-2 所示。

（3）编写程序代码。

写出“圆面积为：”命令按钮 Command1 的 Click 事件代码为：

```
Private Sub Command1_Click( )
    Const pi = 3.14
    Dim r As Single, S As Single
    r = Val(Text1.Text)
    S = pi * r ^ 2
    Label2.Caption = S
End Sub
```

运行程序时，在文本框中输入圆半径的值，单击“圆面积为：”按钮后，输出结果如图 4-3 所示。



图 4-1　建立用户界面



图 4-2　设置各控件的属性

图 4-3　程序运行结果

也可以不用文本框接收输入值，改用 InputBox 函数接收圆的半径 r，求圆面积 S，代码如下：

```
Private Sub Form_Load( )
    Show
    Const pi = 3.1415926
    Dim r As Single, S As Single
    r = Val(InputBox("输入半径：", "计算圆面积", "10"))
```

```
    FontSize = 18
    S = pi * r ^ 2
    Print "圆面积："; S
End Sub
```

程序运行时，首先显示如图 4-4 所示的对话框，在该对话框的文本框中输入数字，按 Enter 键或单击“确定”按钮后，才能显示窗体。

图 4-4　输入对话框

用 InputBox 函数输入文本虽然很方便，但是由于输入框弹出后将暂停程序的运行，直到用户响应，因此输入框不符合 VB 自由环境的精神。输入框适合于像要求用户输入口令等这样不常见的输入方式。还可以用更好的用户输入方式，如文本框、选项按钮等。

4.21　已知平面坐标系中两点的坐标，求两点间的距离。

【解答】由数学知识可知，已知两点坐标（x_A,y_A）、（x_B,y_B），求两点间距离的计算公式为

$$s=\sqrt{(x_B-x_A)^2+(y_B-y_A)^2}$$

建立用户界面如图 4-5 所示。在该界面中用 TextBox 控件输入数据，用 Label 控件输出数据。为了形象地表示两点之间的距离，可用 Picture 控件插入一幅图，该图用画图软件绘制。

命令按钮 Command1 的 Click 事件代码为：

```
Private Sub Command1_Click( )
    Dim xa As Single, xb As Single
    Dim ya As Single, yb As Single
    Dim s As Single
    xa = Val(Text1.Text)
    ya = Val(Text2.Text)
    xb = Val(Text3.Text)
    yb = Val(Text4.Text)
    s = Sqr((xb - xa) ^ 2 + (yb - ya) ^ 2)
    Label6.Caption = s
End Sub
```

程序运行结果如图 4-6 所示。



图 4-5　建立用户界面



图 4-6　程序运行结果

4.22　设计工程，输出在指定范围内的 3 个随机数，范围在文本框中输入。

【解答】随机函数 Rnd( )可以返回一个（0,1）区间中的随机小数，那么，Rnd * a 语句可以返回（0,a）区间中的随机实数（带小数）。

若 n, m 均为整数，则表达式：Int((m + 1 – n) * Rnd) + n 的值是闭区间[n,m]中的一个随机整数。

设计步骤如下。

（1）设计程序界面及设置控件属性。新建一个工程，进入窗体设计器，在窗体中增加一个框架控件 Frame1，一个命令按钮 Command1 和 3 个标签 Label1～Label3。选定 Frame1，在其中增加两个文本框 Text1 和 Text2，以及一些标签。修改对象属性见表 4-1。设置属性后的窗体如图 4-7 所示。

表 4-1　属性设置

对　　象	属　　性	属　性　值	说　　明
Frame1	Caption	请指定随机整数的范围：	框架的标题
Command1	Caption	生成随机数	按钮的标题
Label1～Label3	Caption	0	
Text1, Text2	Text	0, 1	

（2）编写程序代码。

```
Private Sub Command1_Click( )
    Randomize
    n = Val(Text1.Text)
    m = Val(Text2.Text)
    Label4.Caption = Int((m + 1 – n) * Rnd) + n
    Label5.Caption = Int((m + 1 – n) * Rnd) + n
    Label6.Caption = Int((m + 1 – n) * Rnd) + n
End Sub
```

运行程序，在文本框中输入范围值后，单击“生成随机数”按钮，可以不断生成指定范围之内的随机整数，如图 4-8 所示。



图 4-7　设计用户界面



图 4-8　生成随机数

4.23　设某职工应发工资 *x* 元，试求各种面额钞票总张数最少的付款方案。

【解答】可以从最大面额的钞票（100 元）开始，算出所需的张数，然后对剩下的钱数算出较小面额钞票的张数，直到最小面额（1 元）。

设计步骤如下。

（1）建立应用程序用户界面。新建一个工程，进入窗体设计器，增加一个命令按钮 Command1，14 个标签 Label1～Label14，7 个文本框 Text1～Text7。用户界面如图 4-9（a）所示。



（a）　　（b）

图 4-9　建立程序界面与程序运行结果

（2）设置对象属性，见表 4-2。其中标签的标题属性如图 4-9（a）所示。

表 4-2　属性设置

对　象	属　性	属 性 值	说　明
Text1	Text	0	文本框的内容
Text2～Text7	Text		文本框的内容
	Alignment	1-Right Justify	文本内容右对齐
	Locked	True	文本内容只读
Command1	Caption	应付	按钮标题
	Default	True	窗体的默认按钮

（3）编写程序代码。

编写命令按钮 Command1 的 Click 事件代码：

```
Private Sub Command1_Click( )
    x = Val(Text1.Text)              ' x 为实发工资数
    y = x \ 100 : Text2.Text = y     ' 求 100 元票张数并显示
    x = x – 100 * y                  ' 求剩余款项
    y = x \ 50 : Text3.Text = y      ' 求 50 元钞票张数并显示
    x = x – 50 * y                   ' 求剩余款项
    y = x \ 10 : Text4.Text = y      ' 求 10 元钞票张数并显示
    x = x – 10 * y                   ' 求剩余款项
    y = x \ 5   : Text5.Text = y     ' 求 5 元钞票张数并显示
    x = x – 5 * y                    ' 求剩余款项
    y = x \ 2 : Text6.Text = y       ' 求 2 元钞票张数并显示
    x = x – 2 * y : Text7.Text = x   ' 求 1 元钞票张数并显示
End Sub
```

编写文本框 Text1 的 Change 事件代码：

```
Private Sub Text1_Change( )      ' 当在 Text1 中输入新数时，清空其他文本框中的内容
    Text2.Text = ""
    Text3.Text = ""
```

```
    Text4.Text = ""
    Text5.Text = ""
    Text6.Text = ""
    Text7.Text = ""
End Sub
```

运行结果如图 4-9（b）所示。

【说明】文本框的 Change 事件在文本框中的内容被改变时发生。

下面用信息对话框来输出各种面额钞票张数的付款方案，如图 4-10 所示。

图 4-10　信息对话框

只需将图 4-11 中多余的文本框和标签删去，并改写命令按钮的 Click 事件代码如下：

```
Private Sub Command1_Click( )
    x = val(Text1.Text)              ' x 为实发工资数
    y1 = x \ 100                     ' 求 100 元钞票张数
    x = x Mod 100                    ' 求剩余款项
    y2 = x \ 50                      ' 求 50 元钞票张数
    x = x Mod 50                     ' 求剩余款项
    y3 = x \ 10                      ' 求 10 元钞票张数
    x = x Mod 10                     ' 求剩余款项
    y4 = x \ 5                       ' 求 5 元钞票张数
    x = x Mod 5                      ' 求剩余款项
    y5 = x \ 2                       ' 求 2 元钞票张数
    x = x Mod 2                      ' 求 1 元钞票张数
    Text1.SelStart = 0
    Text1.SelLength = Len(Text1.Text)
    Text1.SetFocus
    a = "==========================" & Chr(13)
    a = a & Format(Y1 & "张  百元票,", "@@@@@@@@@@") & _
            Format(Y2 & "张  50 元票", "@@@@@@@@@@") & Chr(13)
    a = a & Format(Y3 & "张  10 元票,", "@@@@@@@@@@") & _
```

```
            Format(Y4 & "张   5 元票", "@@@@@@@@@@") & Chr(13)
    a = a & Format(Y5 & "张   2 元票,", "@@@@@@@@@@@") & _
            Format(x & "张   1 元票", "@@@@@@@@@@") & Chr(13)
    a = a & "==========================" & Chr(13)
    a = a & "共计" & Text1.Text & "元"
    c = MsgBox(a, 0, "应发工资")
End Sub
```

【说明】文本框的 SelStart 属性用来设置（或返回）所选择的文本的起始点，如果没有文本被选中，则指出插入点的位置。

SelLength 属性用来设置（或返回）所选择的字符数。函数 Len( )返回字符串数据的长度。在命令按钮的 Click 事件代码中调用 SetFocus 方法，可使光标重新回到输入框 Text1。使用 Mod 运算可以得到相除的余数。

4.24　使用大小写转换函数设计程序，实现在文本框中输入英文字母，按“转大写”按钮，文本变为大写；按“转小写”按钮，文本变为小写。

【解答】本题主要使用大小写转换函数 UCase( )和 LCase( )，另外还用到了 KeyUp（键抬起）事件。

设计步骤如下。

（1）建立应用程序用户界面，并设置对象属性。新建一个工程，进入窗体设计器，在窗体中增加一个标签 Label1，一个文本框 Text1 和 3 个命令按钮 Command1～Command3，参照图 4-11 设置各个对象的属性。

（a）　　（b）　　（c）

图 4-11　大小写转换

（2）设计代码。

首先设计文本框的 KeyUp 事件代码，使得输入的字符存入文本框的 Tag 属性中：

```
Private Sub Text1_KeyUp(KeyCode As Integer, Shift As Integer)
    Text1.Tag = Text1.Text
End Sub
```

编写命令按钮 Command1 的 Click 事件代码：

```
Private Sub Command1_Click( )
    Text1.Text = UCase(Text1.Tag)
End Sub
```

编写命令按钮 Command2 的 Click 事件代码：

```
Private Sub Command2_Click( )
```

```
    Text1.Text = LCase(Text1.Tag)
End Sub
```

编写命令按钮 Command3 的 Click 事件代码：

```
Private Sub Command3_Click( )
    Text1.Text = Text1.Tag
End Sub
```

4.25　在文本框中输入 3 种商品的单价、购买数量，计算并输出所用的总金额。

【解答】假设第一种商品的单价和购买数量分别是 a1 和 a2，第二种商品的单价和购买数量分别是 b1 和 b2，第三种商品的单价和购买数量分别是 c1 和 c2，所用总金额为 x。通过 TextBox 控件输入这 3 种商品的单价、购买数量，然后利用公式 x = a1*a2+b1*b2+c1*c2 进行计算，最后输出 x 的值。

设计步骤如下。

（1）建立应用程序用户界面，并设置对象属性，如图 4-12 所示。

（2）编写程序代码。

编写 Command1_Click( )的事件代码：

```
Private Sub Command1_Click( )
    Dim a1 As Single, b1 As Single, c1 As Single
    Dim a2 As Integer, b2 As Integer, c2 As Integer
    Dim x As Single
    a1 = Val(Text1.Text)
    a2 = Val(Text2.Text)
    b1 = Val(Text3.Text)
    b2 = Val(Text4.Text)
    c1 = Val(Text5.Text)
    c2 = Val(Text6.Text)
    x = a1 * a2 + b1 * b2 + c1 * c2
    Text7.Text = x
End Sub
```

4.26　在文本框中输入弧度值，将弧度换算为角度值（度、分、秒）的形式，然后输出。例如，弧度值为 1.474 919 573，化为角度的方法为：

（1）先将弧度值变成十进制数，即 1.474 919 573×(180/π) = 84.506 666 65。

（2）去掉整数部分 84，余 0.506 666 65。

（3）0.506 666 65×60 = 30.399 999。

（4）去掉 30，余 0.399 999。

（5）0.399 999×60 = 23.999 94≈24。

（6）最后将 84, 30, 24 拼接成 84° 30' 24"。

【解答】假设某弧度值为 x，角度值为 y。

设计步骤如下。

（1）建立应用程序用户界面，并设置对象属性，如图 4-13 所示。



图 4-12　计算商品总金额

图 4-13　将弧度值换算为角度值

（2）编写程序代码。

写出 Command1 的 Click 事件代码：

```
Private Sub Command1_Click( )
    Const pi = 3.1415926
    Dim x As Single, a As Single, a1 As Single
    Dim d As Integer, f As Integer, m As Integer
    Dim y As String
    x = Val(Text1.Text)
    a = x * (180 / pi)
    d = Fix(a)
    a1 = (a – d) * 60
    f = Fix(a1)
    m = Fix((a1 – f) * 60 + 0.5)
    y = Str(d) & "° " & Str(f) & "′ " & Str(m) & "″ "          ' 符号 ° 、′ 、″ 为全角
    Label1.Caption = y
End Sub
```

第 5 章　选择结构程序设计

一、选择题

5.1　下列程序段的执行结果为（　　）。

```
X=2
Y=5
If X * Y <1 Then Y=Y−1 Else Y=−1
Print Y−X>0
```

A）True　　B）False　　C）−1　　D）1

【解答】正确答案为 B。

分析：Print 方法具有计算和输出双重功能，对于表达式，它先计算后输出，此题中经过第一步的条件语句后，Y 的数值为−1，则 Y−X 的值为−3，小于 0，所以 Y−X>0 为逻辑假，故输出结果应该为 False。

5.2　下面语句正确的是（　　）。

A）If x<3*y And x>y Then y=x^3　　B）If x<3*y And x>y Then y=3x

C）If x<3*y : x>y Then y=x^3　　D）If x<3*y : x>y Then y=x**3

【解答】正确答案为 A。

分析：根据条件语句的结构和条件表达式的格式进行判断。

5.3　下列程序段执行结果为（　　）。

```
x=5
y=-6
If Not x>0 Then x=y−3 Else y=x+3
Print x−y;y−x
```

A）−3　3　　B）5　−9　　C）3　−3　　D）−6　5

【解答】正确答案为 A。

分析：程序开始时分别为 x,y 赋值 5,−6。If 语句的条件 Not x>0 相当于 x<=0，现在 x 的值为 5，比 0 大，所以条件值为 False，执行 Else 语句 y=x+3，此时 y 为 8，执行 Print 方法，即输出 x−y 和 y−x 的值，x−y=−3，y−x=3。

5.4　下列语句正确的是（　　）。

A）If A ≠B Then Print "A 不等于 B"

B）If A < >B Then Printf "A 不等于 B"

C）If A < >B Then Print "A 不等于 B"

D）If A ≠B Print "A 不等于 B"

【解答】正确答案为 C。

分析：题中条件语句都属于条件语句中的“If 条件 Then 语句”类型，从选项中可以看出，条件语句是 A 与 B 的比较语句，其值就是此条件语句的值，也就是说，只有当其值为真时，才执行 Then 语句。选项 A 和 D 中的“≠”符号并非比较运算符，不正确；选项 B 中的 Printf 语句并非 Visual Basic 中的输出方法，不正确，所以只有选项 C 是正确的。

5.5 计算 z 的值，当 x 大于 y 时，z=x；否则 z=y。下列语句错误的是（　　）。

A）If x>=y Then z=x : z=y　　B）If x>=y Then z=x Else z=y

C）z=y : If x>=y Then z=x　　D）If x<=y Then z=y Else z=x

【解答】正确答案为 A。

分析：选项 A 中的语句行有两条语句，执行 If 语句后，执行 z=y 语句，所以不管 x 是否大于 y，最终 z 的值都等于 y。

5.6 下列程序段的执行结果为（　　）。

```
a=95
If a>60 Then I=1
If a>70 Then I=2
If a>80 Then I=3
If a>90 Then I=4
Print "I="; I
```

A）I=1　　B）I=2　　C）I=3　　D）I=4

【解答】正确答案为 D。

分析：此题为 If…Then 结构的条件语句，如果 a>60，则 I=1；如果 a>70，则 I=2；如果 a>80，则 I=3；如果 a>90，则 I=4。

5.7 下面程序段执行结果为（　　）。

```
x=Int(Rnd()+4)
Select Case x
   Case 5
      Print "excellent"
   Case 4
      Print "good"
   Case 3
      Print "pass"
   Case Else
      Print "fail"
End Select
```

A）excellent　　B）good　　C）pass　　D）fail

【解答】正确答案为 B。

分析：在 x=Int(Rnd()+4)语句中，Rnd 是用来产生随机数的，其值在 0～1 之间，而在(Rnd()+4)前面有 Int 进行强制转换，所以 x 总为 4，正确答案是选项 B。

5.8　以下程序段运行时，若从键盘输入字符“–”，则输出结果为（　　）。

```
op $ =InputBox("op=")
If op $ ="+" Then a=a +2
If op $ ="–" Then a=a – 2
Print a
```

A）2　　B）–2　　C）0　　D）+2

【解答】正确答案为 B。

分析：此题根据用户的输入计算变量 a 的值并显示。当输入字符“–”时，第一个 If 条件不成立，跳过，而第二个 If 语句条件满足，因此执行 a=a–2，使 a=–2。所以选项 B 是正确答案。

5.9　在窗体上画一个名称为 Timer1 的计时器控件，要求每隔 0.5 秒发生一次计时器事件，则以下正确的属性设置语句是（　　）。

A）Timer1.Interval=0.5　　B）Timer1.Interval=5

C）Timer1.Interval=50　　D）Timer1.Interval=500

【解答】正确答案为 D。

分析：计时器控件的 Interval 属性用来指定计时器事件之间的毫秒数，本题要求每隔 0.5 秒（即 500 毫秒）发生一次，所以其值为 500，正确答案为选项 D。

二、填空题

5.10　有下面一个程序段，从文本框中输入数据，如果该数据满足条件：除以 4 余 1 且除以 5 余 2，则输出；否则，将焦点定位在文本框中，并清除文本框的内容。请补全程序。

```
Private Sub Command1_Click()
  x=Val(Text1.Text)
  If  ____  Then
    Print x
  Else
    Text1.Text=""
    ____
  End If
End Sub
```

【解答】正确答案为：x Mod 4 = 1 And x Mod 5 = 2；Text1.Setfocus。

分析：程序调用 Val()函数从文本框中得到数据。本题中先通过 Val()函数得到一个整数，进行条件判断，如果满足条件 x Mod 4 = 1 And x Mod 5 = 2 则输出 x；否则，将清空文本框，使焦点落在文本框中。

5.11　给定年份，下列程序用来判断该年是否是闰年。请补全程序。

```
Sub YN()
  Dim x As Integer
  x=InputBox("请输入年号")
  If (x Mod 4=0  ____  x Mod 100 < > 0)  ____  (x Mod 400=0) Then
```

```
        Print "是闰年"
    Else
        Print "不是闰年"
    End If
End Sub
```

【解答】正确答案为：And；Or。

分析：闰年的条件是年份能被 4 整除不能被 100 整除，或者能被 400 整除，由此可以写出逻辑关系式：(x Mod 4 = 0 And x Mod 100 < > 0)Or(x Mod 400 = 0)。

5.12　下列程序的功能是：当 $x<50$ 时，$y=0.8x$；当 $50\leqslant x\leqslant 100$ 时，$y=0.7x$；当 $x>100$ 时，没有意义。请补全程序。

```
Private Sub Command1_Click()
    Dim x As Single
    x=InputBox("输入 x 的值")
    ____
    Case Is<50
        y=0.8 * x
    Case 50 To 100
        y=0.7 * x
    ____
    Print "输入的数据出界！"
    End Select
    Print x , y
End Sub
```

【解答】正确答案为：Select Case x；Case Else。

分析：可以看到，程序中有 Case 和 Select 子句，没有多分支选择结构的起始语句。很明显，第一个空应该填写多分支结构的起始语句 Select Case x。在 Select Case 结构中，只有两个 Case 子句，分别表示 x 取值的两种情况，但题目中将 x 的值分为 3 种情况，可见第二个空应该填写 Case Else 语句。

5.13　在窗体上画一个文本框和一个计时器控件，名称分别为 Text1 和 Timer1，在属性窗中把计时器的 Interval 属性设置为 100，Enabled 属性设置为 False。程序运行后，如果单击命令按钮，则每隔 1 秒在文本框中显示一次当前的时间。请补全程序。

```
Private Sub Command1_Click()
    Timer1.____
End Sub
Private Sub Timer1_Timer()
    Text1.Text=Time
End Sub
```

【解答】正确答案为 Enabled=True。

分析：本题考查的是时钟控件 Timer，该控件的 Interval 属性控制两个计时器事件之间的时间间隔，其值以 ms 为单位，而当 Enabled 属性为 False 时，时钟控件不起作用，故欲使程

序实现每隔 1 秒在文本框中显示一次当前的时间，其前提就是将时钟控件的 Enabled 属性设置为 True。

三、编程题

5.14　编写计算铁路运费的程序。假设铁路托运行李，规定每张客票托运费计算方法是：行李重不超过 50 千克时，每千克 0.25 元；超过 50 千克而不超过 100 千克时，其超过部分每千克 0.35 元；超过 100 千克时，其超过部分每千克 0.45 元。要求输入行李重量，可计算并输出托运的费用。

【分析】设行李重量为 w 千克，应付运费为 x 元，则运费公式为：

$$x=\begin{cases}0.25w & (w\leqslant 50)\\ 0.25\times 50+0.35\times(w-50) & (50<w\leqslant 100)\\ 0.25\times 50+0.35\times 50+0.45\times(w-100) & (w>100)\end{cases}$$

设计步骤如下。

（1）建立应用程序用户界面，并设置对象属性，如图 5-1 所示。

（2）编写程序代码。

写出命令按钮 Command1 的 Click 事件代码为：

```
Private Sub Command1_Click( )
  Dim w As Single, x As Single
  w = Val(Text1.Text)
  If w <= 50 Then
    x = 0.25 * w
  Else
    If w <= 100 Then
      x = 0.25 * 50 + 0.35 * (w  -  50)
    Else
      x = 0.25 * 50 + 0.35 * 50 + 0.45 * (w  -  100)
    End If
  End If
  Text2.Text = x
End Sub
```

【说明】IIf 函数也可以嵌套使用。将命令按钮 Command1 的 Click 事件代码改为：

```
Private Sub Command1_Click( )
  Dim w As Single, x As Single
  w = Val(Text1.Text)
  x = IIf(w <= 50, 0.25 * w, 0.25 * 50 + IIf(w <= 100, 0.35 * (w  -  50), 0.35 * 50 + 0.45 * (w  -  100)))
  Text2.Text = x
End Sub
```

5.15　若基本工资大于等于 600 元，增加 20%工资；若小于 600 元，且大于等于 400 元，则增加 15%工资；若小于 400 元，则增加 10%工资。请根据用户输入的基本工资，计算出增加后的工资。

【解答】设计步骤如下。

（1）建立应用程序用户界面，并设置对象属性，如图 5-2 所示。



图 5-1　计算托运费



图 5-2　计算基本工资

（2）编写程序代码。

编写文本框 Text1 的 KeyPress（按键）事件代码：

```
Private Sub Text1_KeyPress(KeyAscii As Integer)
  If KeyAscii = 13 Then
    n = Val(Text1.Text)
    Select Case n
      Case Is >= 600
        gz = n * 1.2
      Case Is >= 400
        gz = n * 1.15
      Case Else
        gz = n * 1.1
    End Select
    Label1.Caption = "增加后的基本工资为：" & gz & "元"
    Text1.SelStart = 0
    Text1.SelLength = Len(Text1.Text)
  End If
End Sub
```

5.16　求一元二次方程 $ax^2+bx+c=0$ 的根。

【解答】一元二次方程的系数 a, b, c 的取值有以下 3 种情况。

① 当 $a \neq 0$ 时，有两个根。

设 delta = b^2-4ac，则

- 当 delta > 0 时，有两个不同的实根；
- 当 delta = 0 时，有两个相同的实根；
- 当 delta < 0 时，有两个不同的虚根。

② 当 $a = 0$，$b \neq 0$ 时，有一个根。

③ 当 $a = 0$，$b = 0$ 时，方程无意义。

设计步骤如下。

（1）建立应用程序用户界面，并设置对象属性，如图 5-3 所示。

（2）编写代码。

编写命令按钮 Command1 的 Click 事件代码：

```
Private Sub Command1_Click( )
   Dim a As Single, b As Single, c As Single
   Dim sb As Single, xb As Single, re As Single
   a = Val(Text1.Text)
   b = Val(Text2.Text)
   c = Val(Text3.Text)
   If a < > 0 Then                          ' 有两个根
      delta = b ^ 2 – 4 * a * c
      re = –b / (2 * a)
      If delta > 0 Then                     ' 方程有两个实根
         sb = Sqr(delta) / (2 * a)
         Label2.Caption = "方程有两个实根"
         p1 = "x1 = " & Str(re + sb)
         p2 = "x2 = " & Str(re – sb)
         Label3.Caption = p1 & Chr(13) & p2
      ElseIf delta = 0 Then                 ' 方程有两相等实根
         Label2.Caption = "方程有两个相等实根"
         Label3.Caption = "x1 = x2 = " & Str(re)
      Else                                  ' 方程有两个虚根
         xb = Sqr(–delta) / (2 * a)
         Label2.Caption = "方程有两个虚根"
         p1 = "x1 = " & Str(re) & "+" & IIf(xb = 1, "", Str(xb)) & "i"
         p2 = "x2 = " & Str(re) & "–" & IIf(xb = 1, "", Str(xb)) & "i"
         Label3.Caption = p1 & Chr(13) & p2
      End If
   Else
      If b < > 0 Then                       ' 方程仅有一个根
         ygz = –b / c
         Label2.Caption = "方程仅有一个根"
         Label3.Caption = "x=" & Str(ygz)
      Else                                  ' 方程无意义
         Label2.Caption = "方程无意义！"
         Label3.Caption = ""
      End If
   End If
End Sub
```

运行程序，在文本框中输入方程的系数，按“确定”按钮即可判断方程有无实根等情况，并且求出根来，如图 5-3 所示。



（a）

（b）

图 5-3　求一元二次方程的根

5.17　编写一个对输入字符进行大、小写转换的程序。转换规则为：将其中的大写字母转换成小写字母，小写字母转换成大写字母，空格不转换，其余转换成“*”。要求每输入一个字符后，马上就进行判断和转换。

【解答】设计步骤如下。

（1）建立应用程序用户界面，如图 5-4 所示。

（2）设置对象属性，如图 5-5 所示。

（3）编写事件代码。

根据题中要求，在“输入字符串”文本框中每输入一个字符键后，马上就进行判断，这就要求对“输入字符串”文本框对象 Text1 对应的 KeyPress 事件进行编程。

```
Private Sub Text1_KeyPress(KeyAscii As Integer)
  Dim aa As String * 1
  aa = Chr$(KeyAscii)                    ' 将 ASCII 码转换成字符
  Select Case aa
    Case "A" To "Z"                      ' 大写转换成小写
      aa = Chr$(KeyAscii + 32)
    Case "a" To "z"                      ' 小写转换成大写
      aa = Chr$(KeyAscii – 32)
    Case " "
      aa = " "
    Case Else
      aa = "*"
  End Select
  Text2.Text = Text2.Text & aa           ' 将转换文本框中已有的内容与刚输入并转换的字符连接
End Sub
```

编写命令按钮 Command1 的 Click 事件代码：

```
Private Sub Command1_Click( )
  Text1.Text = ""                        ' 清除文本框 Text1 中的内容
  Text2.Text = ""                        ' 清除文本框 Text2 中的内容
End Sub
```

编写命令按钮 Command2 的 Click 事件代码：

```
Private Sub Command2_Click( )
    Unload Me
End Sub
```

程序运行结果如图 5-6 所示。



图 5-4　设置用户界面



图 5-5　设置对象属性

图 5-6　大小写字母转换

5.18　设计个人资料输入界面，使用单选钮组输入性别与民族，用复选框输入个人爱好。

【解答】由于本题要求选择的项目较多，所以需要使用框架控件。

设计步骤如下。

（1）建立应用程序用户界面。

新建一个工程，进入窗体设计器，首先增加 3 个标签 Label1～label3，一个文本框 Text1，两个单选钮 Option1～Option2，一个命令按钮 Command1 及 5 个框架控件 Frame1～Frame5（如图 5-7 所示）。依次激活框架控件 Frame1～Frame5 后，在其中分别增加以下内容。

在 Frame1 中：增加标签 Label1，文本框 Text1。

在 Frame2 中：增加标签 Label2，单选钮 Option1～Option2。

在 Frame3 中：增加标签 Label3，单选钮 Option3～Option4。

在 Frame4 中：增加复选框 Check1～Check4。

在 Frame5 中：增加标签 Label4。

（2）设置对象属性。其中单选钮与复选框的属性，见表 5-1。

表 5-1　属性设置

对　象	属　性	属 性 值
Option1	Caption	男
	Value	True
	Style	1-Graphical
Option2	Caption	女
	Style	1-Graphical
Option3	Caption	汉族
	Value	True
	Style	1-Graphical
Option4	Caption	少数民族
	Style	1-Graphical
Check1	Caption	钓鱼
Check2	Caption	下棋
Check3	Caption	读书
Check4	Caption	打球

其他属性的设置如图 5-7 所示。

（3）编写程序代码。

编写命令按钮 Command1 的 Click 事件代码：

```
Private Sub Command1_Click( )
  If Text1.Text = "" Then
    a = InputBox("您忘了输入姓名！", "注意", "请在此输入姓名")
    If a = "" Or a = "请在此输入姓名" Then Exit Sub
    Text1.Text = a
  End If
  p1 = Text1.Text + "，"
  p2 = IIf(Option1, "男", "女") + "，"
  p3 = IIf(Option3, "汉族", "少数民族")
  p4 = "，喜欢："
  If Check1.Value = 1 Then p4 = p4 + Check1.Caption + "、"
  If Check2.Value = 1 Then p4 = p4 + Check2.Caption + "、"
  If Check3.Value = 1 Then p4 = p4 + Check3.Caption + "、"
  If Check4.Value = 1 Then p4 = p4 + Check4.Caption + "、"
  aa = p1 + p2 + p3 + IIf(p4 = "，喜欢：", "，无爱好。", p4)
  Label4.Caption = Left(aa, Len(aa) - 1) + "。"
  Text1.SetFocus
End Sub
```

编写文本框 Text1 的 Change 事件代码：

```
Private Sub Text1_Change( )
  Label4.Caption = ""
End Sub
```

程序运行结果如图 5-8 所示。

图 5-7　建立用户界面



图 5-8　输入个人资料

5.19　利用单选钮组控制输入文本的字体。

【解答】设计步骤如下。

（1）建立应用程序用户界面并设置对象属性。

新建一个工程，进入窗体设计器，增加一个标签 Label1，一个文本框 Text1 和 4 个单选钮 Option1～Option4，如图 5-9（a）所示。然后设置对象属性，见表 5-2。

表 5-2　属性设置

对　象	属　性	属 性 值	说　明
Label1	Caption	请输入文本	
Text1	Text		清空
Option1	Caption	宋体	
	Value	True	单选钮组中的默认按钮
Option2	Caption	隶书	
Option3	Caption	黑体	
Option4	Caption	楷体	

为文本设置适当的字体大小，如图 5-9（b）所示。



（a）　　（b）

图 5-9　控制输入文本的字体

（2）编写程序代码。

编写单选钮 Option1 的 Click 事件代码：

```
Private Sub Option1_Click( )
    Text1.FontName = "宋体"
End Sub
```

编写单选钮 Option2 的 Click 事件代码：

```
Private Sub Option2_Click( )
    Text1.FontName = "隶书"
End Sub
```

编写单选钮 Option3 的 Click 事件代码：

```
Private Sub Option3_Click( )
    Text1.FontName = "黑体"
End Sub
```

编写单选钮 Option4 的 Click 事件代码：

```
Private Sub Option4_Click( )
    Text1.FontName = "楷体_GB2312"
End Sub
```

【说明】一个单选钮可以用以下方法之一进行选择：

- 在运行期间单击单选钮；
- 用 Tab 键定位到单选钮组，然后在组内使用方向键（箭头键）定位单选钮；

- 用代码将单选钮的 Value 属性设置为 True；
- 按下在 Label 的标题中指定的快捷键，并结合 TabIndex 属性使用。

5.20　文本框的 PasswordChar 属性可以隐蔽用户通过键盘输入的字符。编写程序，利用文本框检查用户口令。

【解答】设计步骤如下。

（1）建立应用程序用户界面，如图 5-10 所示。



（a）　　（b）

图 5-10　检查用户口令

（2）设置对象属性，见表 5-3。

表 5-3　属性设置

对　象	属　性	属 性 值	说　明
Frame1	Caption	请输入您的口令：	
Text1	Text		清空
	PasswordChar	*	只显示字符“*”
Label1	Caption		
	FontName	黑体	字体名称
	FontSize	三号	字体大小

（3）编写代码。

只需编写文本框 Text1 的 KeyPress 事件代码：

```
Private Sub Text1_KeyPress(KeyAscii As Integer)
   If KeyAscii = 13 Then
      If LCase(Text1.Text) = "abcde" Then
   Label1.Caption = "欢迎使用本系统"
Else
   Label1.Caption = "对不起，口令错！"
End If
Text1.SelStart = 0
Text1.SelLength = Len(Text1.Text)
   End If
End Sub
```

5.21　设计一个计时器，能够设置倒计时的时间，并进行倒计时。

【解答】利用计时器控件来控制倒计时，然后通过文本框输入设定的时间。

设计步骤如下。

（1）建立应用程序用户界面并设置属性。

新建一个工程，进入窗体设计器，首先在窗体中增加一个框架 Frame1，一个命令按钮 Command1 和一个计时器控件 Timer1。选定框架 Frame1，在其中增加一个文本框 Text1，如图 5-11 所示。



图 5-11　设计窗体界面

属性设置见表 5-4。

表 5-4　属性设置

对　象	属　性	属 性 值	说　明
Frame1	Caption	请输入倒计时的分钟数：	
Command1	Caption	开始	
	Default	True	设置默认按钮
Timer1	Interval	1000	
	Enabled	False	

（2）编写代码。

编写命令按钮 Command1 的 Click 事件代码：

```
Private Sub Command1_Click( )
  Timer1.Enabled = True
  Timer1.Tag = Text1.Text * 60
  Frame1.Caption = "现在开始倒计时"
End Sub
```

编写计时器 Timer1 的 Timer 事件代码：

```
Private Sub Timer1_Timer( )
  Timer1.Tag = Timer1.Tag – 1
  m = Timer1.Tag
  If m < 0 Then
    Timer1.Enabled = False
    MsgBox "预定的时间到了!", 0, "倒计时"
    Frame1.Caption = "请输入计时的分钟数:"
    Text1.Text = 0
    Exit Sub
  End If
  n1 = Format(m Mod 60, "00")
  n2 = Format((m \ 60) Mod 60, "00:")
  n3 = Format(m \ 3600, "00:")
  Text1.Text = n3 & n2 & n1
End Sub
```

运行程序，输入时间后按“开始”按钮，即开始倒计时，如图 5-12（a）所示。计时结束后将弹出对话框，如图 5-12（b）所示。



（a）　　（b）

图 5-12　倒计时器

5.22　利用图形复选框来控制文本的字体风格。

【解答】设计步骤如下。

（1）建立应用程序用户界面。

新建一个工程，进入窗体设计器，首先在窗体上增加 3 个框架 Frame1～Frame3。依次激活框架 Frame1～Frame3 后，在其中分别增加以下内容。

在 Frame1 中：增加标签 Label1，文本框 Text1。

在 Frame2 中：增加标签 Label2，复选框 Check1。

在 Frame3 中：增加复选框 Check2～Check4。

（2）设置对象属性。

只介绍 4 个复选框 Check1～Check4 的属性设置方法，其余如图 5-13 所示。

依次选中 4 个复选框 Check1～Check4，将其 Style 属性全部改为：1-Graphical（图形方式），将 Picture（图片）属性通过浏览按钮“…”进行查找，并分别改为：

\program files\microsoft visual studio\common\graphics\icons\misc\secur01a.ico

\program files\microsoft visual studio\common\graphics\bitmaps\tlbr_w95\bld.bmp

\program files\microsoft visual studio\common\graphics\bitmaps\tlbr_w95\itl.bmp

\program files\microsoft visual studio\common\graphics\bitmaps\tlbr_w95\undrln.bmp



（a）　　（b）

图 5-13　控制文本的字体风格

Check1 的 DownPicture 属性改为：

\program files\microsoft visual studio\common\graphics\icons\misc\secur01b.ico

Check1 的 Caption 属性改为：锁上；

Check1 的 Value 属性改为：1-Checked；

其他 Check2～Check4 的 Caption 属性改为：（无）。

（3）编写代码。

复选框控件 Check1 的 Click 事件代码：

```
Private Sub Check1_Click( )
    Check2.Enabled = Check1.Value
    Check3.Enabled = Check1.Value
    Check4.Enabled = Check1.Value
    Check1.Caption = IIf(Check4.Caption = "锁上", "打开", "锁上")
End Sub
```

复选框控件 Check2 的 Click 事件代码：

```
Private Sub Check2_Click( )
    Text1.FontBold = Check2.Value
End Sub
```

复选框控件 Check3 的 Click 事件代码：

```
Private Sub Check3_Click( )
    Text1.FontItalic = Check3.Value
End Sub
```

复选框控件 Check4 的 Click 事件代码：

```
Private Sub Check4_Click( )
    Text1.FontUnderline = Check4.Value
End Sub
```

【说明】DownPicture 属性指定复选框被按下时显示的图形。

第 6 章　循环结构程序设计

一、选择题

6.1　阅读下面的程序段：

```
For a=1 To 2
  For b=1 To a
    For c=b To 2
      i=i+1
    Next
  Next
Next
Print i
```

执行上面的三重循环后，i 的值为（　　）。

A）4　　B）5　　C）6　　D）9

【解答】正确答案为 B。

分析：本题是一个多重循环语句，是 For…Next 结构的三重嵌套，仔细分析一下就会看到，在结构中 i=i+1 共执行了 5 次，所以最后 i 的值为 5。

6.2　设有下面的循环：

```
i=1
Do
  i=i+3
  Print i
Loop Until  i> ____
```

程序运行后要执行 3 次循环体，则条件中 i 的最小值为（　　）。

A）6　　B）7　　C）8　　D）9

【解答】正确答案为 B。

分析：本题考查的是 DO…LoopUntil 结构的循环语句，不管满不满足条件，都要先执行一次。经分析当 i>7（或 8 或 9）都满足执行 3 次的条件，但题目要求最小的，所以应该是 7。

6.3　下面程序段的运行结果是（　　）。

```
a=1
b=1
Do
  a=a+1
```

```
    b=b+1
  Loop Until b>5
  Print"k=";a;Spc(4);"b=";b+a
```

A）k=7 b=14　　B）k=6 b=6　　C）k=4 b=8　　D）k=6 b=12

【解答】正确答案为 D。

分析：本题中 Do-Loop Until 循环为直到型循环结构，直到条件 b>5 为止，此时 a=6, b=6，所以最后输出的结果是 D。

6.4　下列程序段的执行结果为（　　）。

```
a=6
b=1
For i=1 To 3
  f=a+b
  a=b
  b=f
  Print f;
Next i
```

A）2　3　6　　B）2　3　5　　C）2　3　4　　D）2　2　8

【解答】正确答案为 B。

分析：本程序段的运行过程如下。

开始时 a=1, b=1。For 循环中步长默认值为 1，循环变量 I 的初值为 1，终值为 3，所以此循环结构可以循环 3 次。

第一次循环后，结果为：f=2, a=1, b=2；

第二次循环后，结果为：f=3, a=2, b=3；

第三次循环后，结果为：f=5, a=3, b=5。

每循环一次，输出 f 当前值，循环 3 次，即输出 3 个 f 值，分别为 2, 3, 5。

6.5　在窗体上画一个名称为 Command1 的命令按钮和一个名称为 Text1 的文本框，然后编写如下事件过程：

```
Private Sub Command1_Click()
  n=Val(Text1.Text)
  For i=2 To n
    For j=2 To Sqr(i)
        If i Mod j=0 Then Exit For
    Next j
    If j>Sqr(i)Then Print i
  Next i
End Sub
```

该事件过程的功能是（　　）。

A）输出 n 以内的奇数　　B）输出 n 以内的偶数

C）输出 n 以内的素数　　D）输出 n 以内能被 j 整除的数

【解答】正确答案为 C。

分析：要理解本事件过程的作用就要看最后打印的条件：j>Sqr(i)，而最后这个条件是否成立又取决于内循环，内循环的作用就是判断在 2 到某个数的平方根之间是否有它本身的因数，显然这是判断素数的条件。

6.6　下列程序段的执行结果为（　　）。

```
a=5
For k=1 To 0
   a=a + k
Next k
Print k;a
```

A）–1　6　　B）–1　16　　C）1　5　　D）11　21

【解答】正确答案为 C。

分析：For 循环语句的执行过程为：“循环变量”首先取得“初值”，检查是否超过“终值”，如果超过，就一次也不循环而跳出循环，属于“先检查后执行”的类型。在本程序段的 For k=1 To 0 语句中，初值为 1，终值为 0，显然当“循环变量”首先取得“初值”1，检查后超过“终值”0，所以一次也不执行，即最后执行 Print 时，k=1, a=5。

6.7　下列程序段的执行结果为（　　）。

```
a=3
b=1
For i=1 To 3
   f=a+b
   a=b
   b=f
   Print f ;
Next I
```

A）4 3 6　　B）4 5 9　　C）6 3 4　　D）7 2 8

【解答】正确答案为 B。

分析：本程序段的运行过程如下。

开始时 a=3, b=1。For 循环中步长默认值为 1，循环变量 i 的初值为 1，终值为 3，所以此循环结构可以循环 3 次。

第一次循环后，结果为：f=4, a=1, b=4；

第二次循环后，结果为：f=5, a=4, b=5；

第三次循环后，结果为：f=9, a=5, b=9。

所以每循环一次，便输出 f 当前值，循环 3 次，即输出 3 个 f 值，分别为 4, 5, 9。

6.8　下列程序段的执行结果为（　　）。

```
i=9
x=5
Do
```

```
    i=i+1
    x=x+2
  Loop Until i >=7
  Print "i="; i;
  Print "x="; x
```

A）i=4 x=5　　B）i=7 x=15　　C）i=6 x=8　　D）i=10 x=7

【解答】正确答案为 D。

分析：本题用的是 Do 循环结构，此循环由于“先执行后检查”，所以至少执行一次。本题中，程序运行到循环条件 i>=7 的值为 True 时，才停止。所以当程序结束运行后 i=10，x=7。

6.9　执行下列程序段后，输出的结果是（　　）。

```
For k1=0 To 4
  y=20
  For k2=0 To 3
    y=10
    For k3=0 To 2
      y=y + 10
    Next k3
  Next k2
Next k1
Print y
```

A）90　　B）60　　C）40　　D）10

【解答】正确答案为 C。

分析：这是一个三重循环嵌套的程序。多重 For 循环的执行过程是：外层循环的循环控制变量每取一个值，内循环的循环控制变量要取遍所有的值。这个程序最外两层循环的最后一次循环执行时 k1=4, k2=3，都要再执行一次 y=10，不管 y 以前为何值，在这里 y 都要重新取得初值 10，这时再执行一次内循环。内层循环体要执行 3 次，即加 3 次 10。y 最终的值是 40。

6.10　新建一个列表框，要实现对列表项可以复选，应设置的属性是（　　）。

A）ScrollBars　　B）MultiSelect　　C）DataField　　D）Stretch

【解答】正确答案为 B。

分析：在列表框的属性中，可以通过设置 MultiSelect 属性来实现是否可以进行列表项的多选，属性值为 0，不可以多选列表；属性为 1，不需要用 Ctrl 键或者 Shift 键就可进行多选；属性值为 2，要利用 Ctrl 键或者 Shift 键，所以选项 B 是正确的。而选项 A 是设置滚动条的，与此无关。选项 C 中 DataField 是用来设置数据的，也无关。选项 D 中 Stretch 不是列表框的属性，所以也不正确。

6.11　在窗体上画一个命令按钮，然后编写如下事件过程：

```
Private Sub Command1_Click()
  x=0
```

```
    Do Until x=–l
      a=InputBox("请输入 A 的值")
      a=Val(A)
      b=InputBox("请输入 B 的值")
      b=Val(B)
      x=InputBox("请输入 x 的值")
      x=Val(X)
      a=a+b+x
    Loop
    Print a
End Sub
```

程序运行后，单击命令按钮，依次在输入对话框中输入 5、4、3、2、l、–1，则输出结果为（　　）。

【解答】正确答案为 2。

分析：本题需要注意的是，每次循环开始时 3 个变量都被重新赋了一次值，所以最后的结果仅仅是 2+1+(–1)。所以最终的输出是 2。

6.12　在 Visual Basic 中，组合框是文本框和（　　）特性的组合。

A）复选框　　B）标签　　C）列表框　　D）目录列表框

【解答】正确答案为 C。

分析：组合框是一个独立的控件，它具有列表框和文本框的功能，它可以像列表框一样，让用户通过鼠标选择需要的项目，也可以像文本框一样，用输入的方式选择项目。

6.13　下列语句中，返回列表框 List1 中项目个数的语句是（　　）。

A）x=List1.ListCount　　B）x=ListCount

C）x=List1.ListIndex　　D）x=ListIndex

【解答】正确答案为 A。

分析：List 控件的 ListCount 属性返回列表部分项目的个数。

二、填空题

6.14　执行下面的程序段，x 的值为____。

```
Private Sub Command1_Click()
    For i=1 To 9
      a=a + i
    Next i
    x=Val(i)
    MsgBox x
End Sub
```

【解答】正确答案为 10。

分析：在程序中 For…Next 语句之后并没有对 i 的值改变，此时 i 为 10，故运行后结果为 10。

6.15　以下程序的功能是从键盘输入若干个学生的考试成绩，统计并输出最高分和最低分，当输入负数时结束输入，输出结果。请补全程序。

```
Dim x,amax,amin As Single
x=InputBox("Enter a score")
amax=x
amin=x
Do While  ____
  If x>amax Then
    amax=x
  End If
  If  _____  Then
    amin=x
  End If
  x＝InputBox("enter a score")
Loop
Print "max="; amax, "min="; amin
```

【解答】正确答案为：x>=0；x<amin。

分析：本程序先定义了 3 个变量 x、amin、amax，它们分别用来接收从键盘输入的数字、最小值和最大值，先用 InputBox 函数从键盘接收一个整数并赋给 x。然后将 x 的值赋给 amin 和 amax，作为它们的初值。因为当输入的值为负数时结束循环，所以 Do While 的控制语句就是判断 x 的值是否大于等于“0”。当 x≥0 时，执行块形式条件语句；当 x>amax 时，将 x 的值赋给 amax；显然，当 x<amin 时，就将 x 的值赋给 amin。每循环一次，就要给 x 赋值一次，然后接着循环直到 x 的值为负数时就跳出循环，输出 amax 和 amin 的值。

6.16　下面程序用来打印乘法“九九表”，请补全程序。

```
Dim i As Integer, j As Integer, Str1$
Str =""
For i=1 To 9
  For j=1 To 9
      If  ____  Then
        Str1=Str1+Str$（j）+"×"+Str$(i)+ "="+Str$(Val(i * j))
      Else
        Str1=Str1 & Chr(13)
          _____
      End If
  Next j
Next i
Print Str1
```

【解答】正确答案为 j < = i ；Exit FOR。

分析：Exit 语句总是出现在 If 语句或 Select Case 语句内部，而 If 语句或 Select Case 语

句在循环内嵌套，用 Exit 语句中断循环。函数首先定义了 Integer 型变量 i、j，并将空格赋给 Str1。第一个 For 循环的变量 i 从 1 到 9 步长为 1，第二个循环的变量 i 也是从 1 到 9 步长为 1。循环体为选择结构，它用来输出 i*j 的值，所以 j 的值应该小于此时 i 的值，那么 If 的判断语句为 j < = i。当满足条件时，执行 Then 后面的语句，即输出 i*j 的值；如果不满足，则执行 Else 后面的语句，并跳出内循环。

6.17　在窗体上画一个命令按钮，然后编写如下程序：

```
Function fun(By Val num As Long)As Long
  Dim k As Long
  k=l
  num=Abs(num)
  Do While num
    k=k*(num Mod 10)
    num=num \ 10
  Loop
  fun=k
End Function
Private Sub Command1_Click()
  Dim n As Long , r As Long
  n=InputBox("请输入一个数")
  n=CLng(n)  :  r=fun(n)
  Print r
End Sub
```

程序运行后，单击命令按钮，在输入对话框中输入"345"，输出结果为____。

【解答】正确答案为 60。

分析：程序首先要求用户输入一个数，然后将这个数转化为长整型数传递给函数 fun。本题中传递的数字是 345。函数 fun 首先对传来的参数取绝对值，然后进入循环。

第一次循环结束的时候：k=5，num=34；

第二次循环结束的时候：k=20，hum=3；

第三次循环结束的时候：k=60，num=0。

6.18　在窗体上画一个命令按钮，然后编写如下事件过程：

```
Private Sub Command1_Click()
  For i=l To 4
    x=4
    For j=1 To 3
      x=3
      For k=1 To 2
        x=x+6
      Next k
```

```
    Next j
  Next i
  Prim x
End Sub
```

程序运行后，单击命令按钮，输出结果是_____。

【解答】正确答案为 15。

分析：观察程序，由于每次执行第二层循环时 x 都被重新赋值，因此只需分析 x=3 时，执行第三层循环所得到的结果就可以了。

k=l，x=x+6=3+6=9

k=2，x=x+6=9+6=15

最终的输出结果是：15。

6.19　以下程序的功能是：从键盘上输入若干个数字，当输入负数时结束输入，统计出全部数字的平均值，输出结果。请补全程序。

```
Private Sub Form_Click()
  Dim x , y As Single , z As Integer
  x=InputBox("Enter a score")
  Do while   ____
    y=y+x
    z=z+1
    x=InputBox("Enter a score")
  Loop
  If z=0 Then
    z=l
  End If
  y=   ____
  Print y
End Sub
```

【解答】正确答案为：x>=0；y/z。

解析：本题的循环首先判断输入是否大于 0，如果是，则处理这个输入，否则结束循环。循环体内的第二个判断语句判断当前输入是否比当前的最小值小，如果是，则改写当前最小值为输入值，否则结束判断语句。

三、编程题

6.20　输入初始值，输出 100 个不能被 3 整除的数。

【解答】

（1）设计程序界面和设置对象属性，如图 6-1 所示。

（2）编写代码。

增加窗体的 Activate 事件代码：

图 6-1　输出 100 个不能被 3 整除的数

```
Private Sub Form_Activate( )
    Text2.SetFocus
End Sub
```

改写命令按钮 Command1 的 Click 事件代码：

```
Private Sub Command1_Click( )
    Dim x As Integer, n As Integer
    x = Val(Text2.Text): n = 1
    Do Until n > 100
        If x Mod 3 < > 0 Then
            Text1.Text = Text1.Text & Str(x) & Chr(13) & Chr(10)
            n = n + 1
        End If
        x = x + 1
    Loop
End Sub
```

6.21　设计程序，求 $s = 1 + (1 + 2) + (1 + 2 + 3) + \cdots + (1 + 2 + 3 + \cdots + n)$的值。

【解答】本题先进行内循环累加，再进行外循环累加，因此，需要两个累加器。但是，由于本题较特殊，也可以用一个循环来实现。括号内每累加一个数，就往 s 中累加一次，其结果一样。

（1）设计程序界面并设置对象属性，如图 6-2（a）所示。其中 Command1 的 Default 属性为 True（窗体的默认命令按钮，使得按 Enter 键等于单击命令按钮）。

（2）编写代码。

首先编写窗体的 Activate 事件代码：

```
Private Sub Form_Activate( )
    Text1.SetFocus
End Sub
```

编写命令按钮 Command1 的 Click 事件代码：

```
Private Sub Command1_Click( )
    Dim n As Integer, i As Integer, j As Long, s As Long
    n = Val(Text1.Text)
    j = 0:    s = 0
    For i = 1 To n
        j = j + i
        s = s + j
    Next i
    Label2.Caption = "和数  s = " & Str(s)
    Text1.SelStart = 0
    Text1.SelLength = Len(Text1.Text)
End Sub
```

运行程序，输入一个整数后按 Enter 键，显示结果如图 6-2（b）所示。



（a）　　（b）

图 6-2　程序界面与程序运行结果

6.22　设 $s = 1^1 \times 2^2 \times 3^3 \times \cdots \times n^n$，求 s 不大于 400 000 时最大的 n 值。

【解答】本题先进行内循环累乘（求幂），再进行外循环累乘，因此需要两个累乘器。计数器为 n，外累乘器 s = s * t，内累乘器 t = t * n。其循环条件是 s <= 400000，又由于求的是最大的 n 值，因此输出语句应在外循环体外。

窗体界面的设计与程序运行的结果，如图 6-3 所示。



（a）　　（b）

图 6-3　程序界面与程序运行结果

编写命令按钮的 Click 事件代码：

```
Private Sub Command1_Click( )
  Dim n As Integer, s As Long
  CurrentY = Label1.Height + 400
  Print " n", " s"
  Print
  n = 1
  s = 1
  Do While s <= 400000
    n = n + 1
    t = 1
    For i = 1 To n
      t = t * n
    Next
    s = s * t
    Print n, s                          ' 通过本行可以看到循环过程
```

```
    Loop
    Label2.Caption = "n = " & Str(n – 1)
End Sub
```

6.23　设有一张厚为 *x* 毫米、面积足够大的纸，将它不断地对折。试问对折多少次后，其厚度可达珠穆朗玛峰的高度（8848 米）。

【解答】设对折后纸的厚度为 *h* 毫米，计数器为 *n*。在没有对折时，纸厚为 *x* 毫米，每对折一次，其厚度是上一次的 2 倍，在未到达 8848 米时，重复进行对折。还要注意，在解题时要把单位换算一致。

（1）建立应用程序用户界面并设置对象属性，如图 6-4（a）所示。



（a）　　（b）

图 6-4　用户界面与程序运行结果

（2）编写程序代码。根据流程图，写出命令按钮 Command1 的 Click 事件代码为：

```
Private Sub Command1_Click( )
    n = 0                                  ' 对折次数
    h = Text1.Text                         ' 纸的厚度
    Do While h < 8848000
        n = n + 1                          ' 对折次数
        h = 2 * h                          ' 对折
    Loop
    Text2.Text = n
End Sub
```

如果改为直到型循环结构，程序代码为：

```
Private Sub Command1_Click( )
    n = 0                                  ' 对折次数
    h = Text1.Text                         ' 纸的厚度
    Do
        n = n + 1                          ' 对折次数
        h = 2 * h                          ' 对折
    Loop Until h >= 8848000
    Text2.Text = n
End Sub
```

运行结果如图 6-4（b）所示。

6.24　我国古代数学家张丘建在“算经”里曾提出一个世界数学史上有名的百鸡问题：“鸡翁一，值钱五；鸡母一，值钱三；鸡雏三，值钱一；百钱买百鸡，问鸡翁、母、雏各几何？”请编写程序，求出结果。

【解答】设公鸡 x 只，母鸡 y 只，小鸡 z 只，依题意可以列出以下方程组：

$$\begin{cases} x+y+z=100 \\ 5x+3y+\dfrac{z}{3}=100 \end{cases}$$

在这两个方程中，由于有 3 个未知数，属于不定方程，无法直接求解。下面用“穷举法”，将各种可能的组合全部一一测试，将符合条件的组合输出。

先设 $x=1$，$y=1$，则 $z=100-x-y=98$，检查这一组的价钱加起来是否为 100 元。经验算，不等于 100 元，所以这一组不合要求。再看下一组，仍保持 $x=1$，而 $y=2$，则 $z=100-1-2=97$，价钱为 $5\times1+3\times2+97/3\approx43$，也不符合要求。保持 $x=1$，使 y 由 3 变到 100，依次测试各组合。

然后设 $x=2$，y 再由 1 变到 100 …… 直到 $x=100$，y 再由 1 变到 100。这样就把全部可能的组合一一测试过了。



图 6-5　百钱买百鸡

（1）建立应用程序用户界面并设置对象属性，如图 6-5 所示。

（2）编写程序代码。命令按钮 Command1 的 Click 事件代码为：

```
Private Sub Command1_Click( )
  List1.Clear
  For x = 1 To 100
    For y = 1 To 100
      z = 100 – x – y
      If 5 * x + 3 * y + z / 3 = 100 Then
        p = Format(x, "@@@@") & Format(y, "@@@@@") & Format(z, "@@@@@")
        List1.AddItem p
      End If
    Next
  Next
End Sub
```

经过对上述代码的分析可知，实际上并不需要使 x 从 1 变到 100，y 从 1 变到 100。因为公鸡每只 5 元，100 元最多买 20 只公鸡，而如果 100 元全买了 20 只公鸡的话，就买不了母鸡和小鸡了，不符合“百钱买百鸡”的要求。所以公鸡不可能是 20 只，最多只能买 19 只。同理，母鸡一只 3 元，100 元最多买 33 只。因此，可将代码修改为：

```
Private Sub Command1_Click( )
  List1.Clear
  For x = 1 To 19
    For y = 1 To 33
      z = 100 – x – y
      If 5 * x + 3 * y + z / 3 = 100 Then
        p = Format(x, "@@@@") & Format(y, "@@@@@") & Format(z, "@@@@@")
```

```
            List1.AddItem p
        End If
    Next
Next
End Sub
```

6.25　打印乘法“九九表”，输出结果分别如图 6-6（a）、（b）所示。



（a）



（b）

图 6-6　“九九表”

【解答】若要输出图 6-6（a）所示的“九九表”，其窗体的 Load 事件代码为：

```
Private Sub Form_Load( )
    Show
    FontSize = 12
    Print Tab(25); "九九表"
    Print
    Print " * ";
    For i = 1 To 9                          ' 输出第一行数字（1～9）
        Print Tab(i * 6); i;
    Next i
    Print
    For j = 1 To 9                          ' 外层循环
        Print j; " ";
        For k = 1 To j                      ' 注意，内层循环的初值为 1，终值为 j
            m = j * k
            Print Tab(k * 6); m; " ";
        Next k
        Print
    Next j
End Sub
```

若要输出图 6-6（b）所示的“九九表”，其窗体的 Load 事件代码为：

```
Private Sub Form_Load( )
    Show
    FontSize = 12
```

```
Print Tab(25); "九九表"
Print
Print " * ";
For i = 1 To 9                          ' 输出第一行数字（1～9）
   Print Tab(i * 6); i;
Next i
Print
For j = 1 To 9                          ' 外层循环
   Print j; " ";
   For k = j To 9                       ' 注意，内层循环的初值为 j，终值为 9
      m = j * k
      Print Tab(k * 6); m; " ";
   Next k
   Print
Next j
End Sub
```

6.26　所谓“水仙花数”，是指一个 3 位数，其各位数的立方和等于该数，如 $153 = 1^3 + 5^3 + 3^3$，编写程序输出所有的“水仙花数”。

【分析】此题的关键是把任意 3 位数的每一位数分离出来。设 a, b, c 分别是 3 位整数 n 的百位数、十位数、个位数，则

```
a = Int(n / 100)
b = Int((n – a * 100) / 10)
c = n – a * 100 – b * 10
```

【解答】窗体界面的设计如图 6-7 所示。

命令按钮的 Click 事件代码为：

```
Private Sub Command1_Click( )
   Dim p As Integer
   List1.Clear
   For n = 100 To 999
      a = Int(n / 100)
      b = Int((n – a * 100) / 10)
      c = n – (a * 100 + b * 10)
      p = a ^ 3 + b ^ 3 + c ^ 3
      If p = n Then List1.AddItem p
   Next
End Sub
```

6.27　求 1000～1100 之间的所有素数。

【分析】对 1000～1100 之间的各整数依次测试其是否为素数，需使用双重循环。

（1）建立用户界面并设置对象属性，如图 6-8 所示。



图 6-7　水仙花数



图 6-8　求素数

（2）编写事件代码。

编写“开始”命令按钮 Command1 的 Click 事件代码：

```
Private Sub Command1_Click( )
  For n = 1001 To 1100 Step 2
    s = 0
    For i = 2 To Int(Sqr(n))
      If n Mod i = 0 Then
        s = 1
        Exit For
      End If
    Next
    If s = 0 Then List1.AddItem n
  Next
End Sub
```

6.28　利用下述公式计算圆周率π的近似值：

$$\frac{\pi}{4}=1-\frac{1}{3}+\frac{1}{5}-\frac{1}{7}+\frac{1}{9}-\cdots$$

当最后一项的绝对值小于 0.000 001 时停止计算。

【解答】在立即窗口输出计算结果的事件代码如下：

```
Private Sub Command1_Click( )
  Dim m!, s#, t#, k!
  k = 1: s = 1: t = 1: m = 1
  While Abs(t) > 0.0000001
    k = k + 2
    m = -m
    t = m / k
    s = s + t
  Wend
  Debug.Print Format(4 * s, "0.0000000")
End Sub
```

6.29　在窗体上输出如图 6-9 所示的图形。

【解答】

（1）建立用户界面并设置对象属性，如图 6-9 所示。

（2）编写命令按钮的 Click 事件代码：

```
Private Sub Command1_Click( )
  Cls
  Print
  For n = 1 To 5
    Print Tab(2 * n + 2);
    For m = n To 10 – n
      Print Spc(1); "*";
    Next
    Print Spc(4);
    For m = 1 To 2 * n – 1
      Print Spc(1); "*";
    Next
    Print
  Next
End Sub
```

6.30　马克思曾经做过这样一道趣味数学题：有 30 个人在一家小饭馆里用餐，其中有男人、女人和小孩。每个男人花了 3 先令，每个女人花了 2 先令，每个小孩花了 1 先令，一共花去 50 先令。问男人、女人和小孩各有几人？

【解答】设有 x 个男人，y 个女人，z 个小孩。依题意，列出以下方程组：

$$\begin{cases} x+y+z=30 \\ 3x+2y+z=50 \end{cases}$$

由于两个方程式中有 3 个未知数，属于不定方程，无法直接求解。可以用“穷举法”来进行“试根”，即将各种可能的 x, y, z 组合一一进行测试，将符合条件者输出即可。

注意，最多只能有 16 个男人，最多只能有 24 个女人。

窗体界面的设计如图 6-10 所示。这里给出命令按钮的 Click 事件代码：



图 6-9　输出图形



图 6-10　趣味数学题

```
Private Sub Command1_Click( )
  List1.Clear
  For x = 1 To 16
    For y = 1 To 24
      z = 30 – x – y
      If 3 * x + 2 * y + z = 50 Then
```

```
            p = Format(x, "@@@") & Format(y, "@@@@@") & Format(z, "@@@@@")
            List1.AddItem p
          End If
        Next
      Next
    End Sub
```

6.31　利用循环语句在窗体中显示不同字号大小。

【解答】窗体界面的设计如图 6-11 所示。Command1 的 Click 事件代码为：

图 6-11　显示不同字号大小

```
Private Sub Command1_Click( )
  Cls
  Form1.FontName = "隶书"
  For i = 10 To 15
    Form1.FontSize = i
    Print "字号大小："; i
  Next i
End Sub
```

6.32　用 1, 2, 3, 4 这 4 个数字组成 4 位数。编写程序，打印出所有可能的 4 位数（4 个数字可以相同），并统计出所组成的 4 位数的个数。

【解答】这个问题可以用四重循环来解决，每层循环的初值均为 1，终值为 4，步长为 1。总的循环次数就是所有 4 位数的个数。程序代码如下：

```
Private Sub Form_Load( )
  Show
  Sum = 0
  For a = 1 To 4
    For b = 1 To 4
      For c = 1 To 4
        For d = 1 To 4
          x = a * 1000 + b * 100 + c * 10 + d
          Print x;
          Sum = Sum + 1
          If Sum Mod 15 = 0 Then Print
        Next d
      Next c
    Next b
  Next a
  Print
  Print "和为："; Sum
End Sub
```

运行结果如图 6-12 所示。



图 6-12　用 1, 2, 3, 4 组成 4 位数

6.33　用“筛法”找出 1～100 之间的全部素数。

【分析】“筛法”求素数表是由希腊著名数学家 Eratost henes 提出来的，其方法是：在纸上写出 1～n 的全部整数，如图 6-13 所示（n=100）。



图 6-13　用户界面和程序运行结果

然后逐一判断它们是否素数，找出一个非素数就把它挖掉（筛掉），最后剩下的就是素数。具体做法如下。

① 先将 1 挖掉。

② 用 2 为除数去除它后面的每个数，即把能被 2 整除的数挖掉，即把 2 的倍数挖掉。

③ 用 3 为除数去除它后面的每个数，即把 3 的倍数挖掉。

④ 分别用 4, 5…各数作为除数去除这些数后面的各数（4 已被挖掉，不必再用 4 当除数，需用未被挖掉的数当除数）。这个过程一直进行到除数为 $\sqrt{n}$ 为止（如果 $\sqrt{n}$ 不是整数就取其整数部分），如图 6-14 所示。



图 6-14　用“筛法”找素数

⑤ 经过如此“筛法”，剩下的部分全部是素数。

【解答】设计步骤如下。

（1）建立应用程序用户界面。新建一个工程，进入窗体设计器，首先增加两个命令按钮 Command1,Command2，一个图片控件 Picture1 作为容器，然后选中 Picture1，在其中增加一个标签控件 Label1，如图 6-15 所示。



（a）　　（b）

图 6-15　建立用户界面

（2）设置对象属性。

修改窗体的 Caption 属性为：用“筛法”找素数。

修改 Command1 的 Caption 属性为：开始。

修改 Command2 的 Caption 属性为：重置。

修改 Picture1 的 BackColor 属性为：（白色）。

修改 Label1 的 BackColor 属性为：（白色），Height 属性为：260，Width 属性为：280。

（3）编写程序代码。筛选的过程主要由“开始”命令按钮的 Click 事件激发。

窗体 Form1 的 Load 事件代码：

```
Private Sub Form_Load( )
   Picture1.Move 0, 0
   Picture1.Height = (Label1(0).Height + 5) * 10 + 3
   Picture1.Width = (Label1(0).Width + 5) * 10 + 3
   Me.Height = Picture1.Height + 900
   Me.Width = Picture1.Width + 80
   Label1(0).Visible = False
   For n = 1 To 100
      i = (n - 1) \ 10: j = (n - 1) Mod 10 : Load Label1(n)
      With Label1(n)
         .Left = 5 + j * Label1(0).Width
         .Top = 5 + i * Label1(0).Height
         .Visible = True
         .Caption = n
      End With
   Next
End Sub
```

“开始”命令按钮 Command1 的 Click 事件代码：

```
Private Sub Command1_Click( )
   Label1(1).Visible = False
   n = 100
   For i = 2 To Sqr(n)
      If Label1(i).Visible = True Then
         MsgBox "现在开始删去" & Str(i) & "的倍数", vbInformation, "用“筛法”找素数"
         For j = i + 1 To n
            If Label1(j).Visible = True And j Mod i = 0 Then
               Label1(j).Visible = False
            End If
         Next
      End If
   Next
   MsgBox "剩下来的整数已全都是素数!", vbInformation, "用“筛法”找素数"
End Sub
```

“重置”命令按钮 Command2 的 Click 事件代码：

```
Private Sub Command2_Click( )
   For n = 1 To 100
      Label1(n).Visible = True
   Next
End Sub
```

6.34　利用列表框，编写能对本学期选修课程进行课程添加、修改和删除的应用程序。

【分析】由于不能直接对列表框中的项目进行添加、修改和删除操作，所以需要利用文本框。列表框中的项目在 Form1_Load 中用 AddItem 方法添加。利用“添加课程”按钮将文本框中的内容添加到列表框中，利用“删除课程”按钮删除列表框中选定的项目。如果要修改列表框，则首先选定项目，然后按“修改课程”按钮，所选的项目将显示在文本框中。当在文本框中修改完之后，再按“修改确定”按钮更新列表框。

【解答】设计步骤如下。

（1）建立用户界面并设置对象属性，如图 6-16（a）所示。



（a）　　（b）

图 6-16　选修课程表

（2）编写事件代码。

窗体 Form 的 Load 事件代码：

```
Private Sub Form_Load()
    List1.AddItem "政治"                    ' 在列表框中加入项目"政治"
    List1.AddItem "体育"                    ' 在列表框中加入项目"体育"
    List1.AddItem "英语"                    ' 在列表框中加入项目"英语"
    List1.AddItem "语文"                    ' 在列表框中加入项目"语文"
    List1.AddItem "数学"                    ' 在列表框中加入项目"数学"
End Sub
```

“添加课程”按钮 Command1 的 Click 事件代码：

```
Private Sub Command1_Click( )
    List1.AddItem Text1.Text                ' 将文本框中的内容添加到列表框中
    Text1.Text = ""                         ' 清空文本框
End Sub
```

“删除课程”按钮 Command2 的 Click 事件代码：

```
Private Sub Command2_Click( )
    List1.RemoveItem List1.ListIndex        ' 除去列表框中选定的项目
End Sub
```

“修改课程”按钮 Command3 的 Click 事件代码：

```
Private Sub Command3_Click( )
    Text1.Text = List1.Text                 ' 将列表框中当前选中的项目内容赋给文本框
    Text1.SetFocus                          ' 设置焦点
    Command1.Enabled = False
    Command2.Enabled = False
    Command3.Enabled = False
    Command4.Enabled = True
End Sub
```

“修改确定”按钮 Command4 的 Click 事件代码：

```
Private Sub Command4_Click( )
    List1.List(List1.ListIndex) = Text1.Text    ' 将文本框中的内容返还给列表框
    Command4.Enabled = False                    ' 使“修改确定”按钮不可用
    Command1.Enabled = True
    Command2.Enabled = True
    Command3.Enabled = True
    Text1.Text = ""
End Sub
```

运行程序，结果如图 6-16（b）所示。

6.35　编写小学加减法算术练习程序。计算机随机给出两位数的加减法算术题，要求学生回答，答对的打“√”，答错的打“×”。将做过的题目存放在列表框中备查，并随时给出答题的正确率。

【分析】随机函数 Rnd 返回一个（0,1）之间的随机小数，为了生成某个范围内的随机整

数，可以使用公式：Int((最大值–最小值+1) * Rnd + 最小值)。其中，最大值和最小值为指定范围中的最大数和最小数。

【解答】设计步骤如下。

（1）建立应用程序用户界面。新建一个工程，进入窗体设计器，增加一个标签 Label1（显示题目），一个文本框 Text1（输入答案），一个列表框 List1（保存做过的题目），一个命令按钮 Command1，一个图像 Image1 和一个框架 Frame1。激活 Frame1 后，在其中增加两个标签，如图 6-17（a）所示。



（a）　　（b）

图 6-17　算术练习

（2）设置对象属性，见表 6-1。其他属性的设置如图 6-17（a）所示。

表 6-1　属性设置

对　象	属　性	属 性 值	说　明
Form1	Tag	0.0	存放题目总数
List1	Tag	0.0	存放答对的题数

（3）编写代码。

出题部分由窗体的 Activate（激活）事件代码完成：

```
Private Sub Form_Activate( )
   Randomize (Time)
   a = Int(10 + 90 * Rnd)
   b = Int(10 + 90 * Rnd)
   p = Int(2 * Rnd)
   Select Case p
      Case 0
         Label1.Caption = a & " + " & b & " = "
         Text1.Tag = a + b                          ' 将本题答案放入 Text1.Tag 中
      Case 1
         If a < b Then t = a: a = b: b = t
         Label1.Caption = a & " – " & b & " = "
         Text1.Tag = a – b                          ' 将本题答案放入 Text1.Tag 中
   End Select
   Form1.Tag = Form1.Tag + 1
   Text1.SelStart = 0
   Text1.Text = ""
End Sub
```

答题部分由文本框的 KeyPress 事件代码完成：

```
Private Sub Text1_KeyPress(KeyAscii As Integer)
  If KeyAscii = 13 Then
    fm = "!@@@@@@@@@@@@@@"
    If Val(Text1.Text) = Text1.Tag Then
      Item = Format(Label1.Caption & Text1.Text, fm) & "  √"
      List1.Tag = List1.Tag + 1
    Else
      Item = Format(Label1.Caption & Text1.Text, fm) & "  ×"
    End If
    List1.AddItem Item, 0                          ' 将题目和回答插入到列表框中的第一项
    Label2.Caption = "共" & Form1.Tag & "题，" & Chr(13) & "正确率为:"
    Label3.Caption = Format(List1.Tag / Form1.Tag, "#0.0#%")
    Form_Activate                                  ' 调用出题部分代码
  End If
End Sub
```

“关闭”按钮的 Click 事件代码：

```
Private Sub Command1_Click( )
  Unload Me
End Sub
```

运行程序，结果如图 6-17（b）所示。

6.36　将习题 6.35 中的列表框改为组合框（下拉列表框）。

【解答】设计步骤如下。

（1）建立应用程序用户界面并设置对象属性。将习题 6.35 中列表框改为组合框，如图 6-18 所示，并修改其属性见表 6-2。另外，再增加一个命令按钮（重置）Command2。其他对象的修改与属性设置如图 6-18 所示。



（a）　　（b）

图 6-18　使用下拉列表框

表 6-2　属性设置

对　象	属　性	属性值	说　明
Combo1	Style	2- Dropdown List	下拉列表框
	Tag	0.0	存放答对的题数

（2）编写程序代码。

修改文本框 Text1 的 KeyPress 事件代码：

```
Private Sub Text1_KeyPress(KeyAscii As Integer)
  If KeyAscii = 13 Then
    Fm = "!@@@@@@@@@@@@@@"
    If Text1.Text = Text1.Tag Then
      Item = Format(Label1.Caption & Text1.Text, Fm) & "  √"
      Combo1.Tag = Combo1.Tag + 1
    Else
      Item = Format(Label1.Caption & Text1.Text, Fm) & "  ×"
    End If
    Combo1.AddItem Item, 0
    Combo1.ListIndex = 0
    Label3.Caption = "共" & Form1.Tag & "题，正确率为:"
    Label2.Caption = Format(Combo1.Tag / Form1.Tag, "#0.0#%")
    Form_Activate
  End If
End Sub
```

编写命令按钮 Command2 的 Click 事件代码：

```
Private Sub Command2_Click( )
  Form1.Tag = 0#
  Combo1.Tag = 0#
  Combo1.Clear
  Label3.Caption = "欢迎重新开始!"
  Label2.Caption = ""
  Form_Activate
  Text1.SetFocus
End Sub
```

其他代码同习题 6.35。

6.37　利用循环结构和列表框控件，设计“选项移动”窗体。

【解答】所谓“选项移动”窗体，是指由两个列表框和 4 个命令按钮所构成的界面，在 Windows 程序中常可见到此类窗口（如图 6-19 所示）。

（a）　　（b）

图 6-19　“选项移动”表单

设计步骤如下。

（1）建立应用程序用户界面并设置对象属性。界面的设计与属性的设置如图 6-19 所示。其中，列表框的属性设置见表 6-3。

表 6-3　属性设置

对　象	属　性	属 性 值	说　明
List1	Multiselect	0-None	多项选择
	Style	1-Checkbox	风格
List2	Multiselect	2-Extended	多项选择

（2）编写事件代码。

编写窗体的 Load 事件代码：

```
Private Sub Form_Load( )
   List1.AddItem "电冰箱"
   List1.AddItem "洗衣机"
   List1.AddItem "彩色电视机"
   List1.AddItem "组合音响"
   List1.AddItem "影碟机"
   List1.AddItem "电水壶"
   List1.AddItem "饮水机"
   List1.AddItem "微波炉"
   List1.AddItem "照相机"
End Sub
```

编写命令按钮 Command1 的 Click 事件代码：

```
Private Sub Command1_Click( )
   i = 0
   Do While i < List1.ListCount
      If List1.Selected(i) = True Then
         List2.AddItem List1.List(i)
         List1.RemoveItem i
      Else
         i = i + 1
      End If
   Loop
End Sub
```

编写命令按钮 Command2 的 Click 事件代码：

```
Private Sub Command2_Click( )
   For i = 0 To List1.ListCount - 1
      List2.AddItem List1.List(i)
   Next
```

```
    List1.Clear
End Sub
```

编写命令按钮 Command3 的 Click 事件代码：

```
Private Sub Command3_Click( )
    i = 0
    Do While i < List2.ListCount
        If List2.Selected(i) = True Then
            List1.AddItem List2.List(i)
            List2.RemoveItem i
        Else
            i = i + 1
        End If
    Loop
End Sub
```

编写命令按钮 Command4 的 Click 事件代码：

```
Private Sub Command4_Click( )
    For i = 0 To List2.ListCount - 1
        List1.AddItem List2.List(i)
    Next
    List2.Clear
End Sub
```

【说明】VB 的列表框支持简单或扩展的多重选择。简单的多重选择可以逐条选择多个选项（按 Shift 键或按 Ctrl 键，同时单击鼠标左键），扩展的多重选择允许用户一次可以选择相邻的多个选项（按 Shift 键，同时单击鼠标左键）或逐条选择不相邻的多项选项（按 Ctrl 键，同时单击鼠标左键）。

列表框的 Multiselect 属性决定了用户是否能够选择多项数据，其中：

- Multiselect=0 —— 不能；
- Multiselect=1 —— 简单多重；
- Multiselect=2 —— 扩展多重。

若 List1.Selected(i) = True，表示列表框 List1 中的第 i+1 项选项被选中。

List1.List(i)表示列表框 List1 中第 i+1 项的值。

第7章　数　　组

一、选择题

7.1　设有声明语句如下，则数组 b 中全部元素的个数为（　　）。

```
Dim b(-1 To 10,2 To9,20 ) As Integer
```

A）2310　　B）2016　　C）1500　　D）1658

【解答】正确答案为 B。

分析：在本题中，Dim b(-1 To 10,2 To 9,20) As Integer 定义了一个三维数组，如果没有指定下界，默认值是 0，所以计算数组中的个数为：(10–(–1)+1)×(9–2+1)×(20–0+1)=12×8×21=2016，所以选项 B 是正确答案。

7.2　设有数组定义语句 Dim a(5) As Integer，下列给数组元素赋值的语句错误的是（　　）。

A）a(3) =3　　B）a(3) =InputBox("input data")

C）a(3) =List1.ListIndex　　D）a =Array(1,2,3,4,5,6)

【解答】正确答案为 D。

分析：Array 函数只能给变体型数组初始化，a 为整型数组，所以选项 D 错误。

7.3　在窗体上画一个名称为 Label1 的标签，然后编写如下事件过程：

```
Private Sub Form_Click()
    Dim arr(10,10) As Integer
    Dim i As Integer , j As Integer
    For i=2 To 4
        For j=2 To 4
            arr(i,j)=i*j
        Next j
    Next i
    Label1.Caption=Str(arr(2,2)+arr(3,3))
End Sub
```

程序运行后，单击窗体，在标签中显示的内容是（　　）。

A）12　　B）13　　C）14　　D）15

【解答】正确答案为 B。

分析：本题具有一定的迷惑性，其实分析一下就可以知道 arr(i,j)的值即为 i*j，所以 arr(2,2)+arr(3,3)就是求 2*2+3*3=13。

7.4　阅读程序：

```
Option Base 1
```

```
Dim arr() As Integer
Private Sub Form_Click()
   Dim i As Integer , j As Integer
   ReDim arr(3,2)
   For i=1 To 3
      For j=1 To 2
         arr(i,j)=i*2+j
      Next j
   Next i
   ReDim Preserve arr(3,4)
   For j=3 To 4
      arr(3,j)=j+9
   Next j
   Print arr(3,2)+arr(3,4)
End Sub
```

程序运行后，单击窗体，输出结果为（　　）。

A）21　　B）13　　C）8　　D）25

【解答】正确答案为 B。

分析：每次使用 ReDim 语句都会使原来数组中的值丢失，也可以在 ReDim 后加 Preserve 参数来保留数组中的数据，但使用 Preserve 只能改变最后一维的大小，前面几维大小不能改变，而在本题中 arr(3,2)始终都未赋值，arr(3,4)=4+9=13，故最后为 13。

7.5　在窗体上画一个名称为 Command1 的命令按钮，然后编写如下程序：

```
Option Base 1
Private Sub Command1_Click()
    Dim c As Integer , d As Integer
    d=0
    c=6
    x=Array(2,4,6,8,10,12)
    For i=1 To 6
        If x(i)>c Then
            d=d+x(i)
            c=x(i)
        Else
            d=d-c
        End If
    Next i
    Print d
End Sub
```

程序运行后，如果单击命令按钮，则在窗体上输出的内容为（　　）。

A）10　　B）16　　C）12　　D）20

【解答】正确答案为 C。

分析：本题考查的是循环和条件判断语句，Option Base 1 的作用是强制使数组下标从 1 开始。

7.6　以下程序段的输出结果为（　　）。

```
Dim I , a(10) , p(3)
k=5
For i=0 To 10
    a(i) =i
Next i
For i =0 To 2
    p(i)=a(i*(i+1))
Next i
For i=0 To 2
    k=k + p(i) * 2
Next i
Print k
```

A）20　　B）21　　C）56　　D）32

【解答】正确答案为 B。

分析：第一个循环对数组 a()进行赋值，第二个循环对数组 p()进行赋值，第三个循环对 k 进行累加。k 的初值为 5，第一个循环 k=k+p(0)*2=5，……，第三个循环 k=k+p(2) *2=21 并输出，所以选项 B 正确。

7.7　假定建立了一个名为 Command1 的命令按钮数组，则以下说法错误的是（　　）。

A）数组中每个命令按钮的名称（Name 属性）均为 Command1

B）数组中每个命令按钮的标题（Caption 属性）都一样

C）数组中所有命令按钮可以使用同一个事件过程

D）用名称 Command1（下标）可以访问数组中的每个命令按钮

【解答】正确答案为 B。

分析：控件数组是由一组相同类型的控件组成的，它们公用一个控件名，具有相同的属性，但它们的按钮标题可以不一样。当建立控件数组时，系统给每个元素赋予一个唯一的索引号（Index），通过该索引号可以访问控件数组中的每个命令按钮。

7.8　下列叙述中，正确的是（　　）。

A）控件数组的每一个成员的 Caption 属性值都必须相同

B）控件数组的每一个成员的 Index 属性值都必须不相同

C）控件数组的每一个成员都执行不同的事件过程

D）对已经建立的多个类型相同的控件，这些控件不能组成控件数组

【解答】正确答案为 B。

分析：控件的 Name 属性是用来标识控件的，几个相同类型的控件，若 Name 属性相同，说明它们是同一个控件，但实际上是几个控件，由此引出控件数组，所以控件数组的每一个成员的 Name 属性值必须相同，而 Caption 属性值可以不同；Index（索引）属性值是用来区别控件数组的，当然取值必须不同；Visual Basic 中控件数组都执行同样的事件过程；对已经建立的多个类型相同的控件，可以通过修改控件的 Name 属性，使其都具有相同的 Name 属性值，这些控件就成为控件数组了。

7.9　在窗体上画一个名称为 Text1 的文本框和一个名称为 Command1 的命令按钮，然后编写如下事件过程：

```
Private Sub Command1Click()
    Dim array1(10,10)As Integer
    Dim i As Integer,j As Integer
    For i=1 To 3
      For j=2 To 4
        Array1(i,j)=i+j
      Next j
    Next i
    Text1.Text=array1(2,3)+array1(3,4)
End Sub
```

程序运行后，单击命令按钮，在文本框中显示的值是（　　）。

A）15　　B）14　　C）13　　D）12

【解答】正确答案为 D。

分析：本题没有太多的技巧，关键就是要搞清楚两层循环给数组的元素所赋的值，不难得出结果。

二、填空题

7.10　用 Dim(1, 3 to 7,10)声明的是一个____维数组。

【解答】正确答案为三。

分析：根据定义数组的语法格式可知，题目中所给数组为三维的。

7.11　在窗体上画一个命令按钮，然后编写如下事件过程：

```
Option Base 1
Private Sub Command1_Click()
  Dim a
  A=Array(1 , 2 , 3 , 4)
  j=1
  For i=4 To l Step–1
    s=s+a(i)*j
    j=j*10
  Next i
```

```
    Print s
End Sub
```

运行程序，单击命令按钮，其输出结果是____。

【解答】正确答案为 1234。

分析：跟踪程序，由 a=Array(1,2,3,4)可知：a(1)=1，a(2)=2，a(3)=3，a(4)=4。

执行循环体：

s=s+a(i)*j=a(4)*1=4，j=j*10=10

s=s+a(i)*j=4+a(3)*10=34，J=j*10=100

s=s+a(i)*j=34+a(2)*100=234，J=j*10=1000

s=s+a(i)*j=234+a(1)*1000=1234，J=j*10=10000

最终的输出结果是：1234。

三、编程题

7.12　输入一串字符，统计各字母出现的次数，不区分大小写。

【分析】统计 26 个英文字母出现的次数，必须声明一个具有 26 个元素的数组，每个元素的下标表示对应字母出现的次数。从输入的字符串中逐一取出字符，并将其转换成大写字母（使得大小写不分），再进行判断。

【解答】设计步骤如下。

（1）建立用户界面和设置对象属性，如图 7-1 所示。



（a）　　（b）

图 7-1　建立用户界面和程序运行结果

（2）编写事件代码。

编写“统计字母”命令按钮 Command1 的单击 Click 事件代码：

```
Private Sub Command1_Click( )
  Dim a(1 To 26) As Integer
  Dim i As Integer, j As Integer, c As String * 1
  tot = Len(Text1)                          ' 求字符串的长度
  For i = 1 To tot
    c = UCase(Mid(Text1, i, 1))             ' 取一个字符，将其转换成相应的大写字母
    If c >= "A" And c <= "Z" Then
      j = Asc(c) – 65 + 1                   ' 将 A～Z 大写字母转换成 1～26 的下标
      a(j) = a(j) + 1                       ' 对应数组元素加 1
```

```
    End If
  Next i
  For j = 1 To 26                              ' 输出字母及其出现的次数
    If a(j) > 0 Then
      Label2.Caption = Label2.Caption & Chr$(j + 64) & "=" & a(j) & " "
    End If
  Next j
End Sub
```

7.13　随机产生 10 个两位整数，找出其中的最大值、最小值和平均值。

【分析】问题可以分为两部分：一是产生 10 个随机整数，二是对这 10 个整数求最大值、最小值及平均值。为此，需要使用数组。

【解答】设计步骤如下。

（1）建立应用程序用户界面并设置对象属性。

新建一个工程，进入窗体设计器，首先增加一个框架 Frame1，3 个标签 Label1～Label3，3 个文本框 Text1～Text3，3 个命令按钮 Command1～Command3。激活 Frame1 后，在其中增加一个标签 Label4，并修改各个控件的属性，如图 7-2 所示。



图 7-2　求最大值、最小值和平均值

（2）编写代码。

考虑到要在不同的过程中使用数组，所以首先在模块的通用段中声明数组：

```
Dim a(1 To 10) As Integer
```

随机整数的生成由窗体的 Load 事件代码完成：

```
Private Sub Form_Load( )
  Dim p As String
  Randomize
  p = ""
  For i = 1 To 10
    a(i) = Int(Rnd * 90) + 10
    p = p & Str(a(i)) & ","
  Next
  Label1.Caption = LTrim(Left(p, Len(p) – 1))
End Sub
```

求最大值、最小值及平均值由“确定”按钮 Command2 的 Click 事件代码完成：

```
Private Sub Command2_Click( )
  Dim n As Integer, m As Integer, s As Single
  min = 100: max = 10: s = 0
  For i = 1 To 10
    If a(i) > n Then max = a(i)
```

```
        If a(i) < m Then min = a(i)
        s = s + a(i)
    Next
    Text1.Text = max
    Text2.Text = min
    Text3.Text = s / 10
End Sub
```

“重置”按钮 Command1 的 Click 事件代码：

```
Private Sub Command1_Click( )
    Form_Load
    Text1.Text = ""
    Text2.Text = ""
    Text3.Text = ""
End Sub
```

最后是“关闭”按钮 Command3 的 Click 事件代码：

```
Private Sub Command3_Click( )
    Unload Me
End Sub
```

【说明】在“重置”按钮的 Click 事件代码中，可以通过调用窗体的 Load 事件过程来重新产生随机数。

如果要求产生的随机整数互不相同，应改写窗体的 Load 事件代码：

```
Private Sub Form_Load( )
    Dim p As String
    Randomize
    p = ""
    For i = 1 To 10
        Do
            x = Int(Rnd * 90) + 10
            yes = 0
            For j = 1 To i - 1
                If x = a(j) Then yes = 1: Exit For          ' 若与前面的元素相同，则返回到 Do 循环
            Next                                            ' 重新产生随机数
        Loop While yes = 1
        a(i) = x
        p = p & Str(a(i)) & ","
    Next
    Label1.Caption = LTrim(Left(p, Len(p) - 1))
End Sub
```

其中，变量 x 用来存放刚产生的随机整数。变量 yes 用来作为标志，如果 x 与已放入数组中的某个随机整数相同，则 yes 为 1，否则为 0。当 yes 为 0 时，将退出第二层 Do…Loop

循环，把随机数放入数组之中：a(i)=x。

7.14　利用随机函数模拟投币结果。设共投币 100 次，求“两个正面”、“两个反面”、“一正一反”3 种情况各出现多少次。

【解答】定义一个二维数组 a(1,1)，其中各元素的含义如下：

a(0,0)——累计“两个正面”的次数；

a(1,1)——累计“两个反面”的次数；

a(0,1)和 a(1,0)——累计“一正一反”或“一反一正”的次数。

为简单起见，本题直接输出结果到窗体上，如图 7-3 所示。

编写命令按钮的 Click 事件代码如下：

```
Private Sub Command1_Click( )
  Dim a(1, 1)
  Dim n As Integer
  n = Val(InputBox("n=", "请输入投币次数"))
  Label1.Caption = "投币次数：" & n
  Randomize Timer
  For i = 1 To n
    n1 = Int(Rnd * 2)
    n2 = Int(Rnd * 2)
    a(n1, n2) = a(n1, n2) + 1
  Next i
  Cls
  CurrentY = Command1.Top + Command1.Height + 50
  Print
  Print Tab(10); "～～～～～～～～～～"
  Print Tab(14); "投币结果如下:"
  Print Tab(10); "～～～～～～～～～～"
  Print
  Print Tab(10); "两个正面的次数为:"; a(0, 0)
  Print
  Print Tab(10); "两个反面的次数为:"; a(1, 1)
  Print
  Print Tab(10); "一正一反的次数为:"; a(0, 1) + a(1, 0)
End Sub
```



图 7-3　模拟投币结果

请读者修改本题，在显示模拟投币结果时，可同时显示钱币图案（正面、反面）。

7.15　某数组有 20 个元素，元素的值由键盘输入，要求将前 10 个元素与后 10 个元素对换，即第 1 个元素与第 20 个元素对换，第 2 个元素与第 19 个元素对换，……，第 10 个元素与第 11 个元素对换。输出数组原来各元素的值和对换后各元素的值。

【解答】根据题意，定义一个具有 20 个元素的数组 a(1 To 20)，首先对其进行赋值，可利用 For 循环来对这 20 个元素依次进行键盘输入。然后按此顺序输出，即对换前各元素的值。进行对换操作时，按要求将前 10 个元素与后 10 个元素进行对换，方法是将 a(i)与 a(20−i+1)

对换。最后输出对换后各元素的值。

设计步骤如下。

（1）设计程序界面和设置对象属性如图 7-4（a）所示。



（a）　　（b）

图 7-4　设计窗体界面与运行程序

（2）编写代码。

首先在“通用”段声明数组：

```
Dim a(1 To 20) As Integer
```

编写“输入”按钮 Command1 的 Click 事件代码：

```
Private Sub Command1_Click( )
  Dim i As Integer, t As Integer
  Dim p As String
  For i = 1 To 20                ' 输入 20 个元素的值
     a(i) = InputBox("输入 a(" & Format(i, "#") & "):")
  Next i
  For i = 1 To 20                ' 连接对换前各元素的值
     p = p & a(i)
     Select Case i
        Case 10
           p = p & "," & Chr(13)
        Case 20
        Case Else
           p = p & ","
     End Select
  Next i
  Label1.Caption = p             ' 输出对换前各元素的值
End Sub
```

编写“对换”按钮 Command2 的 Click 事件代码：

```
Private Sub Command2_Click( )
  Dim p As String
  For i = 1 To 10                                  ' 对换操作
    t = a(i): a(i) = a(20 – i + 1): a(20 – i + 1) = t     ' 对换
  Next i
  For i = 1 To 20                                  ' 连接对换后各元素的值
```

```
            p = p & a(i)
            Select Case i
               Case 10
                  p = p & "," & Chr(13)
               Case 20
               Case Else
                  p = p & ","
            End Select
         Next i
         Label2.Caption = p                    ' 输出对换后各元素的值
      End Sub
```

假设依次输入 10, 11, 12, 13, …, 29，运行程序显示如图 7-4（b）所示。

7.16　有一个 6×6 的矩阵，各元素的值由键盘输入，求全部元素的平均值，并输出高于平均值的元素及它们的行、列号。

【解答】首先应定义一个 6×6 的二维数组 a(6, 6)，用双重循环输入二维数组的各元素值。然后将这些值累加，由此求得平均值 av（累加和/总元素数）。最后按行 n、列 m 逐个元素判断其值 a(n, m)是否大于 av，如果为真，则输出其值 a(n, m)及其所在行 n 和所在列 m。

设计步骤如下。

（1）设计程序界面并设置对象属性。

新建一个工程，进入窗体设计器，在窗体中增加一个图片框 Picture1 和一个框架 Frame1。选中 Frame1，在其中增加一个标签 Label1 和两个命令按钮 Command1, Command2。

设置对象属性如图 7-5（a）所示。



（a）　　（b）

图 7-5　设计窗体界面与运行程序

（2）编写代码。

首先在“通用”段声明数组：

```
Dim a(6 ,6) As Integer
```

“输入”按钮 Command1 的 Click 事件代码：

```
Private Sub Command1_Click( )
   Dim n As Integer, m As Integer
   For n = 1 To 6
      For m = 1 To 6
```

```
            x = InputBox("a(" + Str(n) + "," + Str(m) + ")=")
            a(n, m) = Val(x)
        Next
    Next
    Picture1.CurrentY = 80
    For n = 1 To 6
        For m = 1 To 6
            Picture1.Print Format(a(n, m), "@@@@");
        Next
        Picture1.Print
        Picture1.Print
    Next
End Sub
```

“计算”按钮 Command2 的 Click 事件代码：

```
Private Sub Command2_Click( )
    s = 0
    For n = 1 To 6
        For m = 1 To 6
            s = s + a(n, m)
        Next
    Next
    av = s / (6 * 6)
    Label1.Caption = "平均值= " & Format(av, "######")
    p = ""
    For n = 1 To 6
        For m = 1 To 6
            If a(n, m) > av Then
                p = p & "a(" & n & "," & m & ")=" & a(n, m) & " "
            End If
        Next
    Next
    MsgBox p, 0, "大于平均值的元素是"
End Sub
```

程序运行结果如图 7-5（b）所示。

【说明】也可以用列表框控件来显示矩阵中的元素。

7.17　编写应用程序，输入两个数，根据不同运算符计算结果。

【分析】假设需要进行的有加、减、乘、除、整除、余数、指数和字符串连接 8 种运算，并根据运算方式，显示计算结果。

设计步骤如下。

（1）建立应用程序用户界面并设置对象属性。

利用 Label 控件，在窗体上拖出 4 个标签对象 Label1, Label2, Label3, Label4。利用 TextBox 控件，拖出 Text1 和 Text2 文本框对象。

利用 Frame 控件，拖出 Frame1 框对象。利用 OptionButton 控件，拖出置于 Frame1 中的 Option1 单选钮。单击“复制”按钮，将 Option1 单选钮的对象复制到剪贴板中，再选中 Frame1 对象，单击“粘贴”按钮，单击“是”按钮，将产生一个属于单选钮类别的控件数组，将 Frame1 对象内左上角新产生的 Option1 对象拖到 Frame1 框内已有项目的下方。如此重复粘贴，直到产生 8 个 Option1 对象，如图 7-6（a）所示，并依次改变标题名。



（a）　　（b）

图 7-6　用户界面与程序运行结果

（2）编写程序代码。

编写选项按钮 Option1 的 Click 事件代码为：

```
Private Sub Option1_Click(Index As Integer)
  Dim x As Single, y As Single
  x = Val(Text1.Text)
  y = Val(Text2.Text)
  Select Case Index                    ' Index 值从 0 开始
    Case 0                             ' 加
      Label4.Caption = x + y
    Case 1                             ' 减
      Label4.Caption = x – y
    Case 2                             ' 乘
      Label4.Caption = x * y
    Case 3                             ' 除
      Label4.Caption = x / y
    Case 4                             ' 整除
      Label4.Caption = X \ Y
    Case 5                             ' 余数
      Label4.Caption = x Mod y
    Case 6                             ' 指数
      Label4.Caption = x ^ y
    Case Else                          ' 字符串连接
```

```
        Label4.Caption = x & y
    End Select
End Sub
```

程序运行结果如图 7-6（b）所示。

7.18　编写矩阵转置程序，将一个 6 行 4 列的矩阵行、列互换（其中的矩阵元素随机产生），即

$$\begin{bmatrix} a_{11} & a_{12} & a_{13} & a_{14} \\ a_{21} & a_{22} & a_{23} & a_{24} \\ a_{31} & a_{32} & a_{33} & a_{34} \\ a_{41} & a_{42} & a_{43} & a_{44} \\ a_{51} & a_{52} & a_{53} & a_{54} \\ a_{61} & a_{62} & a_{63} & a_{64} \end{bmatrix} \xrightarrow{\text{转置}} \begin{bmatrix} a_{11} & a_{21} & a_{31} & a_{41} & a_{51} & a_{61} \\ a_{12} & a_{22} & a_{32} & a_{42} & a_{52} & a_{62} \\ a_{13} & a_{23} & a_{33} & a_{43} & a_{53} & a_{63} \\ a_{14} & a_{24} & a_{34} & a_{44} & a_{54} & a_{64} \end{bmatrix}$$

【解答】定义两个二维数组 a(6,4)和 b(4,6)，用双重循环和随机函数产生二维数组 a(6,4)中的各元素值，然后为 b(4,6)赋值：b(n,m)=a(m,n)。

设计步骤如下。

（1）设计程序界面并设置对象属性。

新建一个工程，进入窗体设计器，在窗体中增加两个图片框 Picture1，Picture2 和两个命令按钮 Command1, Command2。

设置对象属性如图 7-7（a）所示。



（a）　　（b）

图 7-7　设计窗体界面与运行程序

（2）编写代码。

首先在“通用”段声明数组：

```
Dim a(6, 4) As Integer, b(4, 6) As Integer
```

“重置”按钮 Command1 的 Click 事件代码：

```
Private Sub Command1_Click( )
  For n = 1 To 6
    For m = 1 To 4
      x = Int(Rnd * 90) + 10
      a(n, m) = Val(x)
    Next
  Next
  Picture1.Cls
```

```
    Picture2.Cls
    Picture1.CurrentY = 80
    For n = 1 To 6
      For m = 1 To 4
        Picture1.Print Format(a(n, m), "@@@");
      Next
      Picture1.Print
      Picture1.Print
    Next
  End Sub
```

“转置”按钮 Command2 的 Click 事件代码：

```
  Private Sub Command2_Click( )
    For n = 1 To 6
      For m = 1 To 4
        b(m, n) = a(n, m)
      Next
    Next
    Picture2.Cls
    Picture2.CurrentY = 80
    For n = 1 To 4
      For m = 1 To 6
        Picture2.Print Format(b(n, m), "@@@");
      Next
      Picture2.Print
      Picture2.Print
    Next
  End Sub
```

程序运行结果如图 7-7（b）所示。

【说明】本题也可以只用一个数组，在输出的时候，将行、列下标互换：

```
  Private Sub Command2_Click( )
    Picture2.Cls
    Picture2.CurrentY = 80
    For n = 1 To 4
      For m = 1 To 6
        Picture2.Print Format(a(m, n), "@@@");
      Next
      Picture2.Print
      Picture2.Print
    Next
  End Sub
```

7.19　编写矩阵的加法运算程序。两个相同阶数的矩阵 *A* 和 *B*（随机产生）相加，就是将相应位置上的元素相加后放到同阶矩阵 *C* 的相应位置。

【分析】定义 3 个二维数组 a(n,m)，b(n,m)，c(n,m)，分别代表矩阵 ***A***、***B***、***C***，利用双重循环和随机函数产生 a(n,m)和 b(n,m)中各元素的值，然后通过双重循环得到 c(n,m)。

【解答】设计步骤如下。



图 7-8　设计窗体界面与运行程序

（1）设计程序界面并设置对象属性。

新建一个工程，进入窗体设计器，在窗体中增加 3 个图片框 Picture1～Picture3，两个标签 Label1, Label2 和两个命令按钮 Command1, Command2。

设置对象属性如图 7-8 所示。

（2）编写代码。

在“通用”段声明数组：

```
Dim a(5, 3) As Integer, b(5, 3) As Integer
```

“重置”按钮 Command1 的 Click 事件代码：

```
Private Sub Command1_Click( )
  For n = 1 To 5                                    ' 控制矩阵行数
    For m = 1 To 3                                  ' 控制矩阵列数
      x = Int(Rnd * 100): a(n, m) = Val(x)          ' 产生随机数
      x = Int(Rnd * 100): b(n, m) = Val(x)          ' 产生随机数
    Next
  Next
  Picture1.Cls: Picture2.Cls: Picture3.Cls          ' 清空图片框
  Picture1.CurrentY = 80: Picture2.CurrentY = 80    ' 控制输出位置
  For n = 1 To 5
    For m = 1 To 3
      Picture1.Print Format(a(n, m), "@@@@");       ' 输出产生的随机数
      Picture2.Print Format(b(n, m), "@@@@");       ' 输出产生的随机数
    Next
    Picture1.Print: Picture2.Print                  ' 换行
  Next
End Sub
```

“计算”按钮 Command2 的 Click 事件代码：

```
Private Sub Command2_Click( )
  Dim c(5, 3) As Integer
  For i = 1 To 5
    For j = 1 To 3
      c(i, j) = a(i, j) + b(i, j)                   ' 求两矩阵相加的各项值
    Next
  Next
  Picture3.Cls                                      ' 清空
```

```
    Picture3.CurrentY = 80                                  ' 确定输出位置
    For n = 1 To 5
      For m = 1 To 3
        Picture3.Print Format(c(n, m), "@@@@");             ' 输出结果数据
      Next
      Picture3.Print                                        ' 换行
    Next
End Sub
```

运行程序，单击“重置”按钮产生原始数据，单击“计算”按钮得到相加结果。如果再次单击“重置”按钮，则重新产生数据，如图 7-9 所示。



（a）　　（b）

图 7-9　两个矩阵相加运行结果

7.20　输出幻方阵。幻方阵也称魔方阵，是指由自然数 1～n^2（n 为奇数）构成的方阵，其各行、各列及对角线元素之和均相等。如图 7-10 所示。

（a）　　（b）　　（c）

图 7-10　幻方阵

【分析】先将数 1 放在第 1 行的当中一列，2～n^2 中的各数依次按以下规律排列：

① 每一个数存放的行数比前一个数的行数减 1，列数加 1；

② 若上一个数在第 1 行，则下一个数在第 n 行（最下一行），列数加 1；

③ 若上一个数在第 n 列，则下一个数在第 1 列，而行数减 1；

④ 若某数按以上规律应放的位置已被其他数占据，则该数放在上一数的下面一列。

【解答】设计步骤如下。

（1）设计程序界面并设置对象属性，如图 7-10 所示。其中单选钮控件数组中的第 1 个单选钮 Option1(0)的 Value 属性必须设为 True。

（2）编写代码。

首先在“通用”段声明动态数组：

```
Dim a( ) As Integer
```

窗体的 Activate 事件代码：

```
Private Sub Form_Activate( )
   Select Case True                    ' 检查哪个按钮被选中
      Case Option1(0)
         n = 3
      Case Option1(1)
         n = 5
      Case Option1(2)
         n = 7
   End Select
   Picture1.Height = 350 * n           ' 设置图片框的大小与位置
   Picture1.Width = 350 * n
   Picture1.Top = (Me.ScaleHeight – Picture1.Height) / 2
   Picture1.Left = (Frame1.Left – Picture1.Width) / 2
   ReDim a(n, n)                       ' 定义数组的大小
   i = 1                               ' 安排数 1 的位置
   j = (n + 1) / 2
   a(i, j) = 1
   Picture1.Cls
   For x = 2 To n * n                  ' 安排数 2 ~ n * n 的位置
      If (x – 1) Mod n = 0 Then
         i = i + 1
      Else
         i = IIf(i = 1, n, i – 1)
         j = IIf(j = n, 1, j + 1)
      End If
      a(i, j) = x
   Next
   For i = 1 To n                      ' 输出数组
      Picture1.CurrentY = (i – 1) * Picture1.Height / n + 60
      Picture1.CurrentX = 15 * n
      For j = 1 To n
         Picture1.Print Format(a(i, j), "@@@");
      Next
      Picture1.Print
   Next
End Sub
```

单选钮控件数组 Option1( )的 Click 事件代码：

```
Private Sub Option1_Click(Index As Integer)
   Form_Activate
End Sub
```

7.21　利用一维数组统计一个班学生 0～9 分，10～19 分，20～29 分，…，90～99 分及 100 分各分数段的人数。

【解答】定义一个有 11 个元素的一维数组 a(0 To 10)，把 0～9 分的学生人数存入 a(0)中，把 11～19 分的学生人数存入 a(1)中……把 100 分的人数存入 a(10)中。

设计步骤如下。

（1）设计程序界面并设置对象属性，如图 7-11 所示。

（2）编写代码。

首先在“通用”段声明动态数组：

```
Dim a(10) As Integer
```

命令按钮 Command1 的 Click 事件代码：

```
Private Sub Command1_Click( )
  Dim i As Integer, p As Integer, n As Integer
  Dim x As Single
  n = InputBox("n=", "请输入学生数")
  Label1.Caption = "共有人数：" & n
  For i = 1 To n
    x = Val(InputBox("请输入第" & Str(i) & "名学生的成绩", ""))
    If x >= 0 And x <= 100 Then
      p = Int(x / 10)
      a(p) = a(p) + 1
    Else
      MsgBox "请输入正确分数！"
      i = i – 1
    End If
  Next i
  For p = 0 To 9
    List1.AddItem p * 10 & "～" & (p * 10 + 9) & "分的人数为:" & a(p)
  Next p
  List1.AddItem "100 分的人数为:" & a(10)
End Sub
```

本题也可以不使用列表框控件，直接使用 Print 方法输出到窗体上或图片框中，如图 7-12 所示。

命令按钮的 Click 事件代码为：

图 7-11　统计各分数段的人数

图 7-12　输出到窗体上

```
Private Sub Command1_Click( )
   Dim p As Integer, n As Integer
   Dim x As Single
   n = InputBox("n=", "请输入学生数")
   Label1.Caption = "共有人数：" & n
   For i = 1 To n
      x = Val(InputBox("请输入第" & Str(i) & "名学生的成绩", ""))
      If x >= 0 And x <= 100 Then
         p = Int(x / 10)
         a(p) = a(p) + 1
      Else
         MsgBox "请输入正确分数！"
         i = i – 1
      End If
   Next i
   Cls
   Print
   For p = 0 To 9
      Print Tab(2); p * 10 & "～" & (p * 10 + 9) & "分的人数为:"; Tab(20); a(p)
   Next p
   Print Tab(2); "100 分的人数为:"; Tab(20); a(10)
End Sub
```

7.22　设计一个“通信录”程序，如图 7-13（a）所示。当用户在“选择姓名”下拉列表框中选择某一人名后，在“电话号码”文本框中显示出对应的电话号码，如图 7-13（b）所示。当用户选择或取消“单位”和“住址”复选框后，将打开或关闭“工作单位”和“家庭住址”文本框，如图 7-13 所示（c）。

【解答】设计步骤如下。

（1）设计程序界面并设置控件属性。

首先在窗体中增加一个框架 Frame1。选中 Frame1，在其中添加一个组合框 Combo1，一个文本框控件数组 Text1(0)～Text1(2)，一个标签控件数组 Label1(0)～Label1(3)和一个复选框控件数组 Check1(0)～Check1(1)。

对象属性设置，见表 7-1。



（a）　　（b）　　（c）

图 7-13　“通信录”程序

（2）编写程序代码。

在窗体的通用段声明对象数组变量：

```
Dim xm(10) As String, dh(10) As String, dw(10) As String, zz(10) As String
```

表 7-1　属性设置

对　　象	属　　性	属　性　值
Frame1	Caption	
Label1(0)～Label1(3)	Caption	电话号码：、工作单位：、家庭住址：、选择姓名：
Text1(1)～Text1(2)	Visible	True
Check1(0)～Check1(1)	Caption	单位、住址

窗体的 Load 事件代码：

```
Private Sub Form_Load( )
  xm(0) = "张大千": dh(0) = "1234567": dw(0) = "计算机系": zz(0) = "1 号楼 15 号"
  xm(1) = "李丽英": dh(1) = "2345678": dw(1) = "计算机系": zz(1) = "2 号楼 16 号"
  xm(2) = "王大鹏": dh(2) = "3456789": dw(2) = "经管系": zz(2) = "2 号楼 1 号"
  xm(3) = "赵晓岩": dh(3) = "4567890": dw(3) = "水利系": zz(3) = "1 号楼 5 号"
  xm(4) = "孙文霞": dh(4) = "5678901": dw(4) = "机电系": zz(4) = "2 号楼 15 号"
  xm(5) = "刘建伟": dh(5) = "6789012": dw(5) = "环建系": zz(5) = "3 号楼 6 号"
End Sub
```

窗体的 Activate 事件代码：

```
Private Sub Form_Activate( )
  For n = 0 To 5
    Combo1.AddItem xm(n)
  Next
  Combo1.ListIndex = 0
End Sub
```

组合框 Combo1 的 Click 事件代码：

```
Private Sub Combo1_Click( )
  Text1(0).Text = dh(Combo1.ListIndex)
  Text1(1).Text = dw(Combo1.ListIndex)
  Text1(2).Text = zz(Combo1.ListIndex)
End Sub
```

组合框 Check1 的 Click 事件代码：

```
Private Sub Check1_Click(Index As Integer)
  Text1(Index + 1).Visible = Check1(Index).Value
  Label1(Index + 1).Visible = Check1(Index).Value
End Sub
```

7.23　设计简易计算器，如图 7-14 所示。



（a）　　（b）　　（c）

图 7-14　简易计算器

【分析】程序中的按钮分为两类：数字类和运算符类，需要分别使用两个命令按钮控件数组。

【解答】设计步骤如下。

（1）建立应用程序用户界面并设置对象属性。

新建一个工程，进入窗体设计器，首先增加一个框架控件 Frame1，选中 Frame1 后，在其中增加一个文本框控件 Text1、两个命令按钮控件数组 Command1(0)～Command1(10)和 Command2(0)～Command2(4)。属性设置见表 7-2。

表 7-2　属性设置

对　象	属　性	属　性　值
Text1	Caption	
	Alignment	1-Right Justify
	Locked	True
Command1(0)～Command1(10)	Caption	0, 1, 2, 3, 4, 5, 6, 7, 8, 9, .（小数点）
Command2(0)～Command2(4)	Caption	+, –, *, /, =

（2）编写程序代码。

首先在模块的通用段声明变量：

```
Dim v As Boolean              ' 是否第一次按运算符
Dim s As Integer              ' 存放上次按的运算符
Dim x As Double               ' 存放第一个操作数
Dim y As Double               ' 存放第二个操作数
```

数字类命令按钮组 Command1( ) 的 Click 事件代码：

```
Private Sub Command1_Click(Index As Integer)
  If Form1.Tag = "T" Then        ' 向显示中的数追加新数
    If Index = 10 Then
      Text1.Text = "0."
    Else
      Text1.Text = Command1(Index).Caption
    End If
    Form1.Tag = ""
```

```
    Else
        Text1.Text = Text1.Text & Command1(Index).Caption
    End If
End Sub
```

运算符类命令按钮组 Command2( ) 的 Click 事件代码：

```
Private Sub Command2_Click(Index As Integer)
    Form1.Tag = "T"
    If v Then                    ' 第一次按运算符
        x = Val(Text1.Text)          ' 将输入的数存入 x 中
        v = Not v
    Else
        y = Val(Text1.Text)
        Select Case s
            Case 0
                Text1.Text = x + y
            Case 1
                Text1.Text = x – y
            Case 2
                Text1.Text = x * y
            Case 3
                If y < > 0 Then
                    Text1.Text = x / y
                Else
                    MsgBox ("不能以 0 为除数")
                    Text1.Text = x
                    v = False
                End If
            Case 4
                y = 0
                v = False
        End Select
        x = Val(Text1.Text)
    End If
    s = Index
End Sub
```

7.24　设有一个 5×5 的方阵，其中元素是由计算机随机生成的小于 100 的整数。求：

（1）对角线上元素之和；

（2）对角线上元素之积；

（3）方阵中最大的元素。

【分析】方阵中的元素可以用一个二维的数组来表示。利用单层的循环就可以计算出对角线上元素的和、积，求方阵中的最大元素则需要利用双层循环。

【解答】设计步骤如下。

（1）建立应用程序用户界面并设置对象属性。

程序界面的建立与各控件属性的设置，如图 7-15 所示。



（a）　　（b）

图 7-15　矩阵计算

（2）编写代码。

考虑到要在不同的过程中使用数组，所以首先在模块的通用段声明数组：

```
Dim a(5, 5) As Integer
```

方阵的生成由窗体的 Load 事件代码完成：

```
Private Sub Form_Load( )
  List1.Clear
  Dim p As String
  Randomize
  For i = 1 To 5
    p = ""
    For j = 1 To 5
      a(i, j) = Int(Rnd * 99) + 1
      p = p & Format(a(i, j), "!@@@")
    Next
    List1.AddItem p, i – 1
  Next
  Text1.Text = ""
  Text2.Text = ""
  Text3.Text = ""
End Sub
```

计算功能由“计算”按钮 Command1 的 Click 事件代码完成：

```
Private Sub Command1_Click( )
  Dim s As Integer, t As Long
  s = 0: t = 1
  For i = 1 To 5
    s = s + a(i, i)
```

```
        t = t * a(i, i)
        If Max < a(i, i) Then Max = a(i, i): p = i
      Next
      Max = a(1, 1): p = 1: q = 1
      For i = 1 To 5
        For j = 1 To 5
          If Max < a(i, j) Then Max = a(i, j): p = i: q = j
        Next
      Next
      Text1.Text = s
      Text2.Text = t
      Text3.Text = "a(" & p & "," & q & ")"
    End Sub
```

“重置”按钮 Command2 的 Click 事件代码完成：

```
    Private Sub Command2_Click( )
      Form_Load
    End Sub
```

第8章　过　　程

一、选择题

8.1　在 VB 工程中，可以作为“启动对象”的程序是（　　）。

A）任何窗体或标准模块　　B）任何窗体或过程

C）Sub Main 过程或其他任何模块　　D）Sub Main 过程或任何窗体

【解答】正确答案为 D。

分析：在 VB 中，除了可以指定某个窗体作为启动对象外，还可以指定 Main 子过程为启动对象，如果不指定启动窗体则默认将工程中建立的第一个窗体作为启动窗体。

8.2　以下关于函数过程的叙述中，正确的是（　　）。

A）函数过程形参的类型与函数返回值的类型没有关系

B）在函数过程中，过程的返回值可以有多个

C）当数组作为函数过程的参数时，既能以传值方式传递，也能以传址方式传递

D）如果不指明函数过程参数的类型，则该参数没有数据类型

【解答】正确答案为 A。

分析：函数的返回值只有一个，选项 B 不正确；当数组作为函数的形参时，只能以传址方式传递，选项 C 不正确；如果不指明函数过程的参数类型，则该参数为变体类型，选项 D 不正确。

8.3　下列程序的执行结果为（　　）。

```
Private Sub Command1_Click( )
   Dim x As Integer , y As Integer
   x=12 : y=20
   Call Value(x , y)
   Print x ; y
End Sub
Private Sub Value(ByVal m As Integer, ByVal n As Integer)
    m=m*2   :   n=n−5
    Print m ; n
End Sub
```

A）20　12　　B）12　20　　C）24　15　　D）24　12

　　20　15　　12　25　　12　20　　12　15

【解答】正确答案为 C。

分析：被调用过程 Value 的两个参数 m 和 n 前面都有关键字 ByVal 修饰，即在主调过程调用此过程时，实参与形参之间是以传值方式传递信息的，而当实参与形参以传值方式相结

合时，形参的改变并没有影响到实参，单单就这一点来说，当程序代码执行主调过程中的 Print x;y 语句时，x 和 y 的值应不变，还是 12 和 20，这样就能够排除选项 A 和选项 B 了。当主调过程调用被调过程时，把实参 x、y 的值分别传给形参 m 和 n，这就是说此时形参 m 值为 12，n 值为 20，执行被调过程中的 m=m*2 : n=n-5 语句后，m 值为 24，n 值为 15，所以执行语句 Print m;n 后，程序输出的结果为 24 和 15。

8.4　在窗体上画一个名称为 Text1 的文本框，一个名称为 Command1 的命令按钮，然后编写如下事件过程和通用过程：

```
Private Sub Command1_Click( )
  n=Val(Text1.Text)
  If n\2=n/2 Then
     f=f1(n)
  Else
     f=f2(n)
  End If
  Print f; n
End Sub
Public Function f1(ByRef x)
  x=x*x
  f1=x+x
End Function
Public Function f2(ByVal x)
  x=x*x
  f1=x+x+x
End Function
```

程序运行后，在文本框中输入 6，然后单击命令按钮，窗体上显示的是（　　）。

A）72　36　　B）108　36　　C）72　6　　D）108　6

【解答】正确答案为 A。

分析：因为输入的 6 满足条件 n\2=n/2，所以执行 f=fl(n)，而 fl 中形参类型为 ByRef，即传址调用，所以在函数里形参的改变会影响到实参。

8.5　在窗体上画一个名称为 Command1 的命令按钮，然后编写如下通用过程和命令按钮的事件过程：

```
Private Function f(m As Integer)
  If m Mod 2=0 Then
    f=m
  Else
    f=1
  End If
End Function
```

```
Private Sub Command1_Click( )
  Dim i As Integer
  s=0
  For i=1 TO 5
    s=s+f(i)
  Next
  Print s
End Sub
```

程序运行后，单击命令按钮，在窗体上显示的是（　　）。

A）11　　B）10　　C）9　　D）8

【解答】正确答案为 C。

分析：本题的结果取决于函数 f，而 f 的作用是：如果参数为偶数，则返回实值；如果为奇数，则返回 1。

8.6　下面程序段，运行后的结果是（　　）。

```
Private Sub Command1_Click( )
  Dim b%(1 To 4), i%, t#
  For i=1 To 4
    b(i) =i
  Next i
  t=Tof(b())
  Print "t=" ; t,
End Sub
Function Tof(a( )As Integer)
  Dim t# , i%
  t=1
  For i=2 To UBound(a)
    t=t *a(i)
  Next i
  Tof=t
End Function
```

A）t=18　　B）t=24　　C）t=30　　D）t=32

【解答】正确答案为 B。

分析：程序开始调用循环语句对数组 b()赋值，b(i)=i，且数组 b 的上下界分别为 1 和 4。接着将数组 b()作为对象，调用 Tof 过程。在 Tof 过程中，定义了两个变量 t 和 i，i 作为 For 循环的循环变量，初始值为 2，终值为数组 b 的上界。循环体为“t=t*a(i)”，即将 b(i)的值乘以 t，然后再赋给 t。当循环结束后，程序返回的就是数组 b 中除第一个元素以外的所有元素值的乘积，此题的结果为 2×3×4=24。输出结果为 t=24。

8.7　单击按钮时，以下程序运行后的输出结果是（　　）。

```
Private Sub Command1_Click( )
  Dim x As Integer,y As Integer,z As Integer
  x=1:y=2:z=3
  Call God(x,x,z)
  Print x;x;z
  Call God(x,y,y)
  Print x;y;y
End Sub
Private Sub God(x As Integer,y As Integer,z As Integer)
  x=3*z+1
  y=2*z
  z=x+y
End Sub
```

A）6　6　12　　B）8　5　10　　C）9　6　12　　D）8　10　10

　　7　11　11　　　5　11　11　　　9　10　15　　　5　9　10

【解答】正确答案为 A。

分析：从整体上看，在两次调用过程 God 后，实参的值将随形参变化而变化。现在分析一下这两次调用实参和形参是怎么变化的。第一次调用被调过程 God，主调过程把实参 x，x 和 z 的地址分别传给形参 x，y 和 z，此时形参 x，y 和 z 的值分别为 1，1 和 3。执行语句 x=3*z+1 后，形参 x 值变为 10，此时相对应的实参 x 也变为 10。执行语句 y=2*z 后，形参 y 值变为 6，则相对应的实参 x 值变为 6。执行语句 z=x+y 后，形参 z 值应为 12，当然这次调用后，程序代码输出的数值为 6，6，12。第二次调用被调过程 God，主调过程把实参 x，x 和 y 的地址分别传给形参 x，y 和 z。应注意，此时实参 x 的值为 6 而不是 1，所以此时形参 x，y 和 z 值分别为 6，2 和 2。执行语句 x=3*z+1 后，形参 x 值为 7，相对实参 x 值也相应变为 7。执行语句 y=2*z 后，形参 y 值变为 4，相应的实参 y 值也变为 4，执行语句 z=x+y 后，形参 z 值变为 11，相应的实参 y 的值变为 11，而形参 y 和 z 的地址相同，所以最后它们值都应为 11，所以此次调用后，输出的数值为 7，11 和 11。

二、填空题

8.8　以下是一个计算矩形面积的程序，调用过程计算矩形面积。请将程序补充完整。

```
Sub RecArea(L , W)
  Dim S As Double
  S=L * W
  MsgBox "Total Area is" & Str(S)
End Sub
Private Sub Command1_Click( )
  Dim M,N
  M=InputBox("What is the L? ")
```

```
M=Val(M)
  ____
N=Val(N)
  ____
End Sub
```

【解答】正确答案为 N = InputBox("What is the W? ")；Call RecArea(M, N)。

分析：程序通过 RecArea 来计算并输出矩形的面积，它有两个形式参数，分别为矩形的长和宽，在 Sub 事件过程中，从键盘上输入矩形的长和宽，调用 InputBox 函数分别将输入的数赋给 M 和 B，所以第一个空应为 N = InputBox("What is the W?")。然后，将 M 和 N 作为实参调用 RecArea 过程，它有两种书写方式，一种是把过程的名字放在一个 Call 语句中，格式为 Call 过程名[(实际参数)]；另外一种是把过程名作为一个语句来使用，与第一种方式相比，它去掉关键字 Call，去掉了参数列表，所以最后一个空为 Call RecArea(M, N)或 RecArea(M, N)。

8.9 在窗体上画一个名称为 Command1 的命令按钮，然后编写如下程序：

```
Option Base 1
Private Sub Command1_Click( )
   Dim a(10)As Integer
   For i=1 To 10
      a(i)=i
   Next
   Call swap   ____
   For i=1 To 10
      print a(i);
   Next
End Sub
Sub swap(b( )As Integer)
   n= ____
   For i=1 To n/2
      t=b(i)
      b(i)=b(n)
      b(n)=t
      ____
   Next
End Sub
```

上述程序的功能是，通过调用过程 swap，调换数组中数值的存放位置，即 a(1)与 a(10)的值互换，a(2)与 a(9)的值互换，……，a(5)与 a(6)的值互换。请填全程序。

【解答】正确答案为 a；UBound(b)；n=n−1。

分析：子过程的功能是实现数组前后对应元素的互换。例如，a(1)和 a(10)，a(2)和 a(9)，其余类推；n=UBound(b)取得数组的上界下标；n=n−1 实现后面元素向前递变。

8.10　设有以下函数过程：

```
Function fun(m As Integer)As Integer
  Dim k As Integer , sum As Integer
  sum=0
  For k=m To 1 Step–2
    sum=sum+k
  Next k
  fun=sum
End Function
```

若在程序中用语句 s=fun(10)调用此函数，则 s 的值为____。

【解答】正确答案为 30。

分析：从函数定义的过程来看，此函数的作用是将从 m 开始依次递减 2 且不小于 1 的数累加起来，赋给 m，然后返回给函数。调用此函数后，s=fun(10)=10+8+6+4+2=30，故 s 的值为 30。

三、编程题

8.11　编写判断奇偶数的函数过程。输入一个整数，判断其奇偶性。

【解答】设计步骤如下。

（1）设计程序界面和设置对象属性，如图 8-1（a）所示。



（a）　　（b）

图 8-1　程序界面与运行程序

（2）编写代码。

首先编写自定义的函数过程代码：

```
Private Function Even(x As Long) As Boolean
  If x Mod 2 = 0 Then
    Even = True
  Else
    Even = False
  End If
End Function
```

命令按钮 Command1 的 Click 事件代码：

```
Private Sub Command1_Click( )
```

```
    Dim n As Long
    n = Val(Text1.Text)
    p = IIf(Even(n), "偶数", "奇数")
    Label1.Caption = n & "是一个" & p
    Text1.SetFocus
    Text1.SelStart = 0
    Text1.SelLength = Len(Text1.Text)
End Sub
```

程序运行结果如图 8-1（b）所示。

8.12　编写随机整数函数过程，产生 30 个 1～100 之间的随机数。

【解答】利用用户定义函数，定义随机整数函数，可以产生指定范围之内的随机整数。在主程序中指定范围和产生的个数。



图 8-2　设计程序界面和设置对象属性

程序的设计步骤如下。

（1）设计程序界面和设置对象属性，如图 8-2 所示。

（2）编写代码。

首先编写自定义函数过程，返回指定范围之内的随机整数，代码如下：

```
Private Function Randomnum(n As Integer, m As Integer)
    Randomize Timer
    Randomnum = Int(Rnd * (m + 1 – n)) + n
End Function
```

命令按钮控件数组 Command1 的 Click 事件代码：

```
Private Sub Command1_Click(Index As Integer)
    Dim k As Integer, x As Integer, y As Integer
    n = Index
    Select Case n
      Case 0
        List1.Clear
      Case 1
        k = Val(InputBox("随机数的个数：", "请输入", "100"))
        x = Val(Text1(0).Text): y = Val(Text1(1).Text)
        For i = 1 To k
          List1.AddItem Randomnum(x, y)
        Next
    End Select
End Sub
```

8.13　编写程序，实现英语单词或短语的加密/解密操作。加密/解密的基本原则是：把英语单词或短语中每个字符的 ASCII 码加上 2，使其变为另外一个字符。例如，把“ABCDE”

中每个字符的 ASCII 码加 2，变为“CEDFG”，从而对原来的单词或短语“加密”。

【解答】根据题目中要求的加密原则进行加密后，为了对加密后的单词或短语进行解密，应对加密后的单词或短语中各字符的 ASCII 码减去所加的值。例如，把“CDEFG”中每个字符的 ASCII 码减去 2，即可恢复为原来的“ABCDE”。

根据上面的分析，可以编写 3 个通用过程，分别执行加密、解密及输入要加密的单词或短语的操作。

（1）设计用户界面及对象属性，如图 8-3（a）所示。

（2）编写代码。

执行加密操作的函数过程如下：

```
Function en(inp As String)
Dim i As Integer
  Dim sout As String, scurrent As String, snew As String
  i = Len(inp)
  For x = 1 To i
    scurrent = Mid$(inp, x, 1)                    ' 截取字符串中的每个字符
    snew = Chr$(Asc(scurrent) + 2)                ' 每个字符的 ASCII 码值加 2 后的字符
    sout = sout & snew                            ' 字符连接
  Next x
  en = sout                                       ' 返回函数值
End Function
```

执行解密操作的函数过程如下：

```
Function de(inp As String)
  Dim i As Integer
  Dim sout As String, scurrent As String, snew As String
  i = Len(inp)
  For x = 1 To i
    scurrent = Mid$(inp, x, 1)                    ' 截取字符串中的每个字符
    snew = Chr$(Asc(scurrent) – 2)                ' 每个字符的 ASCII 码值减 2 后的字符
    sout = sout & snew                            ' 字符连接
  Next x
  de = sout                                       ' 返回函数值
End Function
```

在通用段声明变量：

```
Dim sph As String
```

“输入字符”命令按钮 Command1 的 Click 事件代码：

```
Private Sub Command1_Click( )
  Text1.Text = ""                                 ' 将文本框中的内容置空，准备接收字符
  Text1.SetFocus                                  ' 设置焦点
End Sub
```

“加密”命令按钮 Command2 的 Click 事件代码：

```
Private Sub Command2_Click( )
  Dim sen As String
  sen = en(Text1.Text)                    ' 调用 en 函数过程，对文本框中的字符进行加密
  Label1.Caption = sen                    ' 输出加密后的结果
End Sub
```

“解密”命令按钮 Command3 的 Click 事件代码：

```
Private Sub Command3_Click( )
  Dim sde As String
  sde = de(en(Text1.Text))                ' 调用 en 和 de 函数过程，对加密后的字符进行解密
  Label2.Caption = sde                    ' 输出解密后的结果
End Sub
```

运行程序，单击“输入字符”按钮，在文本框中输入一串字符；单击“加密”按钮，则显示加密后的单词或短语；单击“解密”按钮，则显示解密后的单词或短语，如图 8-3（b）所示。



（a）　　（b）

图 8-3　加密与解密的用户界面和运行结果

8.14　利用子过程或函数过程，求 1～6 的阶乘之和。

【分析】利用求阶乘的函数过程 fact，主程序通过调用该函数依次求得 1!, 2!, 3!,…, 6!的值，然后把这些值进行累加，即为所求的阶乘之和。

【解答】设计步骤如下。

（1）应用程序用户界面的建立与对象属性的设置，如图 8-4 所示。

（2）编写代码。

计算任意整数 n 的阶乘的函数过程 fact：

```
Function fact(x As Integer) As Long
  Dim p As Long, i As Integer
  p = 1
  For i = 1 To x
    p = p * i                             ' 累乘
  Next i
  fact = p                                ' 返回函数值
End Function
```

命令按钮的 Click 事件代码：

```
Private Sub Command1_Click( )
  Dim sum As Long, i As Integer          ' 定义数据类型
  For i = 1 To 6                         ' 求 1～6 的阶乘
    sum = sum + fact(i)                  ' 累加阶乘
  Next i
  Label2.Caption = sum                   ' 输出结果
End Sub
```

程序运行结果，如图 8-4 所示。

8.15　编写判断素数的子过程或函数过程，验证哥德巴赫猜想：一个不小于 6 的偶数可以表示为两个素数之和。例如：6 = 3 + 3, 8 = 3 + 5, 10 = 3 + 7, …。

【解答】假设有一个偶数 n，将它表示为两个整数 a 和 b 之和，即 $n = a + b$。如果 $n = 10$，先令 $a = 2$，判断 2 是否是素数，经检查 2 是素数，由于 $b = n–a$，故 b 的值为 8，经检查 8 不是素数，则这一组合 10 = 2 + 8 不合要求。再使 a 加 1，即 $a = 3$，经检查 3 是素数，$b = n–a = 7$，经检查 7 也是素数，则这一组合 10 = 3 + 7 符合要求。

程序的设计步骤如下。

（1）设计程序界面和设置对象属性，如图 8-5 所示。

图 8-4　求阶乘之和



图 8-5　验证哥德巴赫猜想

（2）编写代码。

方法 1：由于需要多次检查一个整数是否是素数，所以把判断是否是素数这一过程编写为一个子过程 Prime。

编写子过程如下：

```
Private Sub Prime(m As Long, f As Boolean)
  f = True
  If m > 3 Then
    For i = 3 To Sqr(m)
      If m Mod i = 0 Then f = False: Exit For
    Next
  End If
End Sub
```

命令按钮 Command1 的 Click 事件代码如下：

```
Private Sub Command1_Click( )
  Dim n As Long, x As Long, y As Long, p As Boolean
```

```
    n = Val(Text1.Text)
    If n < 6 Or n Mod 2 < > 0 Then
       MsgBox "必须输入大于 6 的偶数,请重新输入!"
       Cancel = True
    Else
       For x = 3 To n / 2 Step 2
          Call Prime(x, p)
          If p Then
             y = n – x: Call Prime(y, p)
             If p Then
                Text2.Text = x & "+" & y
                Exit For
             End If
          End If
       Next
    End If
    Text1.SelStart = 0
    Text1.SelLength = Len(Text1.Text)
End Sub
```

方法 2：也可以编写函数过程。

```
Private Function Prime(m As Long)
    Dim f As Boolean
    f = True
    If m > 3 Then
       For i = 3 To Sqr(m)
          If m Mod i = 0 Then f = False: Exit For
       Next
    End If
    Prime = f
End Function
```

命令按钮 Command1 的 Click 事件代码改为：

```
Private Sub Command1_Click( )
    Dim n As Long, x As Long, y As Long
    n = Val(Text1.Text)
    If n < 6 Or n Mod 2 < > 0 Then
       MsgBox "必须输入大于 6 的偶数,请重新输入!"
       Cancel = True
    Else
       For x = 3 To n / 2 Step 2
```

```
            If Prime(x) Then
               y = n – x
               If Prime(y) Then
                  Text2.Text = x & "+" & y
                  Exit For
               End If
            End If
         Next
      End If
      Text1.SelStart = 0
      Text1.SelLength = Len(Text1.Text)
   End Sub
```

方法 3：可以输出 6～200 之间所有偶数的“1+1”分解，如图 8-6 所示。

编写窗体的 Activate 事件代码：



图 8-6　输出 6～200 之间所有偶数的“1+1”分解

```
Private Sub Form_Activate( )
   Dim n As Long, x As Long, y As Long, p As Boolean
   For n = 6 To 200 Step 2
      For x = 3 To n / 2 Step 2
         If Prime(x) Then
            y = n – x
            If Prime(y) Then
               List1.AddItem "    " & n & " = " & x & " + " & y
               Exit For
            End If
         End If
      Next
   Next
End Sub
```

其中，调用了方法 2 中的函数过程 Prime( )。

8.16　编写判断某数是否能同时被 17 与 37 整除的函数过程，并输出 1000～2000 之间所有能同时被 17 与 37 整除的数。

【解答】设计步骤如下。

（1）设计程序界面和设置对象属性，如图 8-7（a）所示。



（a）　　（b）

图 8-7　程序界面与运行程序

（2）编写代码。

首先编写自定义的函数过程代码：

```
Private Function D(x As Integer) As Boolean
  If x Mod 17 = 0 And x Mod 37 Then
    D = True
  Else
    D = False
  End If
End Function
```

编写窗体的 Activate 事件代码：

```
Private Sub Form_Activate( )
  Dim n As Integer
  For n = 1000 To 2000
    If D(n) Then List1.AddItem n
  Next
End Sub
```

程序运行结果如图 8-7（b）所示。

8.17　使用 Timer 函数设计用来暂停指定时间（秒）的子过程。

【解答】简单的不精确延时，可以用 For…Next 循环来实现，如果要实现比较精确的时间延时，可以采用计时器控件，但要增加额外的“开支”，较为理想的方法是使用 Timer 函数。

Timer 函数返回一个 Single（单精度）数，表示从午夜开始到现在经过的秒数。把开始暂停的时刻加上需要延迟的时间（Start+PauseTime）作为循环结束的条件，当现在的时刻 Timer 超过这个时间时，结束循环。

程序设计步骤如下。

（1）设计程序界面和设置对象属性，如图 8-8 所示。



（a）　　（b）

图 8-8　程序界面和运行结果

（2）编写代码。

首先编写自定义的子过程代码：

```
Private Sub Delay(PauseTime)          ' PauseTime 为暂停时间
  Dim Start
  Start = Timer                       ' 设置开始暂停的时刻
  Do While Timer < Start + PauseTime
  Loop
End Sub
```

命令按钮 Command1 的 Click 事件代码：

```
Private Sub Command1_Click( )
  Cls
  Form1.FontSize = 16                    ' 设置字体大小
  Print
  Form1.ForeColor = QBColor(9)           ' 设置对象中文本或图形的前景色
  Print Tab(4); "暂停 3 秒钟"
  Print
  Delay (3)                              ' 暂停 3 秒
  Form1.ForeColor = QBColor(10)
  Print Tab(4); "暂停 5 秒钟"
  Print
  Delay 5                                ' 暂停 5 秒
  Form1.ForeColor = QBColor(12)
  Print Tab(4); "结束！"
  Print
  Delay (1)                              ' 暂停 1 秒
End Sub
```

为了在程序暂停时不至于使用户误认为“死机”了，可以在窗体的标题栏中显示剩余时间，如图 8-9 所示。

图 8-9　在窗体的标题栏中显示剩余时间

只需修改“延时”子过程代码：

```
Private Sub Delay(PauseTime)
  Start = Timer
  Do While Timer < Start + PauseTime
    m = Start + PauseTime – Timer      ' 计算剩余时间
    m0 = (m * 10) Mod 10               ' 剩余时间的 1 位小数
    m = Int(m)                         ' 剩余秒数
    Me.Caption = Format(m, "0:") & Format(m0, "0")
  Loop
End Sub
```

【说明】在一般情况下，结束时间大于开始时间，因此可以用 Timer < Start + PauseTime 作为循环结束的条件。但是，如果时间延时从午夜前开始（如 23:59:58），延时（如 5 秒）到午夜后结束（如 0:0:3），这时上面设计的延时程序将无法正常延时。请读者自行完善该延时程序，使之适合各种情况。

8.18　编写计算阶乘的函数过程，利用 e^x 的近似公式计算 e 值（直到最后一项小于 10^{-6} 为止）。

$$e^x \approx 1+\frac{x}{1!}+\frac{x^2}{2!}+\frac{x^3}{3!}+\cdots+\frac{x^n}{n!}$$

【解答】编写计算阶乘的函数过程代码如下：

```
Private Function Fact(x As Integer)
    Dim p As Long
    p = 1
    For n = 1 To x
        p = p * n
    Next
    Fact = p
End Function
```

在立即窗口输出计算结果的窗体代码如下：

```
Private Sub Form_Activate( )
    Dim n As Integer
    s = 1: n = 1: t = 1
    Do
        t = 1 / Fact(n)
        s = s + t
        n = n + 1
    Loop While t >= 10 ^ –8
    Debug.Print s
End Sub
```

8.19　编写函数过程返回指定字符、长度的字符串，实现在窗体上输出如图 8-10 所示的图形。



（a）　　（b）

图 8-10　输出图形

【解答】首先编写能够返回指定字符、长度的字符串的函数过程，代码如下：

```
Private Function stri(n As Integer, f As String)
    p = ""
    For i = 1 To n
        p = p & " " & f
    Next
    stri = p
End Function
```

编写命令按钮的 Click 事件代码，调用上述 stri 过程：

```
Private Sub Command1_Click( )
    Dim f As String * 1
    f = InputBox("显示的字符:", "请输入", "*")       ' 指定组成图形的字符
```

```
    If f = "" Then f = "*"
    Cls
    Print
    For n = 1 To 5                          ' 外循环输出 5 行
      Print Tab(2 * n + 2);
      Print stri(11 – 2 * n, f);            ' 输出左半部分
      Print Spc(4);                         ' 左右两部分之间的间隔
      Print stri(2 * n – 1, f);             ' 输出右半部分
      Print
    Next
End Sub
```

8.20　有 5 个人坐在一起。问第 5 个人的岁数，他说比第 4 个人大 2 岁。问第 4 个人，他说比第 3 个人大 2 岁。问第 3 个人，他说比第 2 个人大 2 岁。问第 2 个人，他说比第 1 个人大 2 岁。最后问第 1 个人，他说是 10 岁。请问第 5 个人有多大岁数。

【解答】很明显，这是一个递归问题。要想知道第 5 个人的年龄，就必须先知道第 4 个人的年龄，要知道第 4 个人的年龄就必须先知道第 3 个人的年龄，而第 3 个人的年龄又取决于第 2 个人的年龄，第 2 个人的年龄又取决于第 1 个人的年龄。其中，每一个人都比其前 1 个人大 2 岁。列出算式为：

```
age(5) = age(4) + 2
age(4) = age(3) + 2
age(3) = age(2) + 2
age(2) = age(1) + 2
age(1) = 10
```

可表述为下面的式子：

$$\text{age}(n)=\begin{cases}10 & n=1\\ \text{age}(n-1)+2 & n>1\end{cases}$$

可以看出，当 $n>1$ 时，求第 n 个人的年龄的公式是相同的。在调用时，只是每次的参数不同而已，该递归过程的结束条件是 $n=1$。

使用自定义的函数过程 age 来描述上述的递归过程，过程代码为：

```
Private Function age(n As Integer) As Integer
    If n = 1 Then
      age = 10
    Else
      age = age(n – 1) + 2
    End If
End Function
```

使用下面的窗体过程代码可以将结果输出到立即窗口中：

```
Private Sub Form_Load( )
    Debug.Print age(5)
End Sub
```

运行程序，即可在立即窗口中得到第 5 个人的年龄，如图 8-11 所示。



图 8-11　将结果输出到立即窗口中

8.21　图 8-12 中各图分别由若干个大小不等、形状相同的三角形构成。形成该图的方法是从一个大的等边三角形开始，将其三条边的中点进行连线，分成相同的 4 个三角形，对除中间外的 3 个三角形再重复上述过程，直到到达满足给定条件的底层为止。



（a）　（b）　（c）

图 8-12　递归调用示例

【解答】实现该方法采用递归子过程可使程序最简捷。首先编写子过程：

```
Private Sub triangle(x1!, x2!, x3!, y1!, y2!, k%)
  Dim u1!, u2!, v1!, v2!
  If k > 1 Then                                    ' 当没有到底层时，递归调用
    u1 = (x1 + x2) / 2                             ' 三条边的中点
    u2 = (x2 + x3) / 2
    v1 = (y1 + y2) / 2
    Call triangle(u1, x2, u2, v1, y2, k – 1)       ' 递归调用
    Call triangle(x1, u1, x2, y1, v1, k – 1)
    Call triangle(x2, u2, x3, y1, v1, k – 1)
  Else                                             ' 到达递归底层画三角形
    Picture1.Line (x1, y1) –(x3, y1)               ' 从点(x1, y1)到点(x3, y1)画直线
    Picture1.Line (x1, y1) –(x2, y2)
    Picture1.Line (x2, y2) –(x3, y1)
  End If
End Sub
```

编写 Picture1 的 Click 事件代码：

```
Private Sub Picture1_Click( )
  Dim n As Integer
  n = InputBox("输入 n 的值")                      ' 输入需要递归的层次
  Picture1.Print Tab(5); "递归　n=" & Str(n) & "　时的图形"
  Picture1.Scale (0, 600) –(600, 0)                ' 自定义坐标系
  Call triangle(30, 320, 570, 30, 570, n)          ' 调用画三角形过程
End Sub
```

第 9 章　变量与过程的作用域

一、选择题

9.1　以下叙述中错误的是（　　）。

A）在工程资源管理器窗口中只能包含一个工程文件及属于该工程的其他文件

B）以.BAS 为扩展名的文件是标准模块文件

C）窗体文件包含该窗体及其控件的属性

D）一个工程中可以含有多个标准模块文件

【解答】正确答案为 A。

分析：在工程资源管理器中可以包含多个工程，故选项 A 错误，其他各选项都正确，所以答案为选项 A。

9.2　如果一个工程含有多个窗体及标准模块，则以下叙述中错误的是（　　）。

A）如果工程中含有 Sub Main 过程，则程序一定首先执行该过程

B）不能把标准模块设置为启动模块

C）用 Hide 方法只是隐藏一个窗体，不能从内存中清除该窗体

D）任何时刻最多只有一个窗体是活动窗体

【解答】正确答案为 A。

分析：VB 中程序运行时，首先运行默认的启动窗体，所以选项 A 不正确，其他选项都正确，故答案为选项 A。

9.3　以下关于变量作用域的叙述中，正确的是（　　）。

A）窗体中凡被声明为 Private 的变量只能在某个指定的过程中使用

B）全局变量必须在标准模块中声明

C）模块级变量只能用 Private 关键字声明

D）Static 类型变量的作用域是它所在的窗体或模块文件

【解答】正确答案为 B。

分析：凡被声明为 Private 的变量能被本模块的其他过程使用，但不可以被其他模块使用；模块级变量可以使用 Dim 和 Private 关键字声明；Static 类型的作用域是它所在的过程。

9.4　下列叙述中正确的是（　　）。

A）在窗体的 Form_Load 事件过程中定义的变量是全局变量

B）局部变量的作用域可以超出所定义的过程

C）在某个 Sub 过程中定义的局部变量可以与其他事件过程中定义的局部变量同名，但其作用域只限于该过程

D）在调用过程中，所有局部变量被系统初始化为 0 或空字符串

【解答】正确答案为 C。

分析：模块变量包括窗体变量和标准模块变量。窗体变量可以作用于该窗体的所有过程；全局变量只能在标准模块中声明，不能在过程或窗体模块中声明。

在过程（事件过程或通用过程）内定义的变量叫做局部变量，其作用域是它所在的过程。某一过程的执行只对该过程内的变量产生作用，对其他过程中相同名字的局部变量没有任何影响。因此，在不同的过程中可以定义相同名字的变量，它们之间没有任何关系。如果需要，可以通过“过程名.变量名”的形式分别引用不同过程中相同名字的变量。

在过程中的局部变量，如果过程定义使用了 Static，则过程中的局部变量就是 Static 类型的，即在每次调用过程时，局部变量的值保持不变；如果省略 Static，则局部变量就默认为“自动”的，即在每次调用过程时，局部变量被初始化为 0 或空字符串。

9.5　在窗体上画一个名称为 Command1 的命令按钮，然后编写如下事件过程：

```
Private Sub Command1_Click( )
  Static x As Integer
  Cls
  For i=1 To 2
    y=y+x
    x=x+2
  Next
  Print x , y
End Sub
```

程序运行后，连续 3 次单击 Command1 按钮后，窗体上显示的是（　　）。

A）4　2　　B）12　18　　C）12　30　　D）4　6

【解答】正确答案为 B。

分析：注意此题中 x 被定义为 Static 类型的变量，因此每次单击按钮后，其上次的值被保留下来，而变量 y 每次运算重新归零，单击一次按钮后 x=4, y=2，单击两次按钮后 x=8, y=10，单击 3 次按钮后 x=12, y=18。

9.6　在窗体上画一个命令按钮，下列程序运行后，单击命令按钮，输出结果为（　　）。

```
Private Sub Command3_Click( )
    Tcl 2
    Tcl 3
    Tcl 4
End Sub
Sub Tcl(a As Integer)
    Static x As Integer
    x=x + a
    Print x;
End Sub
```

A）2　3　4　　B）2　5　9　　C）3　5　4　　D）2　4　3

【解答】正确答案为 B。

分析：在过程 Tcl 中，将 Integer 型变量 x 定义为静态变量（Static）。Static 语句的格式与 Dim 语句完全一样，但 Static 语句只能出现在事件过程、Sub 过程或 Function 过程中。在过程中的 Static 变量只有局部的作用域，即只能在本过程中可见，但可以和模块级变量传递参数，即使过程结束后，其值仍能保留。Tcl 过程将变量 a 的值加上 x 赋给 x，然后输出 x 的值。第一次调用 Tcl 过程时，x 未赋值，默认为 0，所以输出结果为 2；第二次调用 Tcl 过程时，因为 x 为静态变量，所以它的值为上次调用后的值，即为 2，加上 a 后，x 的值变为 5，输出结果为 5；同理，第三次调用后输出结果为 9。

9.7　设有如下通用过程：

```
Public Function f(x As Integer)
  Dim y As Integer
  x=20
  y=2
  f=x*y
End Function
```

在窗体上画一个名称为 Command1 的命令按钮，然后编写如下事件过程：

```
Private Sub Command1_Click()
  Static x As Integer
  x=10
  y=5
  y=f(x)
  Print x ; y
End Sub
```

程序运行后，如果单击命令按钮，则在窗体上显示的内容是（　　）。

A）10　5　　B）20　5　　C）20　40　　D）10　40

【解答】正确答案为 C。

分析：此题涉及过程的定义和调用。在调用的过程中应注意参数的传递，如果形参和实参都已赋值，则以形参的值为准。所以，调用了过程之后，x 和 y 的值就成了 20 和 40。答案为选项 C。

二、填空题

9.8　在窗体上画一个名称为 Command1 的命令按钮，然后编写如下事件过程：

```
Private Sub Command1_Click( )
  Static y As Integer
  Cls
  For i=O TO 2
    x=x+y
    y=y+3
  Next
```

```
    Print x , y
End Sub
```

程序运行后，连续单击两次 Command1 按钮后，窗体上显示的是_____。

【解答】正确答案为 36；18。

分析：Static 语句在过程级别中使用，其作用是声明变量并分配存储空间。在整个代码运行期间都能保留使用 Static 语句声明的变量的值。本题中 y 是以 Static 声明的，它的值在离开 Command1_Click 事件后是不会丢失的，而 x 是一个局部变量，它只在过程中有用，离开了过程就不起作用了。单击了一次按钮之后，y=9，单击了两次按钮之后，当 i=0 时，x=x+y=0+9=9，y=y+3=9+3=12；当 i=l 时，x=9+12=21，y=12+3=15；当 i=2 时，x=21+15=36，y=15+3=18。

三、编程题

9.9　编写程序，实现分数化简。要求：在标准模块中编写求最大公约数的函数过程，然后在窗体模块中调用它，来对分数进行化简。

【解答】设计步骤如下。

（1）建立应用程序用户界面并设置对象属性，如图 9-1 所示。

（2）编写代码。

首先向工程中添加标准模块：执行“工程”→“添加模块”菜单命令，打开“添加模块”对话框。选定“新建”选项卡中的“模块”图标，单击“打开”按钮，新建一个标准模块 Module1，并同时打开标准模块窗口。

然后向模块 Module1 中添加一个公有的函数过程 Hcf( )，编写代码如下：

```
Public Function Hcf(m As Long, n As Long)
  If m < n Then
    t = m: m = n: n = t
  End If
  Do While n < > 0
    r = m Mod n
    m = n
    n = r
  Loop
  Hcf = m
End Function
```

最后在窗体模块中编写“化简”按钮 Command1 的 Click 事件代码：

```
Private Sub Command1_Click( )
  Dim n As Long, m As Long
  n = Text1.Text
  m = Text2.Text
  s = Hcf((n), (m))                    ' 调用函数过程
  Text3.Text = n / s
```

```
    Text4.Text = m / s
End Sub
```

执行程序，结果如图 9-2 所示。



图 9-1　分数化简的用户界面和属性



图 9-2　分数化简运行结果

9.10　利用全局变量和函数，设计模拟幸运数字机游戏。设幸运数字为 7，每次由计算机随机产生 3 个 0～10 之间的随机数，当这 3 个随机数中有一个数字为 7 时，就算赢了一次。要求：利用函数过程计算获胜率。

【解答】设计步骤如下。

（1）建立用户界面及设置对象属性，如图 9-3 所示，其中，标签 Label5 用来输出获胜率。



图 9-3　建立用户界面及设置对象属性

（2）编写事件代码。

在标准模块中输入下面的代码，声明全局变量：

```
Public n, wins
```

当程序运行时，程序中的每一事件过程都能访问该变量。

在 Module1 标准模块中声明下面的函数过程：

```
Function rate(hits, att)
    Percent = hits / att                    ' 计算获胜率
    rate = Format(Percent, "0.0%")          ' 返回函数值
End Function
```

编写“生成”按钮 Command1 的 Click 事件代码：

```
Private Sub Command1_Click( )
    Label1.Caption = Int(Rnd * 10)
    Label2.Caption = Int(Rnd * 10)
    Label3.Caption = Int(Rnd * 10)
```

```
    n = n + 1
    ' 如果有一个数字是 7，则响铃
    If (Label1.Caption = 7) Or (Label2.Caption = 7) Or (Label3.Caption = 7) Then
       Beep
       wins = wins + 1
       Label4.Caption = "共产生了" & Str(n) & "次随机数，您赢了" & Str(wins) & "次。"
    Label5.Caption = "获胜率为：" & rate(wins, n)
    End If
End Sub
```

编写“结束”按钮 Command2 的 Click 事件代码：

```
Private Sub Command2_Click( )
    Unload Me
End Sub
```

程序运行结果如图 9-4 所示。



（a）　　（b）

图 9-4　计算“幸运 7”获胜率

第 10 章　用户定义类型与枚举类型

一、选择题

10.1　假定在窗体（名称为 Form1）的代码窗口中定义如下记录类型：

```
Private Type animal
  animalName As String    *20
  aColor As String    *10
End Type
```

在窗体上画一个名称为 Command1 的命令按钮，然后编写如下事件过程：

```
Private Sub Command1_Click( )
  Dim rec As animal
  Open"c:\ vbTest.dat" For Random As#1 Len=Len(rec)
  rec.animalName="Cat"
  rec.aColor ="White"
  Put#1 , , rec
  Close #1
End Sub
```

则以下叙述中正确的是（　　）。

A）记录类型 animal 不能在 Form1 中定义，必须在标准模块中定义

B）如果文件 c:\vbTest.dat 不存在，则 Open 命令执行失败

C）由于 Put 命令中没有指明记录号，因此每次都把记录写到文件的末尾

D）语句“Put #1 , , rec”将 animal 类型的两个数据元素写到文件中

【解答】正确答案为 D。

10.2　表示滚动条控件取值范围最大值的属性是（　　）。

A）Max　　B）LargeChange　　C）Value　　D）Max-Min

【解答】正确答案为 A。

分析：LargeChange 设置滚动条控件的粗调改变值；Value 读取或设置滚动条控件当前的值；Max 设置滚动条的最大值。

10.3　当在滚动条内拖动滚动块时触发

A）KeyUp 事件　　B）KeyPress 事件　　C）Scroll 事件　　D）Change 事件

【解答】正确答案为 C。

分析：本题考查的是滚动条事件的应用。在 VB 中，与滚动条有关的事件是 Scroll 和 Change 事件。当在滚动条内拖动滚动块时触发 Scroll 事件；改变滚动框的位置后，将触发 Change 事件。Scroll 事件用于跟踪滚动条中的动态变化，Change 事件用于得到滚动条的最后的值。

二、填空题

10.4　Visual Basic 对象可以分为两类，分别为____和____。

【解答】正确答案为：预定义对象；用户定义对象。

分析：对象分为两类：预定义对象和用户定义对象。预定义对象是由系统设计好的，可以直接使用或对其进行操作；而用户定义对象中的对象可由程序员自己定义，从而建立自己的对象。

10.5　当滚动条位于最左端或最上端时，Value 属性被设置为____。

【解答】正确答案为：Min。

分析：在一般情况下，垂直滚动条的值由上往下递增，最上端代表最小值，最下端代表最大值；水平滚动条的值从左到右递增，最左端代表最小值，最右端代表最大值。因此当滚动条位于最左端或最下端时，Value 属性被设置为 Min。

三、编程题

10.6　输入学生的姓名、学号、语文成绩、英语成绩、数学成绩，计算每名学生的平均成绩，并显示各科成绩，如图 10-1 所示。



（a）　　（b）

图 10-1　输入学生情况，计算平均成绩、显示各科成绩

【解答】设计方法如下。

（1）在窗体上增加两个框架 Frame1 和 Frame2、3 个命令按钮 Command1～Command3，依次选定 Frame1 和 Frame2，在其中分别增加文本框组 Text1( )、标签组和列表框 List1。如图 10-1 所示。

（2）编写代码。

首先在窗体的通用段创建用户定义类型并声明变量：

```
Private Type studentrec
    na As String * 6                    ' 姓名变量定义为 6 个字符长度
    no As String * 5                    ' 学号变量定义为 5 个字符长度
    ch As Single                        ' 语文变量定义为单精度数
    en As Single                        ' 英语变量定义为单精度数
    ma As Single                        ' 数学变量定义为单精度数
    ag As Single                        ' 平均成绩定义为单精度数
End Type
```

```
Dim stu( ) As studentrec                              ' 定义记录数组
```

编写窗体的 Load 事件代码：

```
Private Sub Form_Load( )
  ReDim stu(0)
End Sub
```

编写“输入”按钮 Command1 的 Click 事件代码：

```
Private Sub Command1_Click( )
  n = UBound(stu)
  ReDim stu(n + 1)
  With stu(n + 1)
    .no = Text1(0).Text
    .na = Text1(1).Text
    .ch = Val(Text1(2).Text)
    .en = Val(Text1(3).Text)
    .ma = Val(Text1(4).Text)
    .ag = Int((.ch + .en + .ma) / 3)                  ' 求个人平均成绩
    cc = Format(.ch, "@@@@@@") & Format(.en, "@@@@@@") & _
                Format(.ma, "@@@@@@") & Format(Str(.ag), "@@@@@@")
    List1.AddItem Format(RTrim(.no), "@@@@@@@") & Format(RTrim(.na), "@@@") & cc
  End With
  Text1(0).SetFocus
End Sub
```

编写“删除”按钮 Command2 的 Click 事件代码：

```
Private Sub Command2_Click( )
  If List1.ListIndex = –1 Then
    MsgBox "请选定欲删除的项！"
    Exit Sub
  End If
  n = List1.ListIndex + 1
  For i = n To UBound(stu) – 1
    stu(i) = stu(i + 1)
  Next
  List1.RemoveItem n – 1                              ' 移除项目
  Text1(0).Text = stu(1).no   :   Text1(1).Text = stu(1).na
  Text1(2).Text = stu(1).ch   :   Text1(3).Text = stu(1).en   :   Text1(4).Text = stu(1).ma
End Sub
```

编写“结束”按钮 Command3 的 Click 事件代码：

```
Private Sub Command3_Click( )
  Unload Me
```

```
End Sub
```

编写列表框 List1 的 Click 事件代码：

```
Private Sub List1_Click( )
    n = List1.ListIndex + 1    :    Text1(0).Text = stu(n).no
    Text1(1).Text = stu(n).na    :    Text1(2).Text = stu(n).ch
    Text1(3).Text = stu(n).en    :    Text1(4).Text = stu(n).ma
End Sub
```

另外编写文本框组的事件代码，使之方便输入：

```
Private Sub Text1_GotFocus(Index As Integer)
    Text1(Index).SelStart = 0
    Text1(Index).SelLength = Len(Text1(Index).Text)
End Sub
Private Sub Text1_KeyPress(Index As Integer, KeyAscii As Integer)
    If KeyAscii = 13 Then
        i = IIf(Index = 4, 0, Index + 1)
        Text1(i).SetFocus
    End If
End Sub
```

10.7　为习题 10.6 增加计算全班平均成绩的功能。

【解答】设计方法如下。

（1）设计程序界面并设置对象属性，如图 10-2 所示。



图 10-2　显示学生成绩

（2）增加命令按钮 Command3 的 Click 事件代码。

```
Private Sub Command3_Click( )
    Dim c As studentrec
    msg = ""
    n = UBound(stu)                          ' 求数组个数
    For i = 1 To n
        c.ch = c.ch + stu(i).ch              ' 求语文平均成绩
        c.en = c.en + stu(i).en              ' 求外语平均成绩
        c.ma = c.ma + stu(i).ma              ' 求数学平均成绩
        c.ag = c.ag + stu(i).ag
    Next i
    b = "语文    外语    数学    平均分" & Chr(13)
    msg = msg & Format(Round(c.ch / n, 1), "####.0") & ", "
    msg = msg & Format(Round(c.en / n, 1), "####.0") & ", "
    msg = msg & Format(Round(c.ma / n, 1), "####.0") & ", "
    msg = msg & Format(Round(c.ag / n, 1), "####.0")
    MsgBox b & msg, 0, "全班平均成绩"
End Sub
```

程序运行结果如图 10-2 所示。

10.8　为习题 10.6 增加按某科成绩排序的功能。

【解答】设计方法如下。

（1）设计程序界面并设置对象属性，如图 10-3 所示。



（a）　　（b）

图 10-3　按单科排序

（2）增加组合框 Combo1 的 Click 事件代码。

```
Private Sub Combo1_Click( )
    i = Combo1.ListIndex
    Dim t As studentrec
    n = UBound(stu)
    Select Case i
      Case 0
        For i = 1 To n – 1
          For j = i + 1 To n
            If stu(i).ch < stu(j).ch Then
              t = stu(i): stu(i) = stu(j): stu(j) = t
            End If
          Next
        Next
      Case 1
        For i = 1 To n – 1
          For j = i + 1 To n
            If stu(i).en < stu(j).en Then
              t = stu(i): stu(i) = stu(j): stu(j) = t
            End If
          Next
        Next
      Case 2
        For i = 1 To n – 1
          For j = i + 1 To n
            If stu(i).ma < stu(j).ma Then
```

```
                t = stu(i): stu(i) = stu(j): stu(j) = t
            End If
        Next
    Next
End Select
Form_Activate
End Sub
```

程序运行结果如图 10-3 所示。由于各科排序的过程完全相同，请读者改为用一个过程实现排序。

10.9　为习题 10.6 增加按姓名查找的功能。

【解答】设计方法如下。

（1）设计程序界面并设置对象属性，如图 10-4 所示。

图 10-4　按姓名查找

（2）编写“查找”按钮 Command3 的 Click 事件代码。

```
Private Sub Command3_Click( )
    Dim c As studentrec
    msg = Trim(InputBox("查找的姓名:", "请输入"))
    If msg = "" Then Exit Sub
    n = UBound(stu)
    f = False
    For i = 1 To n
        If msg = Trim(stu(i).na) Then f = True :    Exit For
    Next i
    If f Then
        Text1(0).Text = stu(i).no    :    Text1(1).Text = stu(i).na
        Text1(2).Text = stu(i).ch    :    Text1(3).Text = stu(i).en    :    Text1(4).Text = stu(i).ma
    Else
        MsgBox "没找到" & msg
    End If
End Sub
```

程序运行结果如图 10-4 所示。

第 11 章　图形与图像

一、选择题

11.1　下面的属性中，用于自动调整图像框中图形内容的大小的是（　　）。

A）Picture　　B）CurrentY　　C）CurrentX　　D）Stretch

【解答】正确答案为 D。

分析：Picture 属性用于窗体、图片框和图像框控件，它可以通过属性窗口进行设置，用来把图形放到这些对象中，因此选项 A 是不正确的；CurrentX 和 CurrentY 用来设置下一个输出的水平坐标和垂直坐标，因此选项 B 和 C 都不正确；Stretch 属性用于图像框控件，用于自动调整图像框中图形内容的大小，既可通过属性窗口设置，也可以通过程序代码设置，该属性的取值为 True 或 False，当其值为 False 时，将自动放大或缩小图像框中的图形使之与图像框的大小相适应，因此选项 D 是正确的。

二、填空题

11.2　为了在运行时把 D:\pic 文件夹下的图形文件 a.jpg 装入图片框 Picture1，所使用的语句为____。

【解答】正确答案为：Picture1.Picture=LoadPicture("D:\\pic\\a.jpg")

或　　Picture1=Load Picture("D:\\pic\\a.jpg")

分析：LoadPicture 函数的功能与 Picture 属性基本相同，即用来把图形文件装入窗体、图片或图像框中，其一般格式为：

［对象.］Picture=LoadPicture(“文件名”)

11.3　下面程序是利用鼠标事件在窗体上画图，如果按下鼠标左键则可以画图，双击窗体可以清除所画图形。补充完整下面的程序。

首先在窗体层定义如下变量：

```
Dim PaintStart As Boolean
```

编写如下事件过程：

```
Private Sub Form_Load( )
    DrawWith=2
    ForeColor=vbGreen
End Sub
Private Sub Form_MouseDown(Button As Integer,Shift As Integer, X As Single,Y As Single)
    ____
End Sub
Private Sub Form_MouseMove(Button As Integer,Shift As Integer, X As Single,Y As Single)
    If PaintStart Then
```

```
        PSet(X,Y)
    End If
End Sub
Private Sub Form_MouseUp(Button As Integer,Shift As Integer, X As Single,Y As Single)
    ____
End Sub
Private Sub Form_DblClick( )
    ____
End Sub
```

【解答】正确答案为：PaintStart = True；PaintStart = False；Cls。

分析：上述过程定义了一个布尔型变量 PaintStart，当按下鼠标左键（触发 MouseDown 事件）时，由题意按下鼠标左键表示画图，所以该变量的值为 True，而松开鼠标左键（触发 MouseUp 事件）时，该变量为 False。如果变量 PaintStart 为 True，则移动鼠标（触发 MouseMove 事件），将在窗体上绘出一个点。除鼠标事件外，上述程序还含有一个 Load 事件过程和一个 DblClick 事件过程，其中 Load 事件过程用来设置画点的大小和颜色，DblClick 事件过程用来清除所画的图形，函数 PSet 是画点语句，用它可以在(x, y)处画一个点。

三、编程题

11.4　编写时钟程序，利用计时器控件来控制指针的转动。

【解答】设计步骤如下。

（1）建立应用程序用户界面。

在窗体上增加一个形状控件 Shape1、一个计时器控件 Timer1，以及表盘上的一些对象：3 个指针（秒针 Line1、分针 Line2 和时针 Line3）和 4 个标签（3 - Label1, 6 - Label2, 9 - Label3, 12 - Labe4）。

最后用一个小的形状控件 Shape2 作为圆盘的中心，如图 11-1（a）所示。



（a）　　（b）

图 11-1　程序界面的设计与程序运行结果

其中，计时器控件 Timer1 可以放在窗体的任何位置。调整 3 个指针的(X2, Y2)坐标（属性），使其位于同一圆心。

（2）设置对象属性，见表 11-1。

（3）编写程序代码。

在窗体的通用段中声明符号常数：

```
Const pi = 3.14159
```

表 11-1　属性设置

对　象	属　性	属 性 值	说　明
Shape1	BackColor	（白色）	表盘
	BackStyle	1-Opaque	背景类型
	BorderWidth	2	线宽
	Shape	2-Oval	形状（椭圆）
Timer1	Interval	100	
Line1	ForeColor	（红色）	秒针
Line2	BorderWidth	2	分针线宽
Line3	BorderWidth	2	时针线宽
Shape2	BackColor	（黑色）	表轴背景色
	BackStyle	1-Opaque	背景类型
	Shape	3-Circle	形状（圆）

编写窗体的 Load 事件代码：

```
Private Sub Form_Load( )
    Line1.Tag = Line1.Y2 – Line1.Y1
    Line2.Tag = Line2.Y2 – Line2.Y1
    Line3.Tag = Line3.Y2 – Line3.Y1
    Form1.Caption = Format(Time, "Medium Time")
    t = Second(Time)
    Line1.X1 = Line1.X2 + Line1.Tag * Sin(pi * t / 30)
    Line1.Y1 = Line1.Y2 – Line1.Tag * Cos(pi * t / 30)
    u = Minute(Time)
    Line3.X1 = Line3.X2 + Line3.Tag * Sin(pi * u / 30)
    Line3.Y1 = Line3.Y2 – Line3.Tag * Cos(pi * u / 30)
    v = Hour(Time)
    s = IIf(v >= 12, v – 12, v) + u / 60
    Line2.X1 = Line2.X2 + Line2.Tag * Sin(pi * s / 6)
    Line2.Y1 = Line2.Y2 – Line2.Tag * Cos(pi * s / 6)
End Sub
```

编写 Timer1 的 Timer 事件代码：

```
Private Sub Timer1_Timer( )
    t = Second(Time)
    Line1.X1 = Line1.X2 + Line1.Tag * Sin(pi * t / 30)
    Line1.Y1 = Line1.Y2 – Line1.Tag * Cos(pi * t / 30)
    If t = 0 Then
        Form1.Caption = Format(Time, "Medium Time")
        u = Minute(Time)
        Line3.X1 = Line3.X2 + Line3.Tag * Sin(pi * u / 30)
        Line3.Y1 = Line3.Y2 – Line3.Tag * Cos(pi * u / 30)
        v = Hour(Time)
```

```
        s = IIf(v >= 12, v – 12, v) + u / 60
        Line2.X1 = Line2.X2 + Line2.Tag * Sin(pi * s / 6)
        Line2.Y1 = Line2.Y2 – Line2.Tag * Cos(pi * s / 6)
    End If
End Sub
```

运行结果如图 11-1（b）所示。

【说明】在窗体的 Load 事件代码中使用了 Line 的 Tag 属性来存放指针的长度。

在 Timer1 控件的 Timer 事件代码中：函数 Second(Time)返回系统时间的秒数，函数 Minute(Time)返回系统时间的分数，函数 Hour(Time)返回系统时间的时数。

下面语句使窗体的标题栏以中文格式显示系统时间（时:分 AM/PM）：

```
Form1.Caption = Format(Time, "Medium Time")
```

还可以改变 Shape1 的 Shape 属性来使时钟的盘面变为圆形、方形或长方形。

11.5　编写程序，实现从蓝色的大圆中分别剪下两个小圆，求大圆剩下部分的面积，如图 11-2 所示。



图 11-2　习题 11.5 的图

【解答】由于本题要多次计算圆面积，所以需要用户定义函数过程，计算圆面积的用户定义函数过程名为 Cir。设计步骤如下。

（1）建立程序界面并设置对象属性。

为使界面形象，用形状控件画出 3 个不同大小的圆形，一个显示信息的标签数组 Label1(0)～Label1(2)，一个命令按钮 Command1。其中各圆的大小不必准确，因为在窗体的代码中将自动进行调整。程序界面如图 11-2 所示。

设置对象属性，见表 11-2。

表 11-2　属性设置

对　象	属　性	属 性 值	说　明
Shape1	BackColor	（蓝色）	大圆
	BackStyle	1-Opaque	背景类型
	Shape	3-Circle	形状（圆）
Shape2	BackColor	（浅绿）	中圆
	BackStyle	1-Opaque	背景类型
	Shape	3-Circle	形状（圆）
Shape3	BackColor	（浅红）	小圆
	BackStyle	1-Opaque	背景类型
	Shape	3-Circle	形状（圆）

（2）编写代码。

首先编写自定义的函数过程 Cir：

```
Private Function Cir(r)
    Const pi = 3.1415
    Cir = pi * r * r
End Function
```

编写命令按钮 Command1 的 Click 事件代码：

```
Private Sub Command1_Click( )
  x1 = Shape1.Width / 2
  x2 = Shape2.Width / 2
  x3 = Shape3.Width / 2
  c = Cir(x1) – Cir(x2) – Cir(x3)
  MsgBox "蓝色部分的面积 = " & c
End Sub
```

使用的坐标单位在窗体事件代码中定义：

```
Private Sub Form_Load( )
  Me.ScaleMode = 6
End Sub
```

为了更加直观，在窗体 Form 的 Resize 事件代码中调整 3 个圆的大小与相互的位置，使得程序运行时，若改变窗体的大小即可相应改变 3 个圆的大小，代码如下：

```
Private Sub Form_Resize( )
  Shape1.Width = IIf(Me.ScaleWidth < Me.ScaleHeight, Me.ScaleWidth, Me.ScaleHeight) * 3 / 4
  Shape1.Height = Shape1.Width
  x = Me.ScaleWidth – Shape1.Width
  y = Me.ScaleHeight – Shape1.Height
  Shape1.Left = x * 3 / 4
  Shape1.Top = y * 3 / 4
  Shape2.Width = Shape1.Width / 2: Shape2.Height = Shape2.Width
  Shape2.Top = Shape1.Top + Shape1.Height / 10: Shape2.Left = Shape1.Left + Shape1.Width / 10
  Shape3.Width = Shape1.Width / 3: Shape3.Height = Shape3.Width
  Shape3.Top = Shape1.Top + Shape1.Height / 2: Shape3.Left = Shape1.Left + Shape1.Width / 2
  Label1(0).Caption = "大圆半径：" & Shape1.Width / 2
  Label1(1).Caption = "中圆半径：" & Shape2.Width / 2
  Label1(2).Caption = "小圆半径：" & Shape3.Width / 2
  Command1.Top = Me.ScaleHeight – Command1.Height – 3
End Sub
```

11.6　编写程序，实现输入两点的坐标，可显示两点的连线并计算两点间的距离，如图 11-3 所示。

【解答】为使界面更加形象，在图片框中画出坐标系，根据输入的坐标自动调整坐标的单位，然后在坐标系中显示两点的连线。设计步骤如下。

（1）建立程序界面并设置对象属性。

在窗体中增加一个图片框 Picture1，两个框架 Frame1 和 Frame2 及一个命令按钮 Command1。在 Frame1 中增加两个文本框控件数组 Text1(0)～Text1(1)，Text2(0)～Text2(1)和一些标签，在 Frame2 中增加一个用于显示计算结果的文本框 Text3。



图 11-3　程序界面

将图片框 Picture1 的 Align 属性设为：3-Align Left，其他属性的设置如图 11-3 所示。

（2）编写代码。

首先在窗体的 Paint 事件代码中画出坐标系，并且调整图片框与其他控件的大小及相互位置，代码如下：

```
Private Sub Form_Paint( )
  Picture1.AutoRedraw = True
  Picture1.Cls
  Picture1.ScaleMode = 6
  x = Picture1.ScaleWidth / 2
  y = Picture1.ScaleHeight / 2
  Picture1.Line (Picture1.ScaleLeft, y) – (Picture1.ScaleWidth, y)
  Picture1.Line (x, Picture1.ScaleTop) – (x, Picture1.ScaleHeight)
  Picture1.Width = Me.Width – Frame1.Width – 200
  Frame1.Left = Picture1.Width + 50
  Frame2.Left = Frame1.Left
  Command1.Left = Frame1.Left + 300
End Sub
```

在窗体的 Resize 事件代码中调用上述代码，以实现窗体与其他控件的协调，代码如下：

```
Private Sub Form_Resize( )
  Form_Paint
End Sub
```

编写命令按钮 Command1 的 Click 事件代码，实现画出两点连线并计算两点之间距离的功能，代码如下：

```
Private Sub Command1_Click( )
  Picture1.AutoRedraw = False
  Picture1.Cls
  Dim x1 As Single, x2 As Single
  Dim y1 As Single, y2 As Single
  x1 = Val(Text1(0).Text): y1 = Val(Text1(1).Text)
  x2 = Val(Text2(0).Text): y2 = Val(Text2(1).Text)
  a = IIf(Abs(x1) < Abs(x2), Abs(x2), Abs(x1))
  b = IIf(Abs(y1) < Abs(y2), Abs(y2), Abs(y1))
  c = IIf(a > b, a, b) + 10
  Picture1.Scale (–c, –c) –(c, c)
  Picture1.Line (x1, –y1) –(x2, –y2), 4
  Picture1.Print "B"
  Text3.Text = Sqr((x1 – x2) ^ 2 + (y1 – y2) ^ 2)
End Sub
```

程序运行结果如图 11-3 所示。用鼠标调整窗体的大小，可以重画坐标系，单击“计算”按钮可以得到两点间的连线及两点之间的距离。

11.7　编写程序，实现在屏幕颜色为 16 位色以上的显示方式下，画一个背景上深下浅的窗体。

【解答】在窗体中增加一个标签 Label1，将 Label1 的 Caption 属性改为上深下浅的窗体背景，并将其 BackStyle 属性改为：0-Transparent，ForeColor 属性改为白色，如图 11-4（a）所示。



（a）　　（b）

图 11-4　上深下浅的窗体背景

编写窗体事件代码为：

```
Private Sub Form_Paint( )
  Dim i As Integer
  ScaleMode = 3
  For i = 0 To 255
  Line (0, i) – (Me.Width, i), RGB(i, i, 255)
  Next i
End Sub
```

程序运行结果如图 11-4（b）所示。

11.8　利用 Circle 方法在窗体上画一个圆桶，如图 11-5 所示。

【解答】利用循环从下到上画一系列的椭圆，最上面的画成实心的即可。窗体的 Paint 事件代码如下：

```
Private Sub Form_Paint( )
  For i = 1000 To 1 Step –1
      Circle (1900, 700 + i), 1000, vbRed, , , 3 / 5
  Next
  Me.FillStyle = 0
  Me.FillColor = RGB(255, 255, 255)
  Circle (1900, 700), 1000, , , , 3 / 5
End Sub
```

11.9　利用 Circle 方法在窗体上画一个有缺口的饼，如图 11-6 所示。



图 11-5　用 Circle 方法在窗体中画一个圆桶



图 11-6　用 Circle 方法在窗体中画一个有缺口的饼

【解答】利用循环从下到上画一系列有缺口的椭圆，最上面的画成实心的即可。窗体的 Paint 事件代码如下：

```
Private Sub Form_Paint( )
  Const pi = 3.14159
  For i = 300 To 1 Step −1
     Circle (1900, 1000 + i), 1000, vbRed, −pi / 3, −pi / 6, 3 / 5
  Next
  Me.FillStyle = 0
  Me.FillColor = RGB(255, 255, 255)
  Circle (1900, 1000), 1000, , −pi / 3,  −pi / 6, 3 / 5
End Sub
```

11.10　编写程序，输入 3 种商品的销售量，可显示销售比例的饼图，如图 11-7 所示。

【解答】首先计算各商品销售量在总量中所占的比例，然后分别按上题画出饼图。设计步骤如下。

（1）设计界面与设置属性。

首先在窗体中增加一个图片框 Picture1、一个框架 Frame1 和一个命令按钮 Command1。然后选中 Frame1，在其中增加一个文本框控件数组 Text1(0)～Text1(2)和一个标签数组。属性的设置如图 11-7 所示。

（2）编写代码。

首先编写画饼图的子过程 hb( )：

```
Private Sub hb(a As Single, b As Single)
  Const pi = 3.14159
  Picture1.FillStyle = 0
  For i = 0 To 200
    Picture1.FillColor = vbRed          ' 用红色填充
    Picture1.Circle (1200, 1200 − i), 800, vbRed, −2 * pi, −2 * pi * a, 2 / 3
    Picture1.FillColor = vbGreen        ' 用绿色填充
    Picture1.Circle (1200, 1200 − i), 800, vbGreen, −2 * pi * a, −2 * pi * (b + a), 2 / 3
    Picture1.FillColor = vbBlue         ' 用蓝色填充
    Picture1.Circle (1200, 1200 − i), 800, vbBlue, −2 * pi * (a + b), −2 * pi, 2 / 3
  Next
  Picture1.FillColor = vbRed            ' 用红色填充
  Picture1.Circle (1200, 1200 − i), 800, vbWhite, −2 * pi, −2 * pi * a, 2 / 3
  Picture1.FillColor = vbGreen          ' 用绿色填充
  Picture1.Circle (1200, 1200 − i), 800, vbWhite, −2 * pi * a, −2 * pi * (b + a), 2 / 3
  Picture1.FillColor = vbBlue           ' 用蓝色填充
  Picture1.Circle (1200, 1200 − i), 800, vbWhite, −2 * pi * (a + b), −2 * pi, 2 / 3
End Sub
```

命令按钮 Click 事件代码：

```
Private Sub Command1_Click( )
    Dim a As Single, b As Single, c As Single
    a = Val(Text1(0).Text)
    b = Val(Text1(1).Text)
    c = Val(Text1(2).Text)
    x = a + b + c
    If c = 0 Then Exit Sub
    a = a / x
    b = b / x
    Call hb(a, b)                    ' 调用画饼图过程
End Sub
```

11.11　在窗体上画五角星，如图 11-8 所示。



图 11-7　显示销售比例的饼图



图 11-8　在窗体上画五角星

【解答】以五角星的左下角为起点，依次向上、右下、左上、右上画线，最后回到左下角。注意：每个角的角度值转换成弧度值后为 2π/10。利用三角函数可以得到顶点的相对坐标。

将画五角星的过程定义为子过程，可以反复调用。

首先在窗体的通用段声明符号常量：

```
Const pi = 3.14159
```

自定义的画五角星子过程代码：

```
Private Sub star(X As Single)
    Randomize
    n = Int(Rnd * 16)
    colr = QBColor(n)
    Line -Step(X * Sin(pi / 10), -X * Cos(pi / 10)), colr
    Line -Step(X * Sin(pi / 10), X * Cos(pi / 10)), colr
    Line -Step(-X * Cos(2 * pi / 10), -X * Sin(2 * pi / 10)), colr
    Line -Step(X, 0), colr
    Line -Step(-X * Cos(2 * pi / 10), X * Sin(2 * pi / 10)), colr
End Sub
```

窗体事件代码：

```
Private Sub Form_MouseUp(Button As Integer, Shift As Integer, X As Single, Y As Single)
    PSet (Rnd * 3000, Rnd * 3000)
    star (Rnd * 2000)
End Sub
```

运行程序，单击窗体可以得到随机大小、颜色的五角星，如图 11-8 所示。

11.12　编写程序，实现输入长方体的长、宽、高，可求出其表面积。

【解答】设长方体的长、宽、高分别为 a, b, c，表面积为 S，根据数学知识，求长方体表面积的公式为：$S = 2(ab+bc+ca)$。利用输入的边长，可以画出长方体，设计步骤如下。



图 11-9　程序界面的设计

（1）建立程序界面并设置对象属性。

在窗体中增加一个图片框 Picture1，一个框架 Frame1。在 Frame1 中增加一个文本框控件数组 Text1(0)～Text1(1)，一个命令按钮 Command1 和一些标签。

将图片框 Picture1 的 Align 属性设为：3-Align Right，其他属性的设置如图 11-9 所示。

（2）编写代码。

命令按钮 Command1 的 Click 事件代码：

```
Private Sub Command1_Click( )
    Picture1.ScaleMode = 6
    Dim a As Single, b As Single, c As Single
    a = Val(Text1(0).Text)
    b = Val(Text1(1).Text)
    c = Val(Text1(2).Text)
    x = a + b + c
    If c = 0 Then Exit Sub
    Call ht(a, b, c)                          ' 调用画图过程
    m = 2 * (a * b + b * c + c * a)
    Label2.Caption = "表面积 = " & Chr(13) & m
End Sub
```

其中调用了画图子过程 ht( )：

```
Private Sub ht(a As Single, b As Single, c As Single)
    Picture1.Cls
    x = (IIf(a < b, b, a) + 0.5 * c) * 1.5        ' 求坐标的最大值
    Picture1.Scale (0, 0) –(x, x)                 ' 设置坐标系
    x0 = x / 6: y0 = x * 5 / 6                    ' 长方体左下角坐标
    Picture1.Line (x0, y0) –Step(a, –b), , B
    Picture1.Line (x0, y0 – b)–Step(1.414 * c / 3, –1.414 * c / 3)
    Picture1.Line –Step(a, 0)
    Picture1.Line –Step(–1.414 * c / 3, 1.414 * c / 3)
    Picture1.Line (x0 + a, y0) –Step(1.414 * c / 3, –1.414 * c / 3)
    Picture1.Line –Step(0, –b)
    With Picture1                                 ' 设置字母"a"的坐标
        .CurrentX = x0 + a / 2
        .CurrentY = y0
```

```
    End With
    Picture1.Print "a"
    With Picture1                                    ' 设置字母"b"的坐标
       .CurrentX = x0 + a + Picture1.TextWidth("b")
       .CurrentY = y0 – b / 2 – Picture1.TextHeight("b")
    End With
    Picture1.Print "b"
    With Picture1                                    ' 设置字母"c"的坐标
       .CurrentX = x0 + 1.414 * c / 6 + Picture1.TextWidth("c")
       .CurrentY = y0 – b – 1.414 * c / 6 – Picture1.TextHeight("c") / 2
    End With
    Picture1.Print "c"
End Sub
```

运行程序，输入任意三边的长度，都可以在图片框中画出长方体的图形，并求出长方体的表面积，如图 11-9 所示。

11.13　编写程序，运行时在图片框中画出坐标系统刻度，在窗体上单击鼠标右键，可在弹出的快捷菜单中选择三角函数名称，可以画出相应的图形。

【解答】设计步骤如下。

（1）首先在窗体上增加一个图片框 Picture1，将其 Align 属性改为：1 - Align Top。然后打开菜单设计器设计菜单，参数设置见表 11-3。

表 11-3　菜单项的设置

标题（Caption）	名称（Name）	索引（Index）	说　明
三角函数	sanjiao		主菜单项
….Sin(x)	hs	0	子菜单项 1
….Cos(x)	hs	1	子菜单项 2
….清除	hs	2	子菜单项 3

将主菜单项“三角函数”的“可见”复选框取消，即将 Visible 属性改为 False。如图 11-10（a）所示。



（a）　　（b）

图 11-10　画出三角函数

（2）编写程序代码。

首先在窗体的通用段声明常量：

```
Const pi = 3.14159
```

然后编写窗体的 Paint 事件代码：

```
Private Sub Form_Paint( )
  With Picture1
    .Height = Me.ScaleHeight
    .ScaleMode = 6
    oldx = .ScaleWidth / 2
    oldy = .ScaleHeight / 2
    .Cls
    Picture1.Line (oldx, .Top) –(oldx, .ScaleHeight), RGB(255, 0, 0)        ' 画坐标轴
    Picture1.Line (0, oldy) –(.ScaleWidth, oldy), RGB(255, 0, 0)
  End With
  Picture1.CurrentX = oldx – 4: Picture1.CurrentY = oldy + 0.5
  Picture1.Print 0                                                          ' 输出坐标原点 0
  For xt = –Int(oldx) To Int(oldx) Step 0.5
    If xt <> 0 Then
      st = xt * 10 * pi
      Picture1.CurrentX = oldx + st – 3: Picture1.CurrentY = oldy + 0.5
      Picture1.Print xt & " π "                                             ' 画 x 轴的刻度
      Picture1.Line (oldx + st, oldy – 1) –(oldx + st, oldy), RGB(255, 0, 0)
    End If
  Next xt
  For yt = –5 To 7
    If yt < > 0 Then
      st = yt * 10
      Picture1.CurrentX = oldx – 4: Picture1.CurrentY = oldy + st – 1
      Picture1.Print yt                                                     ' 画 y 轴的刻度
      Picture1.Line (oldx, oldy + st) –(oldx + 1, oldy + st), RGB(255, 0, 0)
    End If
  Next yt
End Sub
```

在窗体的 Resize 事件代码中调用上述代码，实现改变窗体大小时重画坐标系：

```
Private Sub Form_Resize( )
  Form_Paint
End Sub
```

编写快捷菜单过程代码：

```
Private Sub hs_Click(Index As Integer)
  oldx = Picture1.ScaleWidth / 2
  oldy = Picture1.ScaleHeight / 2
  Select Case Index
```

```
        Case 0
            For t = – oldx To oldx Step 0.01
                xt = 10 * t
                yt = 10 * Sin(t)
                Picture1.PSet (xt + oldx, oldy – yt), RGB(0, 127, 127)
            Next
        Case 1
            For t = – oldx To oldx Step 0.01
                xt = 10 * t
                yt = 10 * Cos(t)
                Picture1.PSet (xt + oldx, oldy – yt), RGB(0, 127, 127)
            Next
        Case 2
            Picture1.Cls
            Form_Paint
            Exit Sub
    End Select
End Sub
```

编写图片框的鼠标事件代码调用快捷菜单：

```
Private Sub Picture1_MouseDown(Button As Integer, Shift As Integer, X As Single, Y As Single)
    If Button = 2 Then
        PopupMenu sanjiao, 6
    End If
End Sub
```

程序运行结果如图 11-10（b）所示。

11.14　编写华氏温度和摄氏温度相互转换的程序，利用文本框交互实现输入和输出。当鼠标指向命令按钮时动态地改变手形状图片的显示，如图 11-11 所示。



（a）　　（b）

图 11-11　动态地改变图片的显示

【解答】设计步骤如下。

（1）建立应用程序用户界面并设置对象属性。

在窗体上，增加两个标签 Label1 和 Label2，两个文本框 Text1 和 Text2，两个命令按钮

Command1 和 Command3 及一个图片框 Picture1。

在属性窗口中设置对象属性，见表 11-4。

表 11-4　属性设置

对　象	属　性	属 性 值	说　明
Label1	Caption	摄氏温度	
Label2	Caption	华氏温度	
Text1	Text		清空
Text2	Text		清空
Command1	Caption	摄氏转华氏	
Command2	Caption	华氏转摄氏	
Picture1	AutoSize	True	
	BorderStyle	0-None	无边框

另外，修改图片框 Picture1 的 Picture 属性（路径与文件名）为：

\program files\microsoft visual studio\common\graphics\icons\arrows\point04.ico

（2）编写程序代码。

编写命令按钮 Command1 的事件代码。

Click 事件：

```
Private Sub Command1_Click( )
   Text2.Text = Format(Text1.Text * (9 / 5) + 32, "##0.0##")
End Sub
```

MouseMove 事件：

```
Private Sub Command1_MouseMove(Button As Integer,Shift As Integer,X As Single,Y As Single)
   Picture1.Picture = LoadPicture("c:\program files\microsoft visual studio\ _
                      common\graphics\icons\arrows\point04.ico")
   Text1.SelStart = 0
   Text1.SelLength = Len(Text1.Text)
   Text1.SetFocus
End Sub
```

编写命令按钮 Command2 的事件代码。

Click 事件：

```
Private Sub Command2_Click( )
   Text1.Text = Format((Text2.Text - 32) * (5 / 9), "##0.0##")
End Sub
```

MouseMove 事件：

```
Private Sub Command2_MouseMove(Button As Integer,Shift As Integer,X As Single,Y As Single)
   Picture1.Picture = LoadPicture("c:\program files\microsoft visual studio\ _
                      common\graphics\icons\arrows\point02.ico")
   Text2.SelStart = 0
   Text2.SelLength = Len(Text2.Text)
```

```
    Text2.SetFocus
End Sub
```

【说明】MouseMove 事件在鼠标移动并经过对象（按钮）时发生（参见第 12 章）。本题中，当鼠标指向左边按钮时，手形图片指向右边，如图 11-11（a）所示；当鼠标指向右边按钮时，手形图片则指向左边，如图 11-11（b）所示。

11.15　编写程序，实现利用滚动条控制图形的大小。

【解答】设计步骤如下。

（1）建立应用程序用户界面并设置对象属性。

首先增加一个框架 Frame1，激活 Frame1 后，在其中增加一个图像框 Image1，一个水平滚动条 HScrollBars1 和一个垂直滚动条 VScrollBars1，如图 11-12（a）所示。

在属性窗口中设置对象属性，见表 11-5。



（a）　　（b）

图 11-12　利用滚动条控制图形的大小

表 11-5　属性设置

对　象	属　性	属 性 值	说　明
Frame1	Caption	单击滚动条，可改变图形的大小	
Image1	Picture	（自选）	
	Stretch	True	根据 Image 控件的大小调整图片
HScrollBars1	Max	3000	
	Min	480	
	LargeChange	200	
	SmallChange	20	
	Value	480	
VScrollBars1	Max	1800	
	Min	480	
	LargeChange	200	
	SmallChange	20	
	Value	480	

（2）编写程序代码。

水平滚动条 HScrollBars1 的 Change 事件代码：

```
Private Sub HScroll1_Change( )
    Image1.Width = HScroll1.Value
End Sub
```

垂直滚动条 VScrollBars1 的 Change 事件代码：

```
Private Sub VScroll1_Change( )
    Image1.Height = VScroll1.Value
End Sub
```

执行程序，拖动滚动条，效果如图 11-12（b）所示。

11.16　编写程序，实现利用滚动条控制色彩，还可以返回色彩的 RGB 值。

【解答】本程序可实现的功能是：直接修改文本框中的 RGB 设置，可以得到相应的颜色；单击滚动条也可得到所需的颜色，并可返回相应的 RGB 设置，如图 11-13 所示。

设计步骤如下。

（1）建立应用程序用户界面并设置对象属性。



（a）　　（b）

图 11-13　调色盘

首先增加一个框架 Frame1，激活 Frame1 后，在其中增加一个图片框 Picture1，一个文本框 Text1，3 个垂直滚动条控件 VScroll1～VScroll3 和 3 个标签控件 Label1～Label3，并设置属性，见表 11-6。

表 11-6　属性设置

对　象	属　性	属 性 值	说　明
VScroll1~VScroll3	LargeChange	32	最大改变值
	SmallChange	4	最小改变值
	Max	0	
	Min	255	
	Value	255	当前值
Frame1	Caption		
Label1~Label3	BackColor	依次为：红、绿、蓝	
	BorderStyle	1-Fixed Single	
	Caption		
Text1	Text	Color = RGB(255,255,255)	
Command1	Caption	确定	
	Default	True	

（2）编写程序代码。

垂直滚动条 VScroll1 的 Change 事件代码：

```
Private Sub VScroll1_Change( )
    Picture1.BackColor = RGB(VScroll1, VScroll2, VScroll3)
```

```
    r = LTrim(Str(VScroll1))
    g = LTrim(Str(VScroll2))
    b = LTrim(Str(VScroll3))
    Text1.Text = "Color=RGB(" & r & "," & g & "," & b & ")"
End Sub
```

垂直滚动条 VScroll2 的 Change 事件代码：

```
Private Sub VScroll2_Change( )
    Picture1.BackColor = RGB(VScroll1, VScroll2, VScroll3)
    r = LTrim(Str(VScroll1))
    g = LTrim(Str(VScroll2))
    b = LTrim(Str(VScroll3))
    Text1.Text = "Color=RGB(" & r & "," & g & "," & b & ")"
End Sub
```

垂直滚动条 VScroll3 的 Change 事件代码：

```
Private Sub VScroll3_Change( )
    Picture1.BackColor = RGB(VScroll1, VScroll2, VScroll3)
    r = LTrim(Str(VScroll1))
    g = LTrim(Str(VScroll2))
    b = LTrim(Str(VScroll3))
    Text1.Text = "Color=RGB(" & r & "," & g & "," & b & ")"
End Sub
```

文本框 Text1 的 GotFocus 事件代码：

```
Private Sub Text1_GotFocus( )
    Text1.SelStart = 10
End Sub
```

命令按钮 Command1 的 Click 事件代码：

```
Private Sub Command1_Click( )
    a = InStr(10, Text1.Text, ",")
    b = InStr(a + 1, Text1.Text, ",")
    c = InStr(b + 1, Text1.Text, ")")
    VScroll1 = Val(Mid(Text1.Text, 11, a - 10))
    VScroll2 = Val(Mid(Text1.Text, a + 1, b - a))
    VScroll3 = Val(Mid(Text1.Text, b + 1, c - b - 1))
End Sub
```

【说明】函数 InStr (n,〈字符串 1〉,〈字符串 2〉)从〈字符串 1〉的第 n 个位置开始查找〈字符串 2〉首次出现的位置，并返回一个整数。

控件也有默认值，大多数的控件都默认为其 Value 属性。因此，

```
VScroll1(0) = Val(Mid(Text1.Text, 11, a - 10))
```

相当于：

```
VScroll1(0).Value = Val(Mid(Text1.Text, 11, a - 10))
```

第 12 章　菜单、工具栏与对话框

一、选择题

12.1　下列说法正确的是（　　）。

A）任何时候都可以使用标准工具栏的“菜单编辑器”按钮打开菜单编辑器

B）只有当代码窗口为当前活动窗口时，才能打开菜单编辑器

C）只有当某个窗体为当前活动窗体时，才能打开菜单编辑器

D）任何时候都可以使用“工具”→“菜单编辑器”菜单命令，打开菜单编辑器

【解答】正确答案为 C。

分析：在 VB 中，因为“菜单编辑器”命令是在“工具”菜单中的，所以只有当某个窗体为当前活动窗体时，才能从“工具”菜单中选择“菜单编辑器”选项，打开菜单编辑器。

12.2　下例叙述错误的是（　　）。

A）下拉式菜单和弹出式菜单都用菜单编辑器建立

B）在多窗体程序中，每个窗体都可以建立自己的菜单系统

C）除分隔线外，所有菜单项都能接受 Click 事件

D）如果把一个菜单项的 Enabled 属性设置为 False，则该菜单项不可见

【解答】正确答案为 D。

分析：Enabled 属性用于决定菜单项是否可用，Visual 属性决定菜单项是否可见。

12.3　设菜单中有一个菜单项为“Open”。若要为该菜单命令设置访问键，即按下 Alt 键及字母 O 时能够执行“Open”命令，则在菜单编辑器中设置“Open”命令的方式是（　　）。

A）把 Caption 属性设置为&Open　　B）把 Caption 属性设置为 O&pen

C）把 Name 属性设置为&Open　　D）把 Name 属性设置为 O&pen

【解答】正确答案为 A。

分析：设置菜单中的标题文本使用 Caption 属性，而 Name 属性是从代码中访问菜单时使用的，因此排除选项 C 和选项 D；设置访问键时在作为访问键的字符前面输入一个&号，就允许用户通过键盘操作菜单项。

12.4　设在菜单编辑器中定义了一个菜单项，名为 menu1。为了在运行时隐藏该菜单项，应该使用的语句是（　　）。

A）menu1.Enabled=True　　B）menu1.Enabled=False

C）menu1.Visible=True　　D）menu1.Visible=False

【解答】正确答案为 D。

分析：Visible 属性可控制控件可见或不可见。

12.5　要使菜单项 MenuOne 在程序运行时失效，使用的语句是（　　）。

A）MenuOne.Visible = True　　B）MenuOne.Visible = False

C）MenuOne.Enabled = True　　D）MenuOne.Enabled = False

【解答】正确答案为 D。

分析：菜单控件的主要属性有：Caption、Name、Index、Enabled、Visible 等，其中 Enabled 和 Visible 属性值为逻辑值，Visible 属性用于控制菜单项是否可见，Enabled 属性用于控制菜单项是否可用。当 Enabled 属性值为 False 时，表示菜单项当前不可用；属性值为 True 时，表示菜单项可用。

12.6　以下关于菜单的叙述中，错误的是（　　）。

A）在程序运行过程中可以增加或减少菜单项

B）如果把一个菜单项的 Enabled 属性设置为 False，则可删除该菜单项

C）弹出式菜单在菜单编辑器中设计

D）利用控件数组可以实现菜单项的增加或减少

【解答】正确答案为 B。

分析：Enabled 属性返回或设置一个值，用来确定一个窗体和控件是否能够对用户产生的事件做出反应。

12.7　下列有关子菜单的说法中，错误的是（　　）。

A）除了 Click 事件之外，菜单项不可以响应其他事件

B）每个菜单项都是一个控件，与其他控件一样也有其属性和事件

C）菜单项的索引号必须从 1 开始

D）菜单的索引号可以不连续

【解答】正确答案为 C。

分析：程序运行后，当用户选择某个菜单标题时会打开下拉菜单，菜单中的菜单项可以是命令、选项、分隔条或子菜单标题，每个菜单项都是一个控件，与其他控件一样也有自己的属性和事件，菜单项的各个属性都能设置和查看，如 Name（名称）和 Caption（标题）属性等，每个菜单项只能响应一个事件，即 Click 事件。菜单项的索引号表示菜单数组中的位置序号。如果不定义菜单数组，可以不理会。其实，菜单的索引号可以不连续，也没有限制必须从 1 开始，所以只有选项 C 是错误的。

12.8　在窗体上建立通用对话框需要添加的控件是（　　）。

A）Data 控件　　B）From 控件

C）CommonDialog 控件　　D）VBComboBox 控件

【解答】正确答案为 C。

分析：通用对话框是一种 ActiveX 控件，它随同 VB 提供给程序设计人员。在一般情况下，启动 VB 后，在工具箱中没有通用对话框控件。为了把通用对话框添加到工具箱中，可以在“部件”对话框的“控件”选项卡的控件列表框中选择“Microsoft Common Dialog Control 6.0”项。

12.9　在窗体上画一个通用对话框，其名称为 CommonDialog1，然后画一个命令按钮，并编写如下事件过程：

```
Private SUb Command1_Click()
  CommonDialog1.Flags=cdlOFNHideReadOnly
```

```
    CommonDialog1.Filter="All Files(*.*)|  *.* | Text Files"& "(*.txt) |*.txt|Batch Files(*.bat) |*.bat"
    CommonDialog1.FilterIndex=2
    CommonDialog1.ShowOpen
    MsgBox CommonDialog1.Filename
End Sub
```

程序运行后，单击命令按钮，将显示一个“打开”对话框，此时在“文件类型”框中显示的是（　　）。

A）All Files(*.*)　　B）Text Files(*.txt)

C）Batch Files(*.Bat)　　D）不确定

【解答】正确答案为 B。

分析：从程序可见，FilterIndex 属性已经设为 2，所以默认过滤器为第 2 个类型，即 Text Files(*.txt)。

12.10　下列程序的功能是调用“字体”对话框来设置文本框字体，单击按钮弹出对话框后，按 Cancel 键退出对话框，则下列说法正确的是（　　）。

```
Private Sub Command1_Click()
    CommonDialog1.CancelError=true
    CommonDialog1.flags=cdlCFEffects Or cdlDFBoth
    CommonDialog1.Action=4
    CommonDialog1.ShowFont
    Text1.Font.Name=CommonDialog1.FontName
    Text1.Font.Size=CommonDialog1.fontSize
    Text1.Font.Bold=CommonDialog1.FontBold
    Text1.Font.Italic=CommonDialog1.FontItalic
    Text1.Font.Underline=CommonDialog1.FontUnderline
    Text1.FontStrikethru=CommonDialog1.FontStrikethru
    Text1.ForeColor=CommonDialog1.Color
End Sub
```

A）Text1 的字体不发生变化

B）Text1 的字体发生变化

C）Text1 的字体和颜色发生变化

D）程序出错！

【解答】正确答案为 D。

分析：对话框的 Cancel Error 属性的作用是指示当单击“取消”按钮时是否出错，本题中该属性为 True，因此按 Cancel 键后将导致程序出错。一般应该在程序中添加错误的处理。

12.11　以下事件过程可以将“打开”对话框的标题改为“新时代”的是（　　）。

A）
```
Private Sub Command2_Click()
    CommonDialog1.DialogTitle ="新时代"
    CommonDialog1.ShowOpen
```

```
    End Sub
```
B）
```
Private Sub Command2_Click()
    CommonDialog1.DialogTitle ="新时代"
    CommonDialog1.ShowFont
End Sub
```
C）
```
Private Sub Command2_Click()
    CommonDialog1.DialogTitle ="新时代"
    CommonDialog1.Show
End Sub
```
D）
```
Private Sub Command2_Click()
    CommonDialog1.DialogTitle ="新时代"
    CommonDialog1.ShowColor
End Sub
```
【解答】正确答案为 A。

分析：首先，设置标题属性用 DialogTitle 属性，4 个选项都一样，所以这个不是区别。再看方法，选项 A 用的是 ShowOpen 方法，即“打开”对话框，所以是正确的；选项 B 用的是 ShowFont 方法，是“字体”对话框，不符合本题的意思，不正确；选项 C 用的是 Show 方法，但是对话框控件没有此方法，所以也不正确；最后选项 D 用的是 ShowColor 方法，是“颜色”对话框，所以也不符合本题的意思。

12.12　下列说法错误的是（　　）。

A）文件对话框可分为两种，即“打开”（Open）对话框和“另存为”（Save As）对话框

B）通用对话框的 Name 属性默认值为 CommonDialogX，此外，每种对话框都有自己的默认标题

C）“打开”对话框可以让用户指定一个文件，由程序使用；而用“另存为”对话框可以指定一个文件，并以这个文件名保存当前文件

D）DefaultEXT 属性和 DialogTitle 属性都是“打开”对话框的属性，不是“另存为”对话框的属性

【解答】正确答案为 D。

分析：文件对话框分为两种：“打开”对话框和“另存为”对话框，所以选项 A 说法正确。通用对话框的 Name 属性的默认值为 CommonDialogX，此外，每种对话框都有自己的默认标题，所以选项 B 说法正确。“打开”对话框可以让用户指定一个文件，由程序使用；而用“另存为”对话框可以指定一个文件，并以这个文件名保存当前文件，所以选项 C 说法正确。除 DefaultEXT、DialogTitle 属性是“打开”对话框和“另存为”对话框共有的外，还有 FileName、FileTitle、Filter、FilterIndex、Flags、InitDir、MaxFileSize、CancelError、HelpCommand、HelpContext 和 HelpFilter 属性，都是它们共有的，所以选项 D 不正确。

12.13　在窗体上画一个名称为 CommonDialog1 的通用对话框，一个名称为 Command1 的命令按钮，要求单击命令按钮时，打开一个“另存为”对话框，该窗口标题为“Save”，默认文件名称为“SaveFile”，在“文件类型”框中显示*.txt，则能够满足上述要

求的程序是（　　）。

A）Private Sub Command1_Click()

```
    CommonDialog1.FileName="SaveFile"
    CommonDialog1.Filter="AllFiles *.* (*.txt) *.txt (*.doc) *.doc"
    CommonDialog1.FilterIndex=2
    CommonDialog1.DialogTitle="Save"
    CommonDialog1.Action=2
End Sub
```

B）Private Sub Command1_Click()

```
    CommonDialog1.FileName="SaveFile"
    CommonDialog1.Filter="AllFiles *.* (*.txt) *.txt (*.doc) *.doc"
    CommonDialog1.FilterIndex=1
    CommonDialog1.DialogTitle="Save"
    CommonDialog1.Action=2
End Sub
```

C）Private Sub Command1_Click()

```
    CommonDialog1.FileName="Save"
    CommonDialog1.Filter="AllFiles *.* (*.txt) *.txt (*.doc) *.doc"
    CommonDialog1.FilterIndex=2
    CommonDialog1.DialogTitle="SaveFile"
    CommonDialog1.Action=2
End Sub
```

D）Private Sub Command1_Click()

```
    CommonDialog1.FileName="SaveFile"
    CommonDialog1.Filter="AllFiles *.* (*.txt) *.txt (*.doc) *.doc"
    CommonDialog1.FilterIndex=1
    CommonDialog1.DialogTitle="Save"
    CommonDialog1.Action=1
End Sub
```

【解答】正确答案为 A。

分析：CommonDialog 控件的 FileName 属性用来设置默认文件名；Filter 属性用来设置在对话框的文件类型列表框中所显示的过滤器，其值从 0 开始；DialogTitle 属性用来设置对话框的标题；Action 属性用来返回或设置被显示对话框的类型，其值为 1 时显示“打开”对话框，为 2 时显示“另存为”对话框。

二、填空题

12.14　菜单编辑器可分为 3 个部分，即菜单属性设置区、____和菜单项显示区。

【解答】正确答案为：编辑区。

分析：菜单编辑器的界面分为菜单属性设置区、编辑区和菜单项显示区 3 个部分。菜单

属性设置区用来输入菜单的名称、索引和标题等信息。编辑区用来对当前的菜单项进行插入、删除等操作。菜单项显示区用来显示当前对菜单操作的结果，以便用户和自己的创建目标进行比照。

12.15　在菜单编辑器中建立一个菜单，其主菜单项的名称为 mnuEdit，Visible 属性为 False。程序运行后，如果用鼠标右键单击窗体，则弹出与 mnuEdit 对应的菜单，以下是实现上述功能的程序，请补全程序。

```
Private Sub Form_ _____(Button As Integer,Shift As Integer,X As Single,Y As Single)
  If Button=2 Then
    ____  mnuEdit
  End If
End Sub
```

【解答】正确答案为：MouseDown；PopupMenu。

12.16　在菜单编辑器中建立了一个菜单，名为 pmenu，用下面的语句可以把它作为弹出式菜单弹出，请补全程序。

```
Form1._____  pmenu
```

【解答】正确答案为：PopupMenu。

分析：Form1.PopupMenu pmenu 可以弹出菜单。

12.17　在窗体上加上一个文本控件 PCSTextBox，画一个命令按钮，当单击命令按钮的时候将显示“打开”对话框。设置该对话框，使之只用于打开文本文件，然后在文本控件中显示打开的文件名。请补全程序。

```
Private Sub Command1_Click()
  CommonDialog1.Filter =  _____
  CommonDialog1.ShowOpen
  PCSTextBox.Text =  _____
End Sub
```

【解答】正确答案为："Text Files(*.txt)|*.txt"；CommonDialog1.FileName。

分析：要设置通用对话框只用于打开文本文件，需要设置通用对话框的 Filter 属性值，在这里应该填写的是"Text Files(*.txt)|*.txt"；在使用了对话框控件的 ShowOpen 方法后出现的“打开”对话框中的文件类型只有 Text Files(*.txt)|*.txt，选择了需要打开的文本文件后，应该让文件的文件名显示在文本框中，因此需要将通用对话框中选择的文件赋给文本控件，即将对话框的 FileName 属性值赋给文本框的 Text 属性。

三、编程题

12.18　设计如图 12-1 所示的菜单程序。“程序”菜单中包含有“Word”、“Excel”和“PowerPoint”3 个选项，“附件”菜单中包含有“画图”和“游戏”两个选项，而“游戏”子菜单中又包含有“纸牌”和“扫雷”两个选项，“退出”菜单中包含有“关闭计算机”和“重新启动”两个选项。当用户选择了“程序”或“附件”中的某一选项时，应能启动相应的程序。当用户选择了“退出”菜单中的某一选项时，将弹出确认对话框，用户确认后将执行相应的操作。



（a）　（b）

（c）　（d）

图 12-1　使用菜单的小程序

【解答】设计步骤如下。

（1）在菜单编辑器中设计菜单选项，菜单中各选项的属性设置参见表 12-1。

表 12-1　菜单选项属性设置

标题（Caption）	名称（Name）	说　明	标题（Caption）	名称（Name）	说　明
文件	File	主菜单项 1	….游戏	Youxi	子菜单 22
….Word	Word	子菜单 11	……纸牌	Zhipai	子子菜单 221
….Excel	Excel	子菜单 12	……扫雷	Saolei	子子菜单 222
….PowerPoint	PowerPoint	子菜单 13	退出	Quit	主菜单 3
附件	Fujian	主菜单 2	….关闭计算机	Guanji	子菜单 31
….画图	Huatu	子菜单 21	….重新启动	Chongqi	子菜单 32

（2）编写程序代码。

```
Private Declare Function ExitWindowsEx Lib "user32" (ByVal uFlags As Long,_
                                                     ByVal dwReserved As Long) As Long
Const EWX_LOGOFF = 0
Const EWX_SHUTDOWN = 1
Const EWX_REBOOT = 2
Const EWX_FORCE = 4
Const EWX_POWEROFF = 8
Const EWX_RESET = EWX_LOGOFF + EWX_FORCE + EWX_REBOOT
Private Sub Chongqi_Click( )
  a = MsgBox("重启计算机将丢失数据！你的程序存盘了吗？", 1 + 48, "警告")
  If a = vbOK Then X = ExitWindowsEx(EWX_RESET, dwReserved)
End Sub
```

```
Private Sub Excel_Click( )
   Shell ("C:\Program Files\Microsoft Office\Office\Excel.exe"), _
   vbNormalFocus
End Sub
Private Sub Guanji_Click( )
   b = MsgBox("关闭计算机将丢失数据！你的程序存盘了吗？", 1 + 48, "警告")
   If b = vbOK Then X = ExitWindowsEx(EWX_SHUTDOWN, dwReserved)
End Sub
Private Sub Huatu_Click( )
   Shell ("C:\Program Files\Accessories\Mspaint.exe"), vbNormalFocus
End Sub
Private Sub PowerPoint_Click( )
   Shell ("C:\Program Files\Microsoft Office\Office\Powerpnt.exe"), vbNormalFocus
End Sub
Private Sub Saolei_Click( )
   Shell ("C:\WINDOWS\Winmine.exe"), vbNormalFocus
End Sub
Private Sub Word_Click( )
   Shell ("C:\Program Files\Microsoft Office\Office\Winword.exe"), vbNormalFocus
End Sub
Private Sub Zhipai_Click( )
   Shell ("C:\WINDOWS\Sol.exe"), vbNormalFocus
End Sub
```

12.19　编写程序设计个人信息查询工具。要求使用菜单来控制数据的编辑、查找，可完成按姓名查找、按工资范围查找、按工作证号查找的功能。查找时使用模糊查找，将满足条件的记录显示在列表框中。

【解答】设计步骤如下。

（1）建立用户界面并设置对象属性，如图 12-2 所示。



（a）　　（b）

图 12-2　程序界面的设计与运行结果

（2）设计菜单。在菜单设计器中按表 12-2 设计菜单项。

表 12-2　设计菜单项

标题（Caption）	名称（Name）	索引（Index）	说　　明
编辑	Edit		主菜单项 1
….增加	Edt	0	二级子菜单项 11
….删除	Edt	1	二级子菜单项 12
….更新	Edt	2	二级子菜单项 13
查找	Find		主菜单项 2
….按姓名查找	Fnd	0	二级子菜单项 21
….按工资范围查找	Fnd	1	二级子菜单项 22
….按工作证号查找	Fnd	2	二级子菜单项 23

（3）编写代码。下面只给出菜单项的代码，其他代码请读者自己编写。

“编辑”菜单控件数组 Edt 的 Click 事件代码：

```
Private Sub Edt_Click(Index As Integer)
  Select Case Index
    Case 0
      With da
        .gh = Text1(0).Text
        .xm = Text1(1).Text
        .xb = Option1(0).Value
        .gz = Val(Text1(2).Text)
        .rq = Text1(3).Text
      End With
      Open "worker.dat" For Random As #1 Len = Len(da)          ' 打开随机文件
      lastrec = LOF(1) / Len(da)
      Put #1, lastrec + 1, da
      Close #1
      Call Form_Activate
      Text1(0).SetFocus
    Case 1
      recnum = n
      Open "rec.tem" For Random As #1 Len = Len(da)             ' 打开临时随机文件
      Open "worker.dat" For Random As #2 Len = Len(da)          ' 打开随机文件
      lastrec = LOF(2) / Len(da)
      For i = 1 To lastrec
        If i < > recnum Then
          Get #2, i, da
          Put #1, , da
        End If
      Next
```

```
            Close #1
            Close #2
            Kill "worker.dat"
            Name "rec.tem" As "worker.dat"
            Call Form_Activate
        Case 2
            With da
                .gh = Text1(0).Text
                .xm = Text1(1).Text
                .xb = Option1(0).Value
                .gz = Val(Text1(2).Text)
                .rq = Text1(3).Text
            End With
            Open "worker.dat" For Random As #1 Len = Len(da)          ' 打开随机文件
            Put #1, n, da
            Close #1
    End Select
End Sub
```

“查找”菜单控件数组 Fnd 的 Click 事件代码：

```
Private Sub Fnd_Click(Index As Integer)
    Dim d(4) As String
    Open "worker.dat" For Random As #2 Len = Len(da)                  ' 打开随机文件
    lastrec = LOF(2) / Len(da)
    Select Case Index
        Case 0                                                        ' 按姓名查找
            a = Trim(InputBox("查找的姓名：", "请输入"))
            If a < > "" Then
                List1.Clear
                For i = 1 To lastrec
                    Get #2, i, da
                    b = Trim(da.xm)
                    If b Like a & "*" Then                            ' 模糊查找
                        d(0) = da.xm
                        d(1) = da.gh
                        d(2) = da.gz
                        d(3) = da.xb
                        List1.AddItem d(1) & d(2) & " " & d(0)
                    End If
                Next
```

```
        If List1.ListCount < 1 Then
            MsgBox "没有找到" & a
        End If
    End If
Case 1                                                  ' 按工资范围查找
    Dim x As Integer, y As Integer
    x = InputBox("请输入查找工资范围的下限：", "请输入")
    y = InputBox("请输入查找工资范围的上限：", "请输入")
    If x <= y Then
        List1.Clear
        For i = 1 To lastrec
            Get #2, i, da
            If da.gz >= x And da.gz <= y Then
                d(0) = da.xm
                d(1) = da.gh
                d(2) = da.gz
                d(3) = da.xb
                List1.AddItem d(1) & d(2) & " " & d(0)
            End If
        Next
        If List1.ListCount < 1 Then
            MsgBox "没有找到"
        End If
    End If
Case 2                                                  ' 按工作证号查找
    a = Trim(InputBox("查找的工作证号：", "请输入"))
    If a < > "" Then
        List1.Clear
        For i = 1 To lastrec
            Get #2, i, da
            b = Trim(da.gh)
            If b Like a & "*" Then                      ' 模糊查找
                d(0) = da.xm
                d(1) = da.gh
                d(2) = da.gz
                d(3) = da.xb
                List1.AddItem d(1) & d(2) & " " & d(0)
            End If
        Next
```

```
            If List1.ListCount < 1 Then
               MsgBox "没有找到" & a
            End If
         End If
      End Select
      Close #2                                          ' 关闭随机文件
End Sub
```

12.20　设计一个能运行可执行文件（.exe, .com, .bat）的对话框程序，程序启动后界面如图 12-3 所示。单击“浏览”按钮将打开图 12-4 所示的“打开”对话框，在选择文件后单击“打开”按钮，返回对话框程序，此时用户选择的文件名将显示在文本框中。通过选择单选钮可以使程序按“常规”、“最大化”或“最小化”方式运行。单击“取消”按钮将清除文本框中的内容。



图 12-3　对话框示例程序



图 12-4　Windows 的“打开”对话框

【解答】程序设计步骤如下。

（1）设计程序界面。在窗体上添加 3 个命令按钮，一个标签，3 个单选钮和一个框架。添加 ActiveX 控件“Microsoft Common Dialog Control 6.0”（公共对话框控件）到工具箱中，并将其添加到窗体中，如图 12-5 所示。

（2）设置对象属性。设置标签的 Caption 属性为“文件名”，AutoSize 属性为 True；设置文本框的 Text 属性为空；设置 3 个按钮的 Caption 属性分别为“运行”、“取消”和“浏览”；设置框架的 Caption 属性为“选项”；设置 3 个单选钮的 Caption 属性分别为“常规”、“最小化”和“最大化”，Option1 的 Value 属性为 True；设置窗体的 Caption 属性为“对话框示例程序”。



图 12-5　设计程序界面

（3）编写程序代码。

```
Private Sub Command1_Click( )          ' 单击“打开”按钮时执行的代码
```

```
    On Error GoTo CUOWU                 ' 若出现错误转去执行以“CUOWU”为标号的程序行
    If Option1 Then Shell Text1.Text, 1 ' 此处使用了 Option 控件的默认属性 Value
    If Option2 Then Shell Text1.Text, 2
    If Option3 Then Shell Text1, 3
    Exit Sub
CUOWU:                                  ' 错误处理程序
    If Text1 = "" Then
        MsgBox "未选择文件，无法运行！", 48, "警告"
    Else
        MsgBox "不能运行该程序！", 48, "注意："
    End If
    Resume Next                         ' 从出错语句的下一个语句开始恢复运行
End Sub
Private Sub Command2_Click( )           ' 单击“取消”按钮时执行的代码
    Text1.Text = ""
    Text1.SetFocus
End Sub
Private Sub Command3_Click( )           ' 单击“浏览”按钮时执行的代码
    CommonDialog1.ShowOpen              ' 显示 Windows 的“打开”对话框
    Text1 = CommonDialog1.FileName      ' 将用户选择的可执行文件名显示到文本框中
End Sub
```

第 13 章　键盘与鼠标事件过程

一、选择题

13.1　在窗体上画一个名称为 TxtA 的文本框，然后编写如下事件过程：

```
Private Sub TxtA_KeyPress(KeyAscii As Integer)
  ……
End Sub
```

若焦点位于文本框中，则能够触发 KeyPress 事件的操作是（　　）。

A）单击鼠标　　B）双击文本框

C）鼠标滑过文本框　　D）按下键盘上的某个键

【解答】正确答案为 D。

分析：KeyPress 事件是键盘的按键事件。

13.2　把窗体的 KeyPreview 属性设置为 True，然后编写如下事件过程：

```
Private Sub Form_KeyPress(KeyAscii As Integer)
  Dim ch As String
  ch=Chr(KeyAscii)
  KeyAscii=Asc(UCase(ch))
  Print Chr(KeyAscii+2)
End Sub
```

程序运行后，按键盘上的 A 键，则在窗体上显示的内容是（　　）。

A）A　　B）B　　C）C　　D）D

【解答】正确答案为 C。

分析：本程序实现字符及其 ASCII 码之间的转换。需要注意的是，UCase(ch)是将字符转化为其相应的大写形式，Asc()是取字符的 ASCII 码，Chr()是将 ASCII 码转化为相应的字符。

13.3　以下叙述中错误的是（　　）。

A）在 KeyUp 和 KeyDown 事件过程中，从键盘上输入 A 或 a 被视做相同的字母（即具有相同的 KeyCode）

B）在 KeyUp 和 KeyDown 事件过程中，将键盘上的 1 键和右侧小键盘上的 1 键视做不同的数字（具有不同的 KeyCode）

C）KeyPress 事件中不能识别键盘上某个键的按下与释放

D）KeyPress 事件中可以识别键盘上某个键的按下与释放

【解答】正确答案为 D。

13.4　假定有如下事件过程：

```
Private Sub Form_MouseDown(Button As Integer, Shift As Integer, X As Single, Y As Single)
  If Button=2 Then
    PopupMenu  popForm
  End If
End Sub
```

则以下描述错误的是（　　）。

A）该过程的功能是弹出一个菜单

B）popForm 是在菜单编辑器中定义的弹出式菜单的名称

C）参数 X、Y 指明鼠标的当前位置

D）Button=2 表示按下的是鼠标左键

【解答】正确答案为 D。

分析：Button=2 表示按下鼠标右键。

13.5　在窗体上画一个文本框，其名称为 Text1，然后编写如下过程：

```
Private Sub Text1_KeyDown(KeyCode As Integer,Shift As Integer)
  Print Chr(KeyCode)
End Sub
Private Sub Text1_KeyUp(KeyCode As Integer,Shift As Integer)
  Print Chr(KeyCode+2)
End Sub
```

程序运行后，把焦点移到文本框中，此时如果按下 A 键，则输出结果为（　　）。

A）A　　B）A　　C）A　　D）A

　　A　　B　　C　　D

【解答】正确答案为 C。

分析：依据键盘与鼠标事件中 KeyDown 事件和 KeyUp 事件的用法，本题选 C。

13.6　在窗体上画一个文本框，然后编写如下事件过程：

```
Private Sub Text1_KeyPress(KeyAscii As Integer)
  Dim char As String
  char=Chr(KeyAscii)
  KeyAscii=Asc(UCase(char))
  Text1.Text=String(6, KeyAscii)
End Sub
```

程序运行后，如果从键盘上输入字母 a，则文本框中显示的内容为（　　）。

A）a　　B）A　　C）aaaaaaa　　D）AAAAAAA

【解答】正确答案为 D。

分析：该事件过程中首先用 Chr 函数记录下输入的字符，然后用 UCase 函数将其转化为大写，再用 Asc 函数记录下其码值。最后使用 String 函数将其重复 6 遍并输出到文本框中。

当文本框中的文本内容发生变化的时候，会触发文本框的 Change 事件，而 KeyPress 事件是在 Change 事件之前触发的。在发生 KeyPress 事件之前，尚未触发 Change 事件。在 KeyPress 事件中，改变了文本框的 Text 属性，所以会触发 Change 事件，之后，键盘输入的信息 a 才在变为大写之后输入文本框，文本框中最终显示的是 7 个 A。

13.7　在窗体上画一个名称为 Command1 的命令按钮，然后编写如下程序：

```
Dim SW As Boolean
Function func(X As Integer)As Integer
    If X<20 Then
        Y=X
    Else
        Y=20+X
    End If
    func=Y
End Function
Private Sub Form_MouseDown(Button As Integer, Shift As Integer,X As Single,Y As Single)
    SW=False
End Sub
Private Sub Form_MouseUp(Button As Integer, Shift As Integer,X As Single,Y As Single)
    SW=True
End Sub
Private Sub Command1_Click()
    Dim intNum As Integer
    intNum=InputBox(" ")
    If SW Then
      Print func(intNum)
    End If
End Sub
```

程序运行后，单击命令按钮，将显示一个输入对话框。如果在对话框中输入 25，则程序的执行结果为（　　）。

A）0　　B）25　　C）45　　D）无任何输出

【解答】正确答案为 C。

分析：当按下鼠标左键（发生 MouseDown）时，SW 的值为 False；当松开鼠标左键（触发 MouseUp）时，SW 的值为 True。

二、填空题

13.8　在 KeyPress 事件过程中，KeyAscii 是所按键的____值。

【解答】正确答案为：ASCII。

分析：当按下键盘上的某个键时，触发 KeyPress 事件过程，该事件用到 KeyAscii 作为

参数，此参数是一个预定义的变量，执行 KeyPress 事件过程时，KeyAscii 是所按键的 ASCII 码值。

13.9　有如下事件过程，当同时按下 Shift 键和 F5 键时，最后输出的信息是____。

```
Const ShiftKey =-1
Const CtrlKey =2
Const Key_F5 =&H74
Const Key_F6 =&H75
Private Sub Text1_KeyDown(KeyCode As Integer,Shift As Integer)
    If KeyCode = Key_F5 And Shift = ShiftKey Then
       Print "Press Shift+F5"
    ElseIf KeyCode = Key_F6 And Shift = CtrlKey Then
       Print "Press Ctrl+F6"
    End If
End Sub
```

A）无任何信息　　B）Press Shift+F5　　C）Press Ctrl+F6　　D）程序出错

【解答】正确答案为 B。

分析：上述事件过程是测试两个参数（KeyCode 和 Shift）是否同时满足给定的条件，如果满足，则输出相应的信息。题中是同时按下 Shift 键和 F5 键，所以满足给定的条件，输出信息 Press Shift+F5。

三、编程题

13.10　编写一个程序，当同时按下 Shift 键和 F6 键时，在窗体上显示“再见！”，并终止程序的运行。

【解答】程序代码如下。

```
Private Sub Form_Load( )
   Form1.KeyPreview = True
End Sub
Private Sub Form_KeyDown(KeyCode As Integer, Shift As Integer)
   If Shift = 1 And KeyCode = vbKeyF6 Then
      MsgBox "再见！"
      End
   End If
End Sub
```

13.11　编写一个程序，当按下某个键时，程序以十六进制数和八进制数的形式输出该键的 KeyCode。

【解答】如图 13-1 所示，可利用 Oct, Hex 函数对 KeyCode 进行转换，代码如下：

```
Private Sub Form_Load( )
   Form1.KeyPreview = True
End Sub
```

```
Private Sub Form_KeyDown(KeyCode As Integer, Shift As Integer)
   Print "Oct:"; Oct(KeyCode); "     Hex:"; Hex(KeyCode)
End Sub
```

13.12　在窗体上画一个文本框、一个图片框和一个命令按钮。编写程序，使得当鼠标光标位于不同的控件或窗体上时，鼠标光标具有不同的形状，此时如果同时按下鼠标右键，则显示相应的信息。例如，当鼠标光标移动到图片框上时，如果按下鼠标右键，则用一个信息框显示“这是图片框”。



图 13-1　输出键值

要求：在文本框和窗体上的鼠标光标使用系统提供的光标形状，而图片框和命令按钮上的鼠标光标使用自己定义的形状。

【解答】本题要求鼠标在不同的对象上时触发不同的 MouseMove 事件，显示不同的 MouseIcon 属性；按键则触发 MouseDown 事件，显示有关的信息。代码如下：

```
Private Sub Form_MouseMove(Button As Integer, Shift As Integer, X As Single, Y As Single)
   Form1.MousePointer = 0
End Sub
Private Sub Text1_MouseDown(Button As Integer, Shift As Integer, X As Single, Y As Single)
   If Button = 2 Then MsgBox ("这是文本框")
End Sub
Private Sub Text1_MouseMove(Button As Integer, Shift As Integer, X As Single, Y As Single)
   Form1.MousePointer = 0
End Sub
Private Sub Command1_MouseDown(Button As Integer, Shift As Integer, X As Single, _
                               Y As Single)
   If Button = 2 Then MsgBox ("这是命令按钮")
End Sub
Private Sub Command1_MouseMove(Button As Integer, Shift As Integer, X As Single, _
                               Y As Single)
   Form1.MousePointer = 99
   Form1.MouseIcon = LoadPicture("c:\Point03.ico")
End Sub
Private Sub Picture1_MouseDown(Button As Integer, Shift As Integer, X As Single, Y As Single)
   If Button = 2 Then MsgBox ("这是图片框")
End Sub
Private Sub Picture1_MouseMove(Button As Integer, Shift As Integer, X As Single, Y As Single)
   Form1.MousePointer = 99
   Form1.MouseIcon = LoadPicture("c:\Point02.ico")
End Sub
```



图 13-2　回收站

13.13　编写一个类似于“回收站”的程序，并用适当的图形作为“回收站”图标。程序运行后，把窗体上其他的对象拖到“回收站”图标上，松开鼠标键后，显示一个信息框，询问是否要把该对象放入回收站，如图 13-2 所示。此时单击“是”按钮即放入回收站，对象从窗体上消失；单击“否”按钮，则对象仍回到原来位置。

【解答】本题主要涉及拖放事件。

```
Private Sub Form_Load( )
   Picture2.DragMode = 0
   Picture2.DragIcon = LoadPicture("c:\Drag2pg.ico")
End Sub
Private Sub Picture1_DragDrop(Source As Control, X As Single, Y As Single)
   i = MsgBox("要将此对象放入回收站", 4)
   If i = 6 Then
      Source.Visible = False
   End If
End Sub
Private Sub Picture2_MouseDown(Button As Integer, Shift As Integer, X As Single, Y As Single)
   Picture2.Drag 1
End Sub
```

13.14　在窗体上画若干个控件，然后画两个列表框，其中一个列表框用来列出当前窗体上控件的名称，另一个列表框列出 15 种鼠标光标的形状。程序运行后，从第一个列表框中选择控件或窗体，从第二个列表框中选择鼠标光标形状，为选择的控件或窗体设置所需要的鼠标光标形状，如图 13-3 所示。

要求：两个列表框隐藏，只在需要时才显示出来。



图 13-3　设置鼠标光标形状

【解答】本题目的是将控件与鼠标光标建立起联系，即 List1 列表框中存放窗体上的有关控件，List2 列表框中存放 15 种鼠标光标样式。当在列表框选择了控件和鼠标光标后，鼠标指向该控件时，鼠标光标改变为所选择的形状。运行界面如图 13-3 所示，程序代码如下：

```
Private Sub Form_Load( )
   List1.AddItem "Label1"
   List1.AddItem "Text1"
   List1.AddItem "Command1"
   List1.AddItem "Picture1"
   For i = 1 To 15
      List2.AddItem i
   Next i
End Sub
Private Sub List1_Click( )
```

```
    Select Case List1.ListIndex
        Case 0
            Label1.MousePointer = Val(List2.Text)
        Case 1
            Text1.MousePointer = Val(List2.Text)
        Case 2
            Command1.MousePointer = Val(List2.Text)
        Case 3
            Picture1.MousePointer = Val(List2.Text)
    End Select
End Sub
```

第 14 章　数 据 文 件

一、选择题

14.1　VB 根据计算机访问文件的方式将文件分成 3 类，其中不包括（　　）。

A）顺序文件　　B）UNIX 文件　　C）二进制文件　　D）随机文件

【解答】正确答案为 B。

分析：VB 访问文件的方式分为 3 类：Input/Output/Append（访问顺序文件）、Binary（访问二进制文件）、Random（访问随机文件）。如果未指定方式，则以 Random 访问方式打开文件。

14.2　下列说明中，不属于随机文件特点的是（　　）。

A）可以随意读取随机文件中任一记录的数据

B）随机文件没有只读或只写的操作方式，随机文件只要一打开，就既可读又可写

C）随机文件的操作是以记录为单位进行的

D）随机文件的读/写操作语句与顺序文件的读/写操作语句一样

【解答】正确答案为 D。

分析：顺序文件的读/写操作语句是 Input#（Line Input#）语句和 Print#（Write#）语句，而随机文件的读/写操作语句是 Get # 语句和 Put # 语句。

14.3　以下叙述中正确的是（　　）。

A）一条记录中所包含的各个元素的数据类型必须相同

B）随机文件中每条记录的长度是固定的

C）Open 命令的作用是打开一个已经存在的文件

D）使用 Input#语句可以从随机文件中读取数据

【解答】正确答案为 B。

分析：随机文件中记录是定长的。

14.4　以下关于文件的叙述中，错误的是（　　）。

A）顺序文件中的记录是一个接一个顺序存放的

B）随机文件中记录的长度是随机的

C）执行打开文件的命令后，自动生成一个文件指针

D）LOF 函数返回给文件分配的字节数

【解答】正确答案为 B。

分析：在随机文件中记录的长度都是固定的，所以选项 B 说法错误；顺序文件顾名思义文件都是顺序存放的，选项 A 说法正确；执行打开文件的命令后，自动生成一个文件指针，选项 C 说法正确；LOF 函数用来返回文件分配的字节数，选项 D 说法正确。

14.5　用 Write 和 Print 语句向文件中写入多个数据的差别在于（　　）。

A）Write 语句不会自动在数据项之间插入逗号

B）Print 语句自动在数据项之间插入逗号

C）Write 语句写入字符串会自动给字符串加上双引号，写入的正数前面没有空格

D）Print 语句写入字符串会自动给字符串加上双引号，写入的正数前面没有空格

【解答】正确答案为 C。

分析：Write 语句和 Print 语句的基本功能相同，其主要区别有以下两点。

① 用 Write 语句向文件写数据时，数据在磁盘上以紧凑格式存放，能自动在数据项之间插入逗号，并给字符串加上双引号。一旦最后一项被写入，就插入新的一行。

② 用 Write 语句写入的正数的前面没有表示符号位的空格。

14.6　设有语句

```
Open "C:\\Test.Dat" For Output As #1
```

则以下错误的叙述是（　　）。

A）该语句打开 C 盘根目录下一个已存在的文件 Test.Dat

B）该语句在 C 盘根目录下建立一个名为 Test.Dat 的文件

C）该语句建立的文件的文件号为 1

D）执行该语句后，就可以通过 Print#语句向文件 Test.Dat 中写入信息

【解答】正确答案为 A。

分析：根据文件操作 Open 的使用方法。执行 Open 语句时，打开文件的文件号与一个具体的文件相关联，其他输入/输出语句或函数通过文件号与文件发生关系。

14.7　要把一个记录型变量的内容写入文件中指定的位置，所使用的语句格式为（　　）。

A）Get 文件号, 记录号, 变量名　　B）Get 文件号, 变量名, 记录号

C）Put 文件号, 变量名, 记录号　　D）Put 文件号, 记录号, 变量名

【解答】正确答案为 D。

14.8　要获得当前驱动器，应使用驱动器列表框的（　　）属性。

A）Path　　B）Drive　　C）Dir　　D）Pattern

【解答】正确答案为 B。

分析：驱动器控件常用的 Drive 属性返回当前驱动器号。

14.9　目录列表框的 Path 属性的作用是（　　）。

A）显示当前驱动器或指定驱动器中的路径

B）显示当前驱动器或指定驱动器中某目录下的文件名

C）显示根目录下的文件名

D）只显示当前路径下的文件

【解答】正确答案为 A。

分析：Path 属性返回或设置当前路径。

14.10　在窗体上画一个名称为 Drive1 的驱动器列表框和一个名称为 Dir1 的目录列表框。当改变当前驱动器时，目录列表框应该与之同步改变。设置两个控件同步的命令放在一个事

件过程中，这个事件过程是（　　）。

A）A）Drive1_Change　　B）Drive1_Click

C）Dir1_Click　　D）Dir1_Change

【解答】正确答案为 A。

分析：当用户在驱动器列表框中选择一个新的驱动器后，Drive1 的 Drive 属性改变，触发 Change 事件。

14.11　要使文件列表框中的文件随目录列表框中所选择的当前目录的不同而发生变化，应该（　　）。

A）在 File1 中的 Change 事件中，输入 File1.Path=Dir1.Path

B）在 Dir1 中的 Change 事件中，输入 File1.Path=Dir1.Path

C）在 File1 中的 Change 事件中，输入 Dir1.Path=File1.Path

D）在 Dir1 中的 Change 事件中，输入 Dir1.Path=File1.Path

【解答】正确答案为 B。

分析：本题涉及文件列表框与目录列表框的关联问题。当用户触发文件的 Change 事件时，文件的列表随着目录位置的改变而改变，其中目录为源操作数，位于等号右边，文件为目的操作数，位于等号的左边，所以正确答案为选项 B。

二、填空题

14.12　对随机文件数据存取是以____为单位进行操作的。

【解答】正确答案为：一条记录。

分析：从存储的角度看，顺序文件一般用来存放同类型或文本类型数据，随机文件一般用来存放一组相关自定义类型的数据。

每组自定义类型数据由不同类型的基本类型数据组成。这样一组自定义的类型数据构成了一条记录。一条记录用一个自定义变量名表示，称之为记录型变量。当向随机文件写入（读取）数据时，每次只可写入（读取）一个记录型变量，即写入（读取）一条记录。

14.13　文件根据数据性质，可分为____文件和____文件。

【解答】正确答案为：程序；数据。

分析：如果一个文件内存放的是程序数据，则该文件是程序文件；否则该文件内存放的就是供其他程序使用的数据，也就是数据文件。

14.14　下列程序的功能是：将数据 1，2，…，8 写入顺序文件 Num.txt，请补充完整。

```
Private Sub Form_Click()
  Dim i As Integer
  Open "Num.txt" For Output As #1
  For i=1 To 8
    ____
  Next i
  Close #1
End Sub
```

【解答】正确答案为：Print #1, i。

分析：向文件中写入数据用 Print 语句。

14.15　改变驱动器列表框的 Drive 属性值将引发____事件。

【解答】正确答案为：Change。

分析：驱动器列表框最常用的事件是 Change 事件，每次重新设置 Drive 属性都会引发该事件。

三、编程题

14.16　在当前目录中有顺序文件 in7.txt（文件中只有字母和空格）。在窗体上画一个文本框，名称为 Text1，允许多行显示；再画 3 个命令按钮，名称分别为 C1, C2, C3，标题分别为“输入”、“转换”、“存盘”（如图 14-1 所示）。请编写适当的事件过程，使得在运行时，单击“输入”按钮，则从考生文件夹中读入 in7.txt 文件，放入 Text1 中；单击“转换”按钮，则把 Text1 中的所有小写字母转换为大写字母；单击“存盘”按钮，则把 Text1 中的内容存入 out7.txt 文件中。

【解答】程序用户界面参照图 14-3 设计。3 个命令按钮的事件过程代码分别为：

```
Private Sub C1_Click( )        ' “输入”命令按钮
   Open App.Path & "\" & "in7.txt" For Input As #1
   a = ""
   Do Until EOF(1)
      Line Input #1, b
      a = a & b & Chr(13) & Chr(10)
   Loop
   Close #1
   Text1.Text = a
End Sub
Private Sub C2_Click( )        ' “转换”命令按钮
   Text1.Text = UCase(Text1.Text)
End Sub
Private Sub C3_Click( )        ' “存盘”命令按钮
   Open App.Path & "\" & "out7.txt" For Output As #1
   Print #1, Text1.Text
   Close #1
End Sub
```



图 14-1　小写字母转换为大写字母

14.17　某单位全年每次报销的经费（假定为整数）存放在一个磁盘文件中，试编写一个程序，从该文件中读出每次报销的经费，计算其总和，并将结果存入另一个文件中。

【解答】求解本题必须考虑存储报销经费的磁盘文件的数据结构，假定每次报销的数据含时间、费用两项数据，数据之间用逗号分界，则可用循环方式每次读出两项数据到两个变量中，并将费用项数据进行累加，文件读取完后，将累加和写入另一个文件中。事件代码为：

```
Private Sub Command1_Click( )
   Dim sj As String, jfsum As Integer
   Open "报销经费.txt" For Input As #1
```

```
    jfsum = 0                                  ' 累加和初值
    Do While Not EOF(1)
       Input #1, sj, jf                        ' 从文件中读时间、费用
       jfsum = jfsum + jf
    Loop
    Open "经费总和.txt" For Output As #2
    Print #1, jfsum
    Close #1, #2
End Sub
```

14.18　编写一个程序，用来处理活期存款的结算事务。每次处理之后，程序都要显示当前的结存，并把它存入一个文件中，要求输出的浮点数保留小数点后两位。

【解答】活期存款结算的数据结构包含时间、支出、收入、余额 4 项，数据之间的关系为：余额＝余额＋收入－支出。在窗体上放置 3 个文本框，用于输入每笔交易的数据。假定活期存款数据文件是顺序文件，要把交易数据添加到文件中，可用 Append 方式打开文件。事件代码为：

```
Dim mye As Single                              ' 声明窗体级变量，保存当前的余额
Private Sub Form_Load( )
   If Dir("活期存款.txt") > " " Then           ' 读出当前的余额
      Open "活期存款.txt " For Input As #1
      Do While Not EOF(1)
         Input #1, mri, mzc, msr, mye
      Loop
      Close #1
   Else
      mye = 0
   End If
    Label2 = Format(mye, "#####.##")
    Open "活期存款.txt" For Append As #1
End Sub
Private Sub Command1_Click( )
   mri = Text1.Text                            ' 时间
   mzc = Val(Text2.Text)                       ' 支出
   msr = Val(Text3.Text)                       ' 收入
   mye = mye + msr - mzc                       ' 余额
   Print #1, mri, mzc, msr, mye;               ' 交易数据添加到文件中
   Label2 = Format(mye, "#####.##")            ' 显示当前的余额
End Sub
```

14.19　编写程序，按下列格式输出月历，并把结果放入一个文件中。

SUN	MON	TUE	WED	THU	FRI	SAT
1	2	3	4	5	6	7
8	9	10	11	12	13	14
15	16	17	18	19	20	21
22	23	24	25	26	27	28
29	30	31				

【解答】用 Print 语句在窗体上输出月历，用 Print #语句把结果放入一个文件中。注意输出数据位置要对齐。事件代码为：

```
Private Sub Command1_Click( )
  Open "月历.txt" For Output As #1
  Print #1, "  SUN   MON   TUE   WED   THU   FRI   SAT"
  Print "  SUN   MON   TUE   WED   THU   FRI   SAT"
  For i = 1 To 31
    If i < 10 Then Print " ";: Print #1, " ";          ' 1 位数前加一个空格
    Print "   "; i;                                    ' 输出到显示屏上
    Print #1, "   "; i;                                ' 输出到文件中
    If i Mod 7 = 0 Then Print: Print #1,               ' 每 7 个数换行一次
  Next i
  Close #1
End Sub
```

14.20　假定在磁盘上已建立了一个通信录文件，文件中的每条记录都包括编号、用户名、电话号码和地址 4 项内容。编写一个程序，用自己选择的检索方法从文件中查找指定用户的编号，并在文本框中输出其名字、电话号码和地址。

【解答】在窗体上放置包含 4 个文本框控件的数组 Text1( )和一个命令按钮 Command1。在 Text1(0)中输入要检索的用户编号。

编写 Command1_Click 子过程用于检索，事件代码如下：

```
Private Type tx1Type
  strID As Integer                          ' 编号
  strname As String * 10                    ' 用户名
  strph As String * 8                       ' 电话号码
  stradd  As String * 20                    ' 地址
End Type
Dim tx1 As tx1Type                          ' 声明变量
Private Sub Command1_Click( )
  Findno = Text1(0)                         ' 要检索的用户编号
  Open "通信录.dat" For Random As #1 Len = Len(tx1)
  Do While Not EOF(1)
    Get #1, , tx1                           ' 读出记录
    If Findno = strID Then
```

```
            Text1(0) = strID
            Text1(1) = tx1.strname
            Text1(2) = tx1.strph
            Text1(3) = tx1.stradd
        End If
    Loop
    If EOF(1) Then MsgBox "无 " & Findno & "用户编号"
    Close #1
End Sub
```

14.21　通过键盘输入数据，包括学号、姓名、性别及数学、英语、电子的成绩等数据，将这些数据输入到一个顺序文件（stu.dat）中。

【解答】以追加方式建立并打开顺序文件 stu.dat，用 Write 语句将数据内容写入该文件中。设计步骤如下。

（1）建立程序界面并设置属性，如图 14-2（a）所示。



（a）　　（b）

图 14-2　程序界面与程序运行结果

（2）编写代码。

首先编写窗体的 Load 事件代码：

```
Private Sub Form_Load( )
    Open "stu.dat" For Append As #1       ' 因文件不能重复打开，故在 Load 事件中打开
End Sub
```

编写命令按钮控件数组 Command1( )的 Click 事件代码：

```
Private Sub Command1_Click(Index As Integer)
    Select Case Index
        Case 0
            Dim xh As String, xm As String, xb As String
            Dim sx As Integer, yy As Integer, dz As Integer
            xh = Text1(0).Text
            xm = Text1(1).Text
            xb = IIf(Option1(0).Value, "男", "女")
            sx = Val(Text1(2).Text)
            yy = Val(Text1(3).Text)
            dz = Val(Text1(4).Text)
            Write #1, xh, xm, xb, sx, yy, dz                ' 写入数据文件
```

```
        For i = 0 To 4
            Text1(i).Text = ""                    ' 清除文本框中的内容
        Next
        Text1(0).SetFocus                         ' 将光标设置到 Text1(0)
    Case 1
        Close #1                                  ' 关闭数据文件
        Unload Me
End Select
End Sub
```

为了方便输入，编写文本框的事件代码如下：

```
Private Sub Text1_GotFocus(Index As Integer)
    Text1(Index).SelStart = 0
    Text1(Index).SelLength = Len(Text1(Index).Text)
End Sub
Private Sub Text1_KeyPress(Index As Integer, KeyAscii As Integer)
    If KeyAscii = 13 Then
        n = Index
        k = IIf(n < 4, n + 1, 0)
        Text1(k).SetFocus
    End If
End Sub
```

运行程序，在文本框中输入各数据后，单击“输入”按钮将数据写入文件中，并清除文本框中的内容，继续输入。单击“结束”按钮，关闭文件，结束程序。如图 14-2（b）所示。

14.22　通过键盘输入数据，包括学号、姓名、性别及数学、英语、电子的成绩等数据，将这些数据输入到一个随机文件（xsda2.dat）中。

【解答】由于要存入随机文件，应先定义一个记录类型 cj，它包含 6 个字段，然后定义一个记录变量 da。打开随机文件 xsda2.dat，由于定义了记录变量，所以可以用 Len(da)测出每条记录的长度，不必自己计算记录长度。最后，将输入的学生数据写到随机文件中。

设计步骤如下。

（1）建立用户界面并设置对象属性，参见图 14-3。



（a）　　（b）

图 14-3　程序界面

（2）编写事件代码。

在通用声明段中编写如下代码，定义用户自定义类型变量：

```
Private Type cj
  xh As String * 10
  xm As String * 6
  xb As String * 2
  sx As Single
  yy As Single
  dz As Single
End Type
Private da As cj
```

编写“输入”命令按钮 Command1 的 Click 事件代码：

```
Private Sub Command1_Click( )
  da.xh = Text1.Text
  da.xm = Text2.Text
  da.xb = IIf(Option1.Value, "男", "女")
  da.sx = Val(Text3.Text)
  da.yy = Val(Text4.Text)
  da.dz = Val(Text5.Text)
  Put #1, , da
  Text1.Text = "" : Text2.Text = "" : Text3.Text = "" : Text4.Text = "" : Text5.Text = ""
  Text1.SetFocus
End Sub
```

编写“结束”命令按钮 Command2 的 Click 事件代码：

```
Private Sub Command2_Click( )
  Close #1
  End
End Sub
```

编写窗体的 Initialize 事件代码：

```
Private Sub Form_Initialize( )
  Open "xsda2.dat" For Random As #1 Len = Len(da)          ' 打开随机文件
End Sub
```

程序运行结果如图 14-4 所示。



图 14-4　运行结果

14.23　从 stu.dat 中读入全部学生成绩数据，将其中获奖学金的学生数据存入一个新文件（stu1.dat）中。评奖的条件是：每门课程成绩均在 85 分以上或 3 门课程总分在 270 分以上。

【解答】以 Input 方式打开顺序文件 stu.dat，用 Input 语句将数据内容读入记录数组中。然后选出获奖学金的学生数据，并写入新文件 stu1.dat 中。设计步骤如下。

（1）建立程序界面并设置属性，如图 14-5（a）所示。



（a）　　（b）

图 14-5　程序界面与程序运行结果

（2）编写事件代码。

为了方便读取数据，使用自定义记录类型的数组来存放学生数据。因此，首先在窗体模块的通用段创建用户定义类型，并声明记录数组如下：

```
Private Type studentrec
  xm As String                    ' 姓名变量定义为字符串
  xh As String                    ' 学号变量定义为字符串
  xb As String                    ' 性别变量定义为字符串
  sx As Single                    ' 数学变量定义为单精度数
  yy As Single                    ' 英语变量定义为单精度数
  dz As Single                    ' 电子变量定义为单精度数
End Type
Dim stu( ) As studentrec          ' 定义记录数组
```

然后在窗体的 Load 事件代码中读入顺序文件 stu.dat 中的数据，并打开顺序文件 stu1.dat：

```
Private Sub Form_Load( )
  Open "stu.dat" For Input As #1          ' 打开顺序文件 stu.dat
  n = 1
  i = 1
  Do Until EOF(1)
    ReDim Preserve stu(1 To n)
    Input #1, stu(i).xh, stu(i).xm, stu(i).xb, stu(i).sx, stu(i).yy, stu(i).dz        ' 读取数据文件
    List1.AddItem stu(i).xh & " " & stu(i).xm & stu(i).xb & _
                                  Str(stu(i).sx) & Str(stu(i).yy) & Str(stu(i).dz)
    n = n + 1
    i = i + 1
  Loop
  Close #1                                ' 关闭顺序文件 stu.dat
```

```
    Open "stu1.dat" For Output As #1                    ' 打开顺序文件 stu1.dat
End Sub
```

编写命令按钮控件数组 Command1 的 Click 事件代码：

```
Private Sub Command1_Click(Index As Integer)
  Select Case Index
    Case 0
      n = UBound(stu)
      For i = 1 To n
        p1 = stu(i).sx >= 85 And stu(i).yy >= 85 And stu(i).dz >= 85
        p2 = stu(i).sx + stu(i).yy + stu(i).dz >= 270
        If p1 Or p2 Then
          Write #1, stu(i).xh, stu(i).xm, stu(i).xb, stu(i).sx, stu(i).yy, stu(i).dz    ' 写入数据文件
        End If
      Next
      Close #1                                          ' 关闭数据文件
    Case 1
      Unload Me
  End Select
End Sub
```

程序运行结果如图 14-5（b）所示。

14.24　编写应用程序，实现功能如下：

（1）建立一个随机文件，管理某单位的职工情况，其中每条记录都由工作证号、姓名、性别、工资、工作日期 5 项组成，可以向此文件中添加新记录；

（2）可以修改、删除记录；

（3）可以按记录浏览所有职工的情况；

（4）可以按姓名查找，并显示找到的记录；

（5）可以按工作证号查找，并显示找到的记录。

【解答】设计步骤如下。

（1）建立程序界面和设置属性，如图 14-6（a）所示。



（a）　　（b）

图 14-6　程序界面与程序运行结果

（2）编写事件代码。因为使用随机文件，所以应先定义一个记录类型 worker，它包含 5 个字段，然后定义一个记录变量 da。

在窗体的通用段声明变量：

```
Private Type worker
  gh As String * 6                          ' 工作证号变量定义为字符串
  xm As String * 6                          ' 姓名变量定义为字符串
  xb As Boolean                             ' 性别变量定义为字符串
  gz As Single                              ' 工资变量定义为单精度数
  rq As Date                                ' 工作日期变量定义为日期型
End Type
Dim n As Integer, lastrec As Integer
Dim da As worker                            ' 定义记录变量
```

编写窗体的 Activate 事件代码：

```
Private Sub Form_Activate( )
  Open "worker.dat" For Random As #1 Len = Len(da)      ' 打开随机文件
  lastrec = LOF(1) / Len(da)
  n = 1
  Get #1, n, da
  Close #1                                  ' 关闭随机文件
  Call xs
End Sub
```

其中调用了显示记录的子过程：

```
Private Sub xs( )
  With da
    Text1(0).Text = .gh
    Text1(1).Text = .xm
    If .xb Then
      Option1(0).Value = True
    Else
      Option1(1).Value = True
    End If
    Text1(2).Text = .gz
    Text1(3).Text = .rq
  End With
  If n = 1 Then
    Command1(0).Enabled = False: Command1(1).Enabled = False
  Else
    Command1(0).Enabled = True: Command1(1).Enabled = True
  End If
  If n = lastrec Then
    Command1(2).Enabled = False: Command1(3).Enabled = False
```

```
    Else
      Command1(2).Enabled = True: Command1(3).Enabled = True
    End If
    Frame2(1).Caption = "移动记录：  " & n & "/" & lastrec
  End Sub
```

编写“移动记录”命令按钮数组 Command1 的 Click 事件代码：

```
  Private Sub Command1_Click(Index As Integer)
    Open "worker.dat" For Random As #1 Len = Len(da)              ' 打开随机文件
    lastrec = LOF(1) / Len(da)
    Select Case Index
      Case 0
        n = 1
      Case 1
        If n > 1 Then n = n – 1
      Case 2
        If n < lastrec Then n = n + 1
      Case 3
        n = lastrec
    End Select
    Get #1, n, da
    Close #1                                                     ' 关闭随机文件
    Call xs
  End Sub
```

编写“编辑”命令按钮数组 Command2 的 Click 事件代码：

```
  Private Sub Command2_Click(Index As Integer)
    Select Case Index
      Case 0
        With da
          .gh = Text1(0).Text
          .xm = Text1(1).Text
          .xb = Option1(0).Value
          .gz = Val(Text1(2).Text)
          .rq = Text1(3).Text
        End With
        Open "worker.dat" For Random As #1 Len = Len(da)          ' 打开随机文件
        lastrec = LOF(1) / Len(da)
        Put #1, lastrec + 1, da
        Close #1
        Call Form_Activate
```

```
            Text1(0).SetFocus
        Case 1
            recnum = n
            Open "rec.tem" For Random As #1 Len = Len(da)          ' 打开临时随机文件
            Open "worker.dat" For Random As #2 Len = Len(da)       ' 打开随机文件
            lastrec = LOF(2) / Len(da)
            For i = 1 To lastrec
                If i < > recnum Then
                    Get #2, i, da
                    Put #1, , da
                End If
            Next
            Close #1
            Close #2
            Kill "worker.dat"
            Name "rec.tem" As "worker.dat"
            Call Form_Activate
        Case 2
            With da
                .gh = Text1(0).Text
                .xm = Text1(1).Text
                .xb = Option1(0).Value
                .gz = Val(Text1(2).Text)
                .rq = Text1(3).Text
            End With
            Open "worker.dat" For Random As #1 Len = Len(da)       ' 打开随机文件
            Put #1, n, da
            Close #1
    End Select
End Sub
```

编写“查找”命令按钮数组 Command3 的 Click 事件代码：

```
Private Sub Command3_Click(Index As Integer)
    Open "worker.dat" For Random As #2 Len = Len(da)               ' 打开随机文件
    lastrec = LOF(2) / Len(da)
    Select Case Index
        Case 0
            a = Trim(InputBox("查找的姓名：", "请输入"))
            If a < > "" Then
                For i = 1 To lastrec
```

```
                Get #2, i, da
                b = Trim(da.xm)
                If b = a Then
                    n = i
                    Call xs
                    Exit For
                End If
            Next
            Close #2                                        ' 关闭随机文件
            If i > lastrec Then
                MsgBox "没有找到" & a
            End If
        End If
    Case 1
        a = Trim(InputBox("查找的工作证号：", "请输入"))
        If a < > "" Then
            For i = 1 To lastrec
                Get #2, i, da
                b = Trim(da.gh)
                If b = a Then
                    n = i
                    Call xs
                    Exit For
                End If
            Next
            If i > lastrec Then
                MsgBox "没有找到" & a
            End If
        End If
End Select
Close #2                                                    ' 关闭随机文件
End Sub
```

编写文本框控件数组 Text1 的 GotFocus 事件代码：

```
Private Sub Text1_GotFocus(Index As Integer)
    Text1(Index).SelStart = 0
    Text1(Index).SelLength = Len(Text1(Index).Text)
End Sub
```

如果在第一次查找时出现不正常的情况，可以试试加上如下代码：

```
Private Sub Form_Load( )
    Open "worker.dat" For Random As #1 Len = Len(da)          ' 打开随机文件
    lastrec = LOF(1) / Len(da)
    Get #1, lastrec, da
    Close #1                                                  ' 关闭随机文件
End Sub
```

运行结果如图 14-6（b）所示。

14.25　设计一个用于登录或注册的对话框程序，程序启动后默认为登录状态，如图 14-7 所示。如果用户尚未获得合法的用户名和密码，可单击“注册”单选钮，屏幕显示隐藏的“确认密码”输入栏，如图 14-8 所示。



图 14-7　登录状态



图 14-8　注册状态

【分析】用户在注册时，可以输入用户名和密码，然后输入确认密码，单击“确定”按钮进行注册。程序将在窗体上显示注册成功的信息，如图 14-9 所示。用户输入的用户名（未加密）和加密后的密码存放在指定的顺序文件中，如图 14-10 所示。

在图 14-10 所示文件中有 4 个注册用户，最后一个用户名和密码本来均为 qwert，可以看出，图 14-10 中密码部分已经被加密，变成了其他字符。

图 14-9　注册成功



图 14-10　存放用户名和密码的文件

用户登录时，输入自己的用户名和密码，程序将对应于该用户名的密码从文件中读出，做解密处理，解密后的密码如果和用户输入密码相同，显示登录成功信息，如图 14-11 所示；否则显示图 14-12 所示的出错信息。

图 14-11　登录成功

图 14-12　口令出错

本题在使用文本框输入密码时，为了使读者清楚地看到全过程，便于对题目的理解，没有使用密码框属性，在实际设计时可以将用户输入的密码用“*”号显示。

本题对密码的加密原则为：将英文字母对应的 ASCII 码加 4，使之变成一个新的字符；其他符号不变。在实际设计时可以自定义适当的加密算法，以提高加密的可靠性。



图 14-13　设计程序界面

【解答】程序设计步骤如下。

（1）设计程序界面。

在窗体上添加 4 个标签，3 个文本框，两个单选钮和两个命令按钮，如图 14-13 所示。并适当调整各控件的大小和位置。

（2）设置对象属性。

设置 4 个标签控件的 AutoSize 属性为 True，其他属性在窗体的 Load 事件代码中设置。

（3）编写程序代码。

```
Dim strInput As String, code As String
Dim reco As String, strTemp As String
Dim i As Integer, length As Integer
Dim intAsc As Integer
Private Sub Form_Load( )                    ' 窗体装入时执行的代码
   Form1.Caption = "登录、注册对话框"
   Text1 = ""
   Text2 = ""
   Text3 = ""
   Text3.Visible = False
   Label1 = "用户名："
   Label2 = "密码："
   Label3 = "确认密码："
   Label3.Visible = False
   Label4 = ""
   Option1.Caption = "登录"
   Option1 = True
   Option2.Caption = "注册"
   Command1.Caption = "确定"
   Command2.Caption = "退出"
End Sub
Private Sub Command1_Click( )               ' 单击“确定”按钮时执行的代码
   If Text1 = "" Then MsgBox "用户名不能为空", 48, "警告": Exit Sub
   If Text2 = "" Then MsgBox "用户名不能为空", 48, "警告": Exit Sub
   If Option2.Value = True Then             ' 表示当前为注册状态
      If Text2 < > Text3 Then MsgBox "两次口令不同，请重新输出", 48, "警告": Exit Sub
```

```
    ' 以下的语句用于加密存放在文件中的用户口令，请读者认真分析程序使用的加密算法
    ' 这里使用的算法虽然简单，但体现了一般加密算法的基本方法
      strInput = Text2
      i = 1
      code = ""
      length = Len(strInput)
      Do While i <= length
        strTemp = Mid(strInput, i, 1)
        If (strTemp >= "A" And strTemp <= "Z") Or (strTemp >= "a" And strTemp <= "z") Then
          intAsc = Asc(strTemp) + 4
          code = Left(code, i - 1) & Chr(intAsc)
        Else
          code = Left(code, i - 1) & strTemp
        End If
        i = i + 1
      Loop
      Open "D:\vb6\习题\login.dat" For Append As #1
      Write #1, Text1, code                    ' 将未加密的用户名和加密后的口令写入文件
      Close #1
      Label4 = "恭喜你，注册成功"
  Else                                         ' 表示当前为登录状态
    ' 以下为解密程序段，用于还原加密的密码
    Open "D:\vb6\习题\login.dat" For Input As #1
    Do Until EOF(1)
      Input #1, username, password
      username = Trim(username)
      password = Trim(password)
      If Text1 = username Then
        i = 1
        code = ""
        length = Len(password)
        Do While i <= length
          strTemp = Mid(password, i, 1)
          If (strTemp >= "A" And strTemp <= "^") Or (strTemp >= "a" And strTemp <= "~") Then
            intAsc = Asc(strTemp) - 4
            code = Left(code, i - 1) & Chr(intAsc)
          Else
            code = Left(code, i - 1) & strTemp
          End If
```

```
            i = i + 1
          Loop
          Exit Do
        End If
      Loop
      Close #1
      If Trim(Text2) = Trim(code) Then
        Label4 = "登录成功"
      Else
        Label4 = "口令错！"
      End If
    End If
End Sub
Private Sub Command2_Click( )
    End
End Sub
Private Sub Option1_Click( )          ' 用户选择“登录”方式时执行的代码
    Text3.Visible = False
    Label3.Visible = False
End Sub
Private Sub Option2_Click( )          ' 用户选择“注册”方式时执行的代码
    Text3.Visible = True
    Label3.Visible = True
    Text1.SetFocus
End Sub
```

14.26　利用随机文件读/写操作设计程序。在图 14-14 所示的程序界面中输入相关信息后，单击“追加记录”按钮，程序将计算出考生的总分，并显示添加成功的信息。如果在当前记录中输入希望查询的记录号后，单击“显示记录”按钮，则程序会将保存在数据文件中的记录显示在屏幕上，如图 14-15 所示。



图 14-14　追加记录



图 14-15　显示指定记录

【解答】程序界面设计如图 14-16 所示。用来显示成绩的 4 个文本框使用控件数组，其

他各控件的有关属性参见图 14-16。可以使用自定义数据类型存放考生信息，随机文件 student.dat 中的数据如图 14-17 所示。



图 14-16　设计程序界面



图 14-17　显示在记事本中的 student.dat 文件的内容

程序设计步骤如下。

（1）设计程序界面。

在窗体上添加 11 个标签，10 个文本框，一个框架和两个命令按钮，如图 14-16 所示。

（2）设置对象属性。

设置标签的 Caption 属性分别为：“准考证号”、“姓名”、“志愿 1”、“志愿 2”、“录取院校”、“数学”、“外语”、“语文”、“物理”、“总分”和“当前记录号”，设置各标签的 AutoSize 属性为 True；设置框架的 Caption 属性为“成绩”；设置两个命令按钮的 Caption 属性分别为“追加记录”和“显示记录”。

（3）编写程序代码。

在工程中添加一个标准模块，在模块中定义变量如下：

```
Type Stud                          ' 在标准模块中声明自定义数据类型
  intNo As String * 8              ' 存放准考证号
  strName As String * 8            ' 存放姓名
  strSex As String * 1             ' 存放性别
  sngMark(0 To 3) As Single        ' 存放 4 科成绩
  sngTotal As Single               ' 存放总分
  fFlag As Boolean                 ' 记录奖惩情况
  strWish(1 To 2) As String * 20   ' 存放第 1, 2 志愿
  strAdmitSchool As String * 20    ' 存放录取学校
End Type
```

在窗体模块中添加如下代码：

```
Dim Student As Stud, Reco As Integer
Private Sub Form_Load( )           ' 窗体装入时执行的代码
  For i = 0 To 3
    Text6(i) = ""
  Next
  Text1 = ""
  Text2 = ""
```

```
    Text3 = ""
    Text4 = ""
    Text5 = ""
    Text7 = ""
    Option1.Value = 1
End Sub
Private Sub Command1_Click( )        ' 单击“追加记录”按钮时执行的代码
    Student.intNo = Text1
    Student.strName = Text2
    If Option1.Value = True Then
        Student.strSex = "男"
    Else
        Student.strSex = "女"
    End If
    For i = 0 To 3
        Student.sngTotal = Student.sngTotal + Text6(i)
        Student.sngMark(i) = Text6(i)
    Next
    If Check1.Value = 1 Then
        Student.fFlag = True
    Else
        Student.fFlag = False
    End If
    Student.strWish(1) = Text3
    Student.strWish(2) = Text4
    Student.strAdmitSchool = Text5
    Open "D:\vb6\习题\student.dat" For Random As #1 Len = Len(Student)
    Reco = LOF(1) / Len(Student) + 1
    Put #1, Reco, Student
    Close #1
    Label8 = "总分：" & Student.sngTotal & "          记录已加入数据文件"
End Sub
Private Sub Command2_Click( )                    ' 单击“显示”按钮时执行的代码
    Reco = Text7
    Open "D:\vb6\习题\student.dat" For Random As #1 Len = Len(Student)
    Get #1, Reco, Student
    Close #1
    Text1 = Student.intNo
    Text2 = Student.strName
    Text3 = Student.strWish(1)
```

```
Text4 = Student.strWish(2)
Text5 = Student.strAdmitSchool
If Student.strSex = "男" Then
   Option1.Value = True
Else
   Option2.Value = True
End If
If Student.fFlag = True Then
   Check1.Value = 1
Else
   Check1.Value = 0
End If
For i = 0 To 3
   Text6(i) = Student.sngMark(i)
Next
Label8 = "总分：" & Student.sngTotal
```

End Sub

14.27　设计一个使用文件控件的示例程序，执行这个程序可以找到磁盘中保存的任何文件。在驱动器列表框中选择驱动器后，目录列表框和文件列表框中的内容自动发生相应的变化。选择目录列表框后，文件列表框中的内容自动发生相应的变化。双击某文件夹时，窗体上显示出当前路径。单击某文件时，在文本框中显示包括路径的文件名称。如果用户在文本框中输入正确的完整文件名（包括路径的文件名），并按回车键后，程序可以找到该文件；若找不到则显示错误信息。

【解答】程序设计步骤如下。

（1）设计程序界面。

在窗体上添加 5 个标签，一个文本框，一个命令按钮，一个驱动器列表框，一个文件夹列表框和一个文件列表框。适当调整各控件的大小及位置，如图 14-18 所示。

（2）设置对象属性。

设置所有标签的 AutoSize 属性为 True，其他对象的属性在窗体的 Load 事件代码中设置。



（a）　　（b）

图 14-18　文件控件

（3）编写程序代码。

```
Private Sub Form_Load( )                          ' 窗体装入时执行的代码
  Form1.Caption = "文件控件使用示例"
  Label1 = Dir1.Path                              ' 在标签中显示当前路径
  ' 下面的代码将始终使 Label1 标签水平居中
  Label1.Left = (Form1.Width - Label1.Width) / 2
  Label2 = "驱动器："
  Label3 = "目录："
  Label4 = "文件："
  Label5 = "文件名："
  Text1 = ""
  Command1.Caption = "退出"
End Sub
Private Sub Command2_Click( )
  End
End Sub
Private Sub Command1_Click( )                     ' 单击“退出”按钮时执行的代码
  End
End Sub
Private Sub Dir1_Change( )                        ' 目录列表框中发生改变时执行的代码
  File1.Path = Dir1.Path
  Label1 = Dir1.Path
  Label1.Left = (Form1.Width - Label1.Width) / 2
End Sub
Private Sub Drive1_Change( )                      ' 驱动器列表框中发生改变时执行的代码
  Dir1.Path = Drive1.Drive
End Sub
Private Sub File1_Click( )                        ' 文件列表框中发生单击事件时执行的代码
  If Right(Dir1.Path, 1) < > "\" Then
    Text1 = Dir1.Path & "\" & File1.FileName
  Else
    Text1 = Dir1.Path & File1.FileName
  End If
End Sub
Private Sub Text1_KeyPress(KeyAscii As Integer)   ' 文本框中发生键盘按键事件时执行的代码
  On Error GoTo 2002                              ' 程序出错转去执行行标号为 2002 的程序段
  If KeyAscii = 13 Then                           ' 按 Enter 键时执行的代码
   ' 找出最后一个“\”符号所在的位置，并以此为分界，左边是路径名，右边是文件名
    Do Until a = "\" Or i = Len(Text1) - 2
```

```
        a = Mid(Text1, Len(Text1) - i, 1)
        i = i + 1
      Loop
       If i < > Len(Text1) - 2 Then
        Drive1.Drive = Left(Text1, 2)                    ' 完整路径名的前两位是驱动器名
        Dir1.Path = Left(Text1, Len(Text1) - i)
        File1.FileName = Right(Text1, i - 1)
      End If
   End If
   Exit Sub
2002:                                                    ' 错误处理程序
   MsgBox "文件名或路径错误！", vbExclamation, "警告"
End Sub
```

第 15 章　数据库访问技术

一、选择题

15.1　下述关于数据库系统的叙述中正确的是（　　）。

A）数据库系统减少了数据冗余　　C）数据库系统中数据的一致性是指数据类型一致

B）数据库系统避免了一切冗余　　D）数据库系统比文件系统能管理更多的数据

【解答】正确答案为 A。

分析：数据库系统的数据具有高共享性和低冗余性，但不能完全避免数据冗余；数据的一致性是指在系统中同一数据的不同出现应保持相同的值。

15.2　下列有关数据库的描述，正确的是（　　）。

A）数据处理是将信息转化为数据的过程

B）数据的物理独立性是指当数据的逻辑结构改变时，数据的存储结构不变

C）关系中的每一列称为元组，一个元组就是一个字段

D）如果一个关系中的属性或属性组并非该关系的关键字，但它是另一个关系的关键字，则称其为本关系的外关键字

【解答】正确答案为 D。

分析：数据处理是指将数据转换成信息的过程，故选项 A 叙述错误；数据的物理独立性是指数据的物理结构的改变，不会影响数据库的逻辑结构，故选项 B 叙述错误；关系中的行称为元组，对应存储文件中的记录，关系中的列称为属性，对应存储文件中的字段，故选项 C 叙述错误。

15.3　应用数据库的主要目的是（　　）。

A）解决数据保密问题　　B）解决数据完整性问题

C）解决数据共享问题　　D）解决数据量大的问题

【解答】正确答案为 C。

分析：数据库中的数据具有“集成”与“共享”的特点，也就是说，数据库集中了各种应用的数据，进行统一构造与存储，而使它们可以被不同应用程序所使用，故选项 C 正确。

二、填空题

15.4　数据库系统的核心是_____。

【解答】正确答案为：数据库管理系统。

分析：数据库管理系统（Database Management System，DBMS）是数据库的机构，它是一种系统软件，负责数据库中的数据组织、数据操纵、数据维护、控制及保护和数据服务等。数据库管理系统是数据库系统的核心。

15.5　数据库设计包括两个方面的设计内容，它们是_____和_____。

【解答】正确答案为：概念设计；逻辑设计。

分析：数据库设计包括数据库概念设计和数据库逻辑设计两个方面的设计内容。

三、编程题

15.6　设某校规定，超过全班平均成绩 10%者可以享受一等奖学金，超过全班平均成绩 5%者可以享受二等奖学金。试编写一个程序，使用数据控件，建立与存放学生成绩的 Access 数据库的连接，设数据库中包括“学号”、“姓名”、“成绩”3 个字段。程序执行后，输出奖学金等级一览表。

【解答】设数据库中的数据内容如图 15-1 所示。程序执行后，输出奖学金等级一览表，如图 15-2 所示。

	姓名	学号	成绩
▶	1111	001	65
	2222	002	78
	3333	003	84
	4444	004	61
	5555	005	74
	6666	006	94
	7777	007	86
	8888	008	76
	9999	009	68
	1010	010	79
*			0

图 15-1　数据库中的数据

图 15-2　程序运行结果

程序设计步骤如下。

（1）设计程序界面。

在窗体上添加一个 Data1 控件和两个命令按钮。

（2）设置对象属性。

设置两个命令按钮的 Caption 属性分别为“开始”和“退出”；设置 Data1 数据控件的 DatabaseName 属性为数据库文件的完整文件名（包括路径），设置其 RecordSource 属性为需要使用的表或查询的名称，这里选择了“学生成绩”表，设置 Data1 控件的 Connect 属性为“Access2000”。

（3）编写程序代码。

```
Private Sub Command1_Click( )                    ' 单击“开始”按钮时执行的代码
  Command1.Enabled = False
  Data1.Recordset.MoveFirst                      ' 将记录指针移到第 1 条记录处
  Do While Data1.Recordset.EOF = False           ' 指针指向数据库文件尾时循环结束
    zcj = zcj + Data1.Recordset.Fields("成绩").Value   ' 累加全班学生的成绩（总成绩）
    Data1.Recordset.MoveNext
  Loop
  pingjun = zcj / 10
  Data1.Recordset.MoveFirst
  Print "姓名", "学号", "成绩", "等级"
  Do While Data1.Recordset.EOF = False
   Select Case Data1.Recordset.Fields("成绩").Value
     Case Is > pingjun * 1.1
```

```
            Print Data1.Recordset.Fields("姓名").Value,        ' Print 语句结尾带有“,”表示不换行
            Print Data1.Recordset.Fields("学号").Value,
            Print Data1.Recordset.Fields("成绩").Value,
            Print "一等奖"
        Case Is > pingjun * 1.05
            Print Data1.Recordset.Fields("姓名").Value,
            Print Data1.Recordset.Fields("学号").Value,
            Print Data1.Recordset.Fields("成绩").Value,
            Print "二等奖"
        Case Else
            Print Data1.Recordset.Fields("姓名").Value,
            Print Data1.Recordset.Fields("学号").Value,
            Print Data1.Recordset.Fields("成绩").Value
        End Select
        Data1.Recordset.MoveNext                              ' 将记录指针指向下一条记录
    Loop
    Print
    Print "  平均", "  一等", "  二等"
    Print pingjun, pingjun * 1.1, pingjun * 1.05
End Sub
Private Sub Command2_Click( )
    End
End Sub
```

15.7　使用数据控件，结合 Access 数据库设计一个“通信录”程序。程序启动后显示图 15-3 所示的界面，用户可以使用“姓名”和“地址”下拉列表框查询需要的记录。单击“更新”按钮时，显示图 15-4 所示的“验证口令”对话框。

用户在正确回答了口令之后，显示网格式更新、删除、添加记录窗体。修改完毕后单击“更新”按钮将新数据写入数据库，输入新的数据可以添加记录，单击“删除”按钮，将删除当前记录，将光标移到最后的空白记录处。



图 15-3　程序启动时的界面



图 15-4　验证更新权限口令

【解答】程序设计步骤如下。

（1）创建程序主界面。

在窗体 Form1 上添加 6 个标签 Label1～Label6，4 个文本框 Text1～Text4，两个下拉列

表框 Combo1～Combo2，两个命令按钮和一个数据控件 Data1。适当调整各控件的大小及位置，如图 15-5 所示。

设置 6 个标签的 Caption 属性分别为“姓名”、“地址”、“性别”、“电话”、“传呼”和“E-mail”，设置两个命令按钮的 Caption 属性为“更新”和“退出”，设置各文本框的 Text 属性为空，设置 Data1 的 DatabaseName 属性为 Access 数据库所在的路径及文件名，RecordSource 属性为数据表名称（本例设为“通信录”）。

（2）创建登录对话框界面。

执行“工程”→“添加窗体”菜单命令，在“添加窗体”对话框中选择“登录对话框”后，单击“打开”按钮。此时系统会自动为程序创建登录对话框的窗体及代码框架。系统自动生成的代码如图 15-6 所示，读者应注意与后面的登录对话框窗体模块代码进行对照。根据本例的需要，可以将“用户名”文本框删除。更改系统生成的登录对话框窗体的 Name 属性为 pass。



图 15-5　设计程序主界面



图 15-6　系统为登录对话框生成的代码

（3）使用向导创建更新删除窗体界面。

执行“工程”→“添加窗体”菜单命令，在打开的“添加窗体”对话框中选择“VB 数据窗体向导”后，单击“打开”按钮启动向导。按屏幕提示填写需要的参数，在图 15-7 中填写数据库文件名，在图 15-8 中选择窗体布局类型和数据绑定类型，本例中选择了网格（数据表）和 ADO 数据控件的方式。

图 15-7　填写需要使用的数据库文件名

图 15-8　设置窗体类型和数据绑定类型

另外，在记录源设置对话框中，应选择记录源为数据库中的表“通信录”，并选择显示所有字段，适当调整各字段的排列顺序。在“控件选择”对话框中选择窗体具有“更新”、“删除”和“关闭”按钮。

数据网格窗体生成后，可添加一个标签控件用来提示用户，不要忘记输入“姓名”和“地址”字段。

（4）编写程序代码。

程序主窗体模块 Form1 代码如下：

```
Sub xianshi( )                                    ' 显示记录信息的自定义过程
  If Data1.Recordset("性别") < > "" Then
    Text1 = Data1.Recordset("性别")
  Else
    Text1 = "不详"
  End If
  If Data1.Recordset("电话") < > "" Then
    Text2 = Data1.Recordset("电话")
  Else
    Text2 = "无"
  End If
  If Data1.Recordset("传呼") < > "" Then
    Text3 = Data1.Recordset("传呼")
  Else
    Text3 = "无"
  End If
  If Data1.Recordset("E-mail") < > "" Then
    Text4 = Data1.Recordset("E-mail")
  Else
    Text4 = "无"
  End If
End Sub
Sub chushihua( )                                  ' 显示数据初始化的自定义过程
  Combo1.Clear
  Combo2.Clear
  Data1.Recordset.MoveFirst
  Do While Data1.Recordset.EOF = False
    Combo1.AddItem Data1.Recordset("姓名")        ' 将姓名字段的内容添加至组合框列表
    Combo2.AddItem Data1.Recordset("地址")        ' 将地址字段的内容添加至组合框列表
    Data1.Recordset.MoveNext
  Loop
  Data1.Recordset.MoveFirst
```

```
    Combo1.Text = Data1.Recordset("姓名")
    Combo2.Text = Data1.Recordset("地址")
End Sub
Private Sub Combo1_Click( )                    ' 姓名下拉列表框中发生单击事件时执行的代码
    Data1.Recordset.MoveFirst                  ' 将指针移到第一条记录
' 查找与“姓名”相符的第一条记录（指针指向符合条件的第一条记录）
    Data1.Recordset.FindFirst "姓名 ='" & Combo1.Text & "'"
' 将找到记录的“地址”字段内容显示到“地址”下拉列表框中
    Combo2.Text = Data1.Recordset("地址")
    Call xianshi          ' 调用 xianshi 自定义过程在各文本框中显示其他字段内容
End Sub
Private Sub Combo2_Click( )                    '“地址”下拉列表框中发生单击事件时执行的代码
    Data1.Recordset.MoveFirst
    Data1.Recordset.FindFirst "地址 ='" & Combo2.Text & "'"
    Combo1.Text = Data1.Recordset("姓名")
    Call xianshi
End Sub
Private Sub Command1_Click( )                  ' 单击“更新”按钮时执行的代码
    pass.Show                                  ' 显示“登录”对话框
End Sub
Private Sub Command2_Click( )                  ' 单击“退出”按钮时执行的代码
    End
End Sub
Private Sub Form_Initialize( )                 ' 窗体初始化时执行的代码
    Call chushihua
    Call xianshi
End Sub
```

登录对话框窗体模块代码如下：

```
' 其中带有下画线的内容需要用户自行编写，其他由系统自动生成
Option Explicit
Public LoginSucceeded As Boolean
Private Sub cmdCancel_Click( )
    LoginSucceeded = False
    Unload Me
    Form1.Show
    Call Form1.chushihua
    Call Form1.xianshi
End Sub
Private Sub cmdOK_Click( )
```

```
    If txtPassword = "654321" Then                    ' 如果用户输入的是正确密码
        LoginSucceeded = True
        Unload Me
        Unload Form1
        gengxin.Show
    Else
        MsgBox "无效的密码，请重试!", 48, "验证口令"
        txtPassword.SetFocus
        SendKeys "{Home}+{End}"
    End If
End Sub
```

更新、删除窗体模块的代码如下：

```
' 只有带下画线的部分需要用户自行编写，其他均由系统自动生成
Private Sub cmdClose_Click( )                         ' 单击“关闭”按钮时执行的代码
    Unload Me
    Form1.Show
    Call Form1.chushihua
    Call Form1.xianshi
End Sub
' 窗体卸载前（用户单击标题栏中的“关闭”按钮 ✕ 时）执行的代码
' 如果没有该过程，关闭时将无法回到程序主窗口
Private Sub Form_QueryUnload(Cancel As Integer, UnloadMode As Integer)
    cmdClose.Value = True      ' 调用“关闭”按钮 Command1 的 Click 事件过程
End Sub
Private Sub Form_Resize( )
    On Error Resume Next
' 当窗体调整时会调整网格
    grdDataGrid.Height = Me.ScaleHeight - datPrimaryRS.Height - 600 - picButtons.Height
End Sub
Private Sub Form_Unload(Cancel As Integer)
    Screen.MousePointer = vbDefault
End Sub
' 错误处理过程
Private Sub datPrimaryRS_Error(ByVal ErrorNumber As Long, Description As String, _
ByVal Scode As Long, ByVal Source As String, ByVal HelpFile As String, _
ByVal HelpContext As Long, fCancelDisplay As Boolean)
    ' 错误处理程序代码置于此处
    ' 想要忽略错误，注释掉下一行
    ' 想要捕获它们，在此处添加代码以处理它们
```

```
    MsgBox "Data error event hit err:" & Description
End Sub
Private Sub datPrimaryRS_WillChangeRecord(ByVal adReason As _
ADODB.EventReasonEnum, ByVal cRecords As Long, _
adStatus As ADODB.EventStatusEnum, ByVal pRecordset As ADODB.Recordset)
    ' 验证代码置于此处
    ' 下列动作发生时该事件被调用
    Dim bCancel As Boolean
    Select Case adReason
        Case adRsnAddNew
        Case adRsnClose
        Case adRsnDelete
        Case adRsnFirstChange
        Case adRsnMove
        Case adRsnRequery
        Case adRsnResynch
        Case adRsnUndoAddNew
        Case adRsnUndoDelete
        Case adRsnUndoUpdate
        Case adRsnUpdate
    End Select
    If bCancel Then adStatus = adStatusCancel
End Sub
Private Sub cmdAdd_Click( )
    On Error GoTo AddErr
    datPrimaryRS.Recordset.MoveLast
    grdDataGrid.SetFocus
    SendKeys "{down}"
    Exit Sub
AddErr:
    MsgBox Err.Description
End Sub
Private Sub cmdDelete_Click( )
    On Error GoTo DeleteErr
    With datPrimaryRS.Recordset
        .Delete
        .MoveNext
        If .EOF Then .MoveLast
    End With
```

```
    Exit Sub
DeleteErr:
    MsgBox Err.Description
End Sub
Private Sub cmdUpdate_Click( )
    On Error GoTo UpdateErr
    datPrimaryRS.Recordset.UpdateBatch adAffectAll
    Exit Sub
UpdateErr:
    MsgBox Err.Description
End Sub
```

15.8　使用 DBGrid 控件和数据绑定技术，创建一个数据库浏览程序。要求：使用 Access 创建一个数据库，该数据库中包含有一个“学生成绩”表，表中有“学号”、“姓名”、“班级”、“年龄”、“性别”、“数学”、“语文”、“英语”和“计算机”字段。

【解答】程序设计步骤如下。

（1）设计程序界面。

在窗体上添加 3 个标签 Label1～Label3，两个文本框 Text1 和 Text2，一个数据网格控件 MSFlexGrid1 和一个数据控件 Data1。适当设置各控件的大小和位置，如图 15-9 所示。注意：MSFlexGrid 控件是 ActiveX 控件（Microsoft MSFlexGrid Control 6.0），使用前应当执行“工程”→“部件”菜单命令将其添加到工具箱中，控件图标为 。



图 15-9　设计程序界面

（2）设置对象属性。

设置窗体的 Caption 属性为“数据绑定示例”，设置 Label1 和 Label2 的 Caption 属性分别为“学号”和“姓名”，3 个标签的 AutoSize 属性均为 True。

设置 MSFlexGrid 的 DataSource 属性为 Data1（与数据控件绑定）。

设置 Data1 的 Connect 属性为 Access 2000，DatabaseName 属性为数据库的实际路径及文件名，RecordSource 属性为“学生成绩”表，Align 属性为 2-Align Bottom，使控件对齐到窗体的底部。

设置 Text1, Text2 的 DataSource 属性均为 Data1，它们的 DataField 属性分别为“学号”和“姓名”（与对应字段绑定）。

（3）编写程序代码。

```
' 一般，在数据表中不存放计算字段（如总分、平均分等）的内容以减少数据库的数据冗余
```

```
'计算字段值通常在本地，由应用程序自行计算
Private Sub Text1_Change( )              ' 当“学号”文本框中的内容变化时执行的代码
   zf = Data1.Recordset("数学") + Data1.Recordset("语文") _
+ Data1.Recordset("英语") + Data1.Recordset("计算机")      ' 计算“总分”
   pjf = Format(zf / 4, "0.0")                            ' 计算“平均分”，并保留 1 位小数点
   Label3 = "总分：" & zf & "    平均分：" & pjf           ' 将计算结果显示在 Label3 中
End Sub
```

从本题可以看出，使用数据绑定技术，几乎不需要编写任何代码就可以创建一个一般的数据库浏览程序。程序启动后界面如图 15-10 所示。



图 15-10　数据绑定示例

15.9　使用 ADO 控件设计一个数据库管理程序，使之具有浏览、更新、添加和删除记录的功能。程序启动时，界面如图 15-11 所示，单击 ADO 控件的对应按钮，可以浏览数据表中的内容。单击“添加”按钮，出现图 15-12 所示的空记录界面，输入信息后在最后一个“性别”文本框中按 Enter 键可继续添加新记录，直到用户按下“更新”按钮。



图 15-11　程序启动时的界面



图 15-12　添加记录

单击“删除”按钮程序显示警告，确认后将删除当前记录，如图 15-13 所示。用户修改或添加了记录之后，单击“更新”按钮时才会将最后的更新一次性地写入数据表中，如图 15-14 所示。

图 15-13　删除记录



图 15-14　更新记录

【解答】程序设计步骤如下。

（1）设计程序界面。

在窗体上添加 6 个标签 Label1～Label6，5 个文本框 Text1～Text5，一个由 4 个命令按钮组成的按钮组 Command1(0)～Command1(3)和一个 ADO 控件 Adodc1，适当设置各控件的大小及位置，如图 15-15 所示。

（2）设置对象属性。

设置 5 个标签 Label1～Label5 的 Caption 属性分别为“学号”、“姓名”、“班级”、“年龄”和“性别”，Label6 的 Caption 属性设为空，所有标签的 AutoSize 属性均设置为 True。设置 4 个命令按钮的 Caption 属性分别为“添加”、“删除”、“更新”和“退出”。

在属性窗口中单击 Adodc1 ConnectionString（连接字符串）属性右边的 ... 按钮，如图 15-16 所示。在“属性页”对话框中选择“使用连接字符串”后，单击“生成”按钮，按屏幕提示选择“提供者”，单击“下一步”按钮。在“连接”选项卡中选择并测试数据库连接，测试成功后，单击“确定”按钮。按屏幕提示完成该属性的设置。



图 15-15　设计程序界面



图 15-16　Adodc1 的连接字符串属性

设置 Adodc1 的 RecordSource 属性为“学生成绩”表。5 个文本框 Text1～Text5 的 RecordSource 属性均设置为 Adodc1，分别与“学号”、“姓名”、“班级”、“年龄”和“性别”字段绑定。

（3）编写程序代码。

```
Private Sub Command1_Click(Index As Integer) ' 按钮组中某按钮被单击时执行的代码
  Select Case Index                          ' 添加
    Case 0                                   ' 在数据库的尾部添加记录
      Adodc1.Recordset.MoveLast
     Adodc1.Recordset.AddNew                 'AddNew 方法用于向数据库的尾部添加一条新记录
      Text1.SetFocus
      Label6 = "在“学号”处按 Esc 键放弃，在“性别”处按回车键继续添加"
    Case 1                                   ' 删除当前记录（屏幕上看到的记录）
      a = MsgBox("当前记录将被删除，确定吗？", 4 + 48, "警告")
      If a = vbNo Then Exit Sub
      Adodc1.Recordset.Delete
      Adodc1.Recordset.MoveLast
```

```
        Adodc1.Recordset.MoveFirst
    Case 2                                          ' 将缓存中已被更新的数据写入数据表中
        If Text1 = "" And Text2 = "" And Text3 = "" And Text4 = "" And Text5 = "" Then
            MsgBox "不能保存空记录", vbExclamation, "警告"
            Text1.SetFocus
            Exit Sub
        Else
            Adodc1.Recordset.UpdateBatch        ' 保存添加的新记录或修改后的所有记录内容（批更新）
            Adodc1.Recordset.MoveFirst          ' 将记录指针移动到第一条记录处
            MsgBox "所有修改和添加已被保存到数据库", vbInformation, "提示"
            Label6 = ""
        End If
    Case 3
        End
    End Select
End Sub
Private Sub Text1_Change( )                     ' 使显示记录变化时 ADO 控件的提示随之变化
    currec = Adodc1.Recordset.AbsolutePosition  ' 变量 currec 用于存放当前记录号
    recount = Adodc1.Recordset.RecordCount      ' 变量 recount 用于存放记录总数
    Adodc1.Caption = "          第 " & currec & " 条记录" & "      共 " & recount & " 条记录"
End Sub
' 可以将文本框设置为控件组，以使输入记录过程中随时按下 Enter 键或 Esc 键均可保存
' 或放弃输入的内容
Private Sub Text1_KeyUp(KeyCode As Integer, Shift As Integer)
    If KeyCode = vbKeyEscape Then       ' 用户在“学号”文本框中按下 Esc 键时执行的程序代码
        Adodc1.Recordset.CancelUpdate
        '上句的作用是取消在调用 Update 方法前对当前记录或新记录所做的任何更改
        Adodc1.Recordset.MoveFirst
        Label6 = ""
    End If
End Sub
' 在最后一个文本框（“性别”文本框）中按 Enter 键时执行的程序代码
Private Sub Text5_KeyUp(KeyCode As Integer, Shift As Integer)
    If KeyCode = 13 Then
        ' 如果所有文本框均为空，则表明该记录是一条空记录
        If Text1 = "" And Text2 = "" And Text3 = "" And Text4 = "" And Text5 = "" Then
            MsgBox "不能保存空记录", vbExclamation, "警告"
            Text1.SetFocus
```

```
            Exit Sub
        Else
            Adodc1.Recordset.UpdateBatch    ' 保存添加的新记录或修改后的所有记录内容（批更新）
        End If
        Adodc1.Recordset.AddNew
        Text1.SetFocus
    End If
End Sub
```

第 16 章　调 试 程 序

调试是在应用程序中查找并修改错误的过程。

16.1　调试程序

在程序设计过程中，错误是难免的，查找和修改错误的过程称为程序调试。VB 为调试程序提供了一组交互的、有效的调试工具。为了便于学习和上机实习，本节介绍 VB 基本的调试功能，如设置断点、观察变量和过程跟踪等。

16.1.1　错误类型

为了易于找出程序中的错误，我们将错误分为 4 种类型，即编辑错误、编译错误、运行错误和逻辑错误。

1．编辑错误

当用户在代码窗口中编辑代码时，VB 会对程序直接进行语法检查。当发现程序中存在输入错误，如语句没输入完、关键字输入错误等，VB 会弹出对话框，提示出错信息，如图 16-1 所示。这时，用户单击“确定”按钮，关闭提示框，程序中出错的位置显示为红色，出错部分被高亮度显示，提示用户进行修改。



（a）　　（b）　　（c）

图 16-1　编辑错误

2．编译错误

编译错误是指单击了“启动”按钮后，VB 开始运行程序前，编译执行程序段时产生的错误。此类错误是由于用户未定义变量、遗漏关键字等原因造成的。这时，VB 也将弹出对话框，提示出错信息，如图 16-2 所示。出错的位置被高亮度显示，同时 VB 停止编译。这时，用户必须单击“确定”按钮，关闭出错对话框，然后对出错行进行修改。

3．运行错误

运行错误是指在编译通过后，运行代码时发生的错误。这类错误往往是由于指令代码执

行了非法操作引起的，如类型不匹配、试图打开一个不存在的文件等。



（a）　　（b）

图 16-2　编译错误

例如，属性 FontSize 的类型为整型，若对其赋值的类型为字符串，系统运行时将显示图 16-3 所示的出错信息。若用户单击“调试”按钮，则进入中断模式，光标停留在引起出错的位置，此时允许修改代码。

图 16-3　运行错误

4．逻辑错误

程序运行后，如果得不到期望的结果，这说明程序存在逻辑错误。例如，运算符使用不正确，语句的次序不对，循环语句的起始值、终值不正确等。通常，逻辑错误不会产生错误提示信息，因此错误较难排除，这时就需要程序员仔细地阅读分析程序，并要有一定的调试程序的经验。

16.1.2　调试和排错

为了更正程序中发生的不同错误，VB 提供了功能强大的调试工具，主要通过设置断点、插入观察变量、逐行执行和过程跟踪等手段进行调试，并在调试窗口中显示所关注的信息。

1．VB 的 3 种模式

作为一个集编辑、编译与运行于一体的集成环境，VB 的工作状态可分为 3 种模式。为了测试和调试应用程序，用户在任何时候都应清楚地知道自己正处在何种模式下。

（1）设计模式

在设计模式下，可以进行程序的界面设计、属性设置、代码编写等操作，此时标题栏显示“[设计]”字样，如图 16-4 所示。在此模式下不能运行程序，也不能使用调试工具。



图 16-4　设计模式

（2）运行模式

执行“运行”→“启动”菜单命令，或按 F5 键，或单击工具栏上的“启动”按钮 ▶，即可由设计模式进入运行模式，标题栏显示“[运行]”字样，如图 16-5 所示。



图 16-5　运行模式

在此模式下，可以查看程序代码，但不能修改。若要修改代码，必须选择“运行”→“结束”菜单命令，或单击工具栏上的“结束”按钮 ■，回到设计模式；也可以选择“运行”→“中断”菜单命令，或单击工具栏上的“中断”按钮 ⏸，进入中断模式。

（3）中断模式

当程序运行时，执行“运行”→“中断”菜单命令，或单击工具栏上的“中断”按钮 ⏸，进入中断模式，如图 16-6 所示，此时标题栏显示“[break]”字样。当程序出现运行错误时，也可以进入中断模式。

在中断模式下，运行的程序被挂起，可以查看代码、修改代码、检查数据。修改结束，再单击“继续”按钮 ▶ 继续程序的运行，或单击“结束”按钮 ■ 停止程序运行。



图 16-6　中断模式

2．插入断点和逐语句跟踪

在调试程序时，通常会设置断点来中断程序的运行，然后逐语句跟踪检查相关变量、属性和表达式的值是否在预期的范围内。

可在中断模式下或设计模式下设置或删除断点。当应用程序处于空闲时，也可在运行时设置或删除断点。按下 F9 键，在程序运行到断点语句处（该语句未执行）停下，进入中断模式，如图 16-7 所示，在此之前所关心的变量、属性、表达式的值都可以查看。

在 VB 中提供了在中断模式下直接查看某个变量值的功能，只要把鼠标指向所关心的变量处，稍停片刻，则在鼠标下方显示该变量的值，如图 16-8 所示。



图 16-7　插入断点和逐语句跟踪



图 16-8　显示变量值

如果要继续跟踪断点以后的语句执行情况，只需按 F8 键或选择“运行”→“逐语句”菜单命令即可。

将设置断点和逐语句跟踪相结合，是初学者调试程序最简捷的方法。

3．调试窗口

在中断模式下，除了用鼠标指向要观察的变量可直接显示其值外，还可以通过立即窗口、本地窗口和监视窗口观察有关变量的值。可以通过单击“视图”菜单中的对应命令打开这些窗口。

（1）立即窗口

立即窗口是在调试窗口中最方便、最常用的窗口，如图 16-9 所示。可以在程序代码中利

用 Debug.Print 方法，把输出送到立即窗口中，也可以直接在该窗口使用 Print 语句或“?”符号显示变量的值。



图 16-9　立即窗口

（2）本地窗口

本地窗口用来显示当前过程中所有变量的值，如图 16-10 所示。当程序的执行从一个过程切换到另一个过程时，本地窗口的内容会发生改变，它只反映当前过程中可用的变量。

图 16-10　本地窗口

（3）监视窗口

监视窗口中可显示当前的监视表达式，如图 16-11 所示。在此之前必须在设计阶段，利用“调试”菜单中的“添加监视命令”或“快速监视”命令，添加监视表达式并设置监视类型，运行时，在监视窗口中将根据所设置的监视类型进行相应的显示。



图 16-11　监视窗口

16.2　错误陷阱

再有经验的程序员无论如何细心地调试程序，都不可能绝对避免错误的发生。在 VB 中常采用错误陷阱的方法防止致命错误的发生。

16.2.1 On Error 语句

设置错误陷阱可以使用 On Error 语句，其语法形式见表 16-1。

表 16-1 On Error 语句的语法形式

语　　句	说　　明
On Error GoTo line	转去以 line 为行号的程序行继续执行
On Error Resume Next	从出错语句的下一条语句继续执行
On Error GoTo 0	关闭当前过程中所有已启动的错误处理程序

错误处理程序的设计一般可分为以下 3 步。

① 使用 On Error 语句捕获错误，并把程序流程转向由标号指示的错误处理程序段。

② 编写错误处理代码，对所有可能预见的错误都做出相应的安排。

③ 根据错误类型可使用 Resume 语句重新执行出错语句，或使用 ResumeNext 语句执行出错语句的下一条语句继续运行程序。

【例 16-1】错误处理程序示例。建立一个 10 次的循环，每次产生两个 0～9 的随机整数，并输出两数的商。若出错，则执行错误处理语句，显示信息如图 16-12 所示；否则显示正常信息，如图 16-13 所示。



图 16-12　出错处理后结束的程序



图 16-13　正常结束的程序

```
Private Sub Form_Load( )                    ' 窗体装入时执行的代码
   Show
   Randomize                                ' 初始化随机数发生器
   On Error GoTo aa                         ' 若出现错误（b=0）转去执行行标号为 aa 的程序段
   For i = 1 To 10
      a = Int(Rnd * 10) : b = Int(Rnd * 10)  ' 产生两个 0～9 的随机整数
      Print a / b                           ' 输出两个随机数的商
   Next
   Print "程序正常结束！"
   Exit Sub                                 ' 退出过程，不执行错误处理语句
aa:
   Print "分母为零了,程序结束!"
End Sub
```

16.2.2 Err 对象

Err 对象中包含有关运行时错误的信息。Err 对象的属性由错误的生成者设置，这个生成者可以是 VB 系统，可以是对象，也可以是程序设计员。

1．Err 对象的方法

使用 Clear 方法，其语法格式为：

```
Err.Clear
```

通常，在处理错误之后使用 Clear 方法来清除 Err 对象，例如，在对 On Error Resume Next 使用拖延错误处理时就可使用 Clear 方法。每当执行下列语句时就会自动调用 Clear 方法：

- 任意类型的 Resume 语句
- Exit Sub, Exit Function, Exit Property 语句
- 任何 On Error 语句

注意，当处理因访问其他对象产生的错误时，与其使用 On Error GoTo 语句，不如使用 On Error Resume Next 语句。每一次与对象打交道之后都检查 Err，可消除代码访问对象时的含混之处，也可以确认是哪个对象将错误引入了 Err.Number 中，还可以确认最初是哪个对象产生了这个错误（Err.Source 中指定的对象）。

【例 16-2】Clear 方法使用示例。本示例使用 Err 对象的 Clear 方法将 Err 对象之数值属性重新设置为 0，并将其字符串属性设置为零长度字符串。如果在代码中省略 Clear 方法，则每完成一次循环便会显示一次错误信息（发生错误之后），且不管程序中的计算结果是否真的有错误。

程序代码如下：

```
Dim Result(10) As Integer          ' 声明数组变量
' 其元素容易溢出
Dim indx
On Error Resume Next               ' 将错误处理的方式设为“继续下一行”
Do Until indx = 10
    ' 下面计算若有错误发生，便显示错误信息
    Result(indx) = Rnd * indx * 20000
    If Err.Number < > 0 Then
        MsgBox Err, , "Error Generated: ", Err.HelpFile, Err.HelpContext        ' 弹出一个信息框
        Err.Clear                  ' 清除 Err 对象的属性
    Else
        indx = indx + 1
    End If
Loop
```

2．Raise 方法

若发生运行错误，Err 对象的属性被填入明确识别错误的信息及处理这个错误所使用的信息。为了在代码中生成运行时错误，应使用 Raise 方法。

其语法格式为：

Err.Raise number, source, description, helpfile, helpcontext

各参数的含义见表 16-2。

在任意形式的 Resume 或 On Error 语句之后，以及在错误处理子程序内的 Exit Sub, Exit Function 或 Exit Property 语句之后，将 Err 对象的属性重新设置为 0 或长度为 0 的字符串（""）。可使用 Clear 方法重新明确设置 Err 对象。

表 16-2　Raise 参数的含义及说明

参　数	说　明
number	必需的。长整型数，用来识别错误性质。VB 错误（既有 VB 定义的错误也有用户定义的错误）的范围在 0～65 535 之间：0～512 保留为系统错误，513～65 535 可以用做用户定义的错误。当在类模块中将 Number 属性设置成自己的错误代码时，可将错误代码号添加到 vbObjectError 常数上。例如，为了产生错误号 513，可将 vbObjectError+513 赋值给 Number 属性
source	可选的。字符串表达式，为产生错误的对象或应用程序命名。当设置对象的这一属性时，要使用窗体 project.class。如果没有指定 source，则使用当前 Visual Basic 工程的程序设计 ID
description	可选的。描述错误的字符串表达式。如果没有指定，则检查 Number 的值。如果可以将错误映射成 VB 运行时的错误代码，则将 Error 函数返回的字符串作为 Description 使用。如果没有与 Number 对应的错误，则要用到消息“应用程序定义的错误或对象定义的错误”
helpfile	可选的。帮助文件的完整限定路径，在帮助文件中可以找到有关错误的帮助信息
helpcontext	可选的。识别 helpfile 内的标题的上下文 ID，而 helpfile 中提供了有助于了解错误的描述

对于系统错误和类模块生成运行时错误，要使用 Raise 方法而不使用 Error 语句。在其他代码中是否使用 Raise 方法，这要看想要返回的信息量有多大。另外，Err 对象是具有全局范围的固有对象，在代码中没有必要建立这些对象的实例。

3．Err 对象的属性

Err 对象的默认属性是 Number。因为该默认属性可以用对象名称 Err 表示，所以不必修改以前用 Err 函数或 Err 语句书写的代码，Err 对象的常用属性见表 16-3。

表 16-3　Err 对象的常用属性

属　性	说　明
Number	返回或设置表示错误的数值，是 Err 对象的默认属性
Description	返回或设置一个字符串表达式，包含与对象相关联的描述性字符串
HelpContext	返回或设置一个字符串表达式，包含 Windows 帮助文件中主题的上下文 ID
HelpFile	返回或设置一个字符串表达式，表示帮助文件的完整限定路径
LastDLLError	返回因调用动态链接库（DLL）而产生的系统错误号
Source	返回或设置一个字符串表达式，指明最初生成错误的对象或应用程序的名称

附录 A　VB 6.0 中的属性和事件及其含义

A.1　VB 6.0 中的属性

属 性 名	含　　义
ActiveControl	活动控件
ActiveForm	活动窗体
Alignment	文本对齐类型
Align	指定图形在图片框中的位置
Archive	文本列表框是否含有文档属性
AutoRedraw	控制对象自动重画
AutoSize	控制对象自动调整大小
BackColor	背景颜色
BackStyle	指定线型与背景的结合方式
BorderColor	边框颜色
BorderStyle	边框类型
BorderWidth	边框宽度
Cancel	命令按钮是否为 Cancel
Caption	标题
Checked	菜单项加标记
ClipControls	设置 Paint 事件是否重画整个控件
Columns	指定列表框水平方向显示的列数
ControlBox	窗体是否有控制框
Count	对象的数量
CurrentX	当前 *X* 坐标
CurrentY	当前 *Y* 坐标
Default	指定默认按钮
DragIcon	控件拖动过程中作为图标显示
DragMode	拖动方式
DrawMode	绘图方式
DrawStyle	设置线型

DrawWidth	设置线宽
Drive	指定驱动器（驱动器列表框）
Enabled	对象是否可用
EXEName	活动文件名称
FileName	文件名
FileNumber	文件号
FillColor	填充颜色
FillStyle	填充方式
FontBold	字体加粗
FontCount	字体种类计数
FontItalic	字体斜体
FontName	字体名称
Fonts	按序号返回可用字体名称
FontSize	字体大小
FontStrikethru	加中画线
FontTransparent	字体与背景叠加
FontUnderline	加下画线
ForeColor	前景颜色
Height	设置或返回对象的高度
HelpContextID	对象与 Help 文件连接的 ID 号
HelpFile	在应用程序中调用 Help 文件
Hidden	指定文件列表框内显示的文件是否是隐含文件
Icon	窗体最小化后显示的图标
Image	窗体或图片框的图形句柄
Index	设置或返回控件数组中控件的下标
Interval	设置或返回计时器时间间隔的毫秒数
ItemData	用于列表框或组合框，与 List 属性相同
KeyPreview	窗体先收到键盘事件还是控件先收到键盘事件
LargeChange	滚动框在滚动条内变化的最大值
Left	控件与窗体左边界的距离
ListCount	列表框计数
List	字符串数组
ListIndex	指定控件当前选项的序号

Max, Min	指定滚动条的最大值和最小值
MaxButton	最大化按钮
MaxLength	指定文本框所接收的最大字符数
MDIChild	指定一个窗体为 MDI 子窗体
MinButton	最小化按钮
MousePointer	鼠标形状
MultiLine	设置多行文本框
MultiSelect	指定文本框或列表框为多项选择
Name	对象名称
NewIndex	列表框或组合框中最近一次加入的项目的下标
Normal	指定文件列表框内显示的文件的属性
Page	指定打印机当前页号
Parent	返回控件所在的窗体
PasswordChar	口令字符
Path	设置或返回当前路径
Pattern	程序运行在文件列表框中显示的文件类型
Picture	图片属性
ReadOnly	文件属性为只读
ScaleHeight	用户定义坐标系的纵坐标轴
ScaleLeft	用户定义坐标系起点的横坐标
ScaleMode	用户定义坐标系的单位
ScaleTop	用户定义坐标系起点的纵坐标
ScaleWidth	用户定义坐标系的横坐标轴
ScrollBars	决定一个文本框是否有水平或垂直滚动条
Selected	返回文件列表框或列表框内项目的选择状态
SelLength	所选文本的长度
SelStart	所选文本的起点
SelText	所选文本的字符串
Shape	形状控件的显示类型
Shortcut	设置菜单项热键
SmallChange	滚动条最小变化值
Sorted	列表框或组合框中的项目是否按字母顺序排列
Stretch	图形装入图片框的方式
Style	指定组合框的类型
System	设置或返回文件列表框内显示的文件是否是系统文件

TabIndex	设置或返回控件的选取顺序
TabStop	用 Tab 键移动光标时是否在某个控件上停留
Tag	控件的别名
Text	文本
Title	标题属性
Top	控件与窗体上边界的距离
TopIndex	设置列表框或文件列表框显示的第一个项目
TwipsPerPixelX	屏幕或打印机水平方向的点数
TwipsPerPixelY	屏幕或打印机垂直方向的点数
Value	滚动条移动后的值
Visible	控件是否可见
Width	对象宽度
WindowList	指定菜单项是否含有 MDI 窗体的窗口列表
WindowState	窗口状态
WordWrap	标签显示文本的方式
X1	设置或返回线型控件起点的横坐标
X2	设置或返回线型控件终点的横坐标
Y1	设置或返回线型控件起点的纵坐标
Y2	设置或返回线型控件终点的纵坐标

A.2 VB 6.0 中的事件

事 件 名	含　义
Activate	控件激活
Change	改变
Click	单击
DblClick	双击
Deactivate	窗体非激活，在激活另一个窗体时发生
DragDrop	拖放
DropOver	拖动
DropDown	拖动后放下

KeyDown	按下键盘
KeyPress	键盘按键
KeyUp	键盘放开
Load	装入
LostFocus	失去指针
MouseDown	鼠标按下
MouseMove	鼠标移动
MouseUp	鼠标松开
Paint	控件重画
PathChange	路径改变
PatternChange	属性改变
QueryUnload	窗体队列关闭
Resize	改变尺寸
Scroll	滚动条滚动
Timer	计时器
Unload	卸载对象
Updated	更新

附录 B　VB 常用对象的属性表

对象 / 属性	窗体	标签	文本框	命令按钮	复选框	单选钮	框架	滚动条	列表框	组合框	驱动器列表框	目录列表框	文件列表框	直线	形状	计时器	图片框	图像框	通用对话框	菜单	打印机	屏幕
Action																			#			
ActiveControl	#																					*
ActiveForm																						*
Align																	*					
Alignment		*	*		*	*																
Archive													*									
Hidden													*									
Normal													*									
System													*									
AutoRedraw	*																*					
AutoSize		*															*					
BackColor	*	*	*	*	*	*	*	*	*	*	*	*	*		*	*	*					
BackStyle		*													*							
BorderColor														*	*							
BorderStyle		*													*							
BorderWidth		*												*	*							
Cancel				*																		
CancelError																			*			
Caption	*	*		*	*	*	*													*		
Checked																				*		
ClipControls	*						*										*					
Color																			*			
Columns									*													
ControlBox	*																					
Copies																			*			
CurrentX	#																#				*	
CurrentY	#																#				*	
DataChanged		*	*		*												*	*				
DataField		*	*		*												*	*				
DataSource		*	*		*												*	*				
Default				*																		

续表

对象 / 属性	窗体	标签	文本框	命令按钮	复选框	单选钮	框架	滚动条	列表框	组合框	驱动器列表框	目录列表框	文件列表框	直线	形状	计时器	图片框	图像框	通用对话框	菜单	打印机	屏幕
DefaultExt																			*			
DialogTitle																			*			
DragIcon		*	*	*	*	*	*	*	*	*	*	*	*				*	*				
DragMode		*	*	*	*	*	*	*	*	*	*	*	*				*	*			*	
DrawMode	*													*	*		*				*	
DrawStyle	*																*				*	
Drive											#											
Enabled	*	*	*	*	*	*	*	*	*	*	*	*	*			*	*	*		*		
FileName													#						*			
FileTitle																			#			
FillColor	*														*		*				*	
FillStyle	*														*		*				*	
Filter																			*			
FilterIndex																			*			
Flags																			*			
FontBold	*	*	*	*	*	*	*	*	*	*	*	*	*				*		*		*	
FontItalic	*	*	*	*	*	*	*	*	*	*	*	*	*				*		*		*	
FontName	*	*	*	*	*	*	*	*	*	*	*	*	*				*		*		*	
FontSize	*	*	*	*	*	*	*	*	*	*	*	*	*				*		*		*	
FontStrikethru	*	*	*	*	*	*	*	*	*	*	*	*	*				*		*		*	
FontTransparent	*	*	*	*	*	*	*	*	*	*	*	*	*				*		*		*	
FontUnderline	*	*	*	*	*	*	*	*	*	*	*	*	*				*		*		*	
FontCount																					*	*
ForeColor	*	*	*	*	*	*	*	*	*	*	*	*	*				*		*		*	
FromPage																			*			
ToPage																			*			
hDC	#																#		#			
Height,Width	*		*	*	*	*	*	*	*	*	*	*	*	*	*		*	*			*	*
HelpCommand		*																	*			
HelpContext																			*			
HelpContextID	*		*	*	*	*	*	*	*	*	*	*	*				*			*		
HelpFile																			*			
HelpKey																			*			
HideSelection			*																			
hWnd	#		#	#	#	#	#	#	#	#	#	#	#				#					

续表

对象 / 属性	窗体	标签	文本框	命令按钮	复选框	单选钮	框架	滚动条	列表框	组合框	驱动器列表框	目录列表框	文件列表框	直线	形状	计时器	图片框	图像框	通用对话框	菜单	打印机	屏幕
Icon	*																					
Image	#																#					
Index		*	*	*	*	*	*	*	*	*	*	*	*	*	*	*	*	*	*	*		
InitDir																			*			
Interval																*						
ItemData									#	#												
KeyProview	*																					
LargeChange								*														
SmallChange								*														
Left,Top	*	*	*	*	*	*	*	*	*	*	*	*	*		*	*	*	*	*			
LinkItem		*	*														*					
LinkMode	*	*	*														*					
LinkTimeout		*	*														*					
LinkTopic	*	*	*														*					
List									#	#	#	#	#									
ListCount									#	#	#	#	#									
ListIndex									#	#	#	#	#									
Max,Min																			*			
MaxButton	*																					
MinButton	*																					
MaxFileSize																			*			
MaxLength			*																			
MDIChild	*																					
MousePointer	*	*	*	*	*	*	*	*	*	*	*	*	*				*	*				*
MultiLine			*																			
MultiSelect									*				*									
Name	*	*	*	*	*	*	*	*	*	*	*	*	*	*	*	*	*	*	*			
NewIndex									#	#												
Page																					*	
Parent		#	#	#	#	#	#	#	#	#	#	#	#	#	#	#	#	#		#		
PasswordChar			*																			
Path												#	#									
Pattern													*									
Picture	*																*	*				
PrinterDefault																			*			

续表

对象 / 属性	窗体	标签	文本框	命令按钮	复选框	单选钮	框架	滚动条	列表框	组合框	驱动器列表框	目录列表框	文件列表框	直线	形状	计时器	图片框	图像框	通用对话框	菜单	打印机	屏幕
ReadOnly													*									
ScaleHeight	*																*				*	
ScaleWidth	*																*				*	
ScaleLeft	*																*				*	
ScaleTop	*																*				*	
ScaleMode	*																*				*	
ScrollBars			*																			
Selected									#				*									
SelLength			#							#												
SelStart			#							#												
SelText			#							#												
Shape															*							
Shortcut																				*		
Stretch																		*				
Style										*												
TabIndex		*	*	*	*	*	*	*	*	*	*	*	*				*					
TabStop			*	*	*	*		*	*	*	*	*	*				*					
Tag	*	*	*	*	*	*	*	*	*	*	*	*	*	*	*	*	*	*	*	*		
Text			*						#	*												
TopIndex									#				#									
TwipsPerPixelX																					*	*
TwipsPerPixelY																					*	*
Value				#	*	*		*														
Visible	*	*	*	*	*	*	*	*	*	*	*	*	*	*	*	*	*	*	*	*		
WindowList																				*		
WindowState	*																					
WordWrap					*																	
X1,Y1														*								
X2,Y2														*								

* 表示在属性窗口中具有的属性，可直接在设计阶段设置。

表示在属性窗口中没有的属性，只能通过程序代码设置、修改或读取。

附录 C　VB 常用对象可响应的事件表

对象 / 事件	窗体	标签	文本框	命令按钮	复选框	单选钮	框架	滚动条	列表框	组合框	驱动器列表框	目录列表框	文件列表框	计时器	图片框	图像框	菜单
Active	*																
Deactivate	*																
Change		*	*					*		*	*	*			*		
Click	*	*	*	*	*	*	*		*	*		*	*		*		*
DblClick	*	*	*			*	*		*	*			*		*	*	
DragDrop	*	*	*	*	*	*	*	*	*				*		*	*	
DragOver	*	*	*	*	*	*	*	*	*	*	*	*	*		*	*	
DropDown										*							
GotFocus	*		*	*	*	*		*	*	*	*	*	*		*		
KeyPress	*		*	*	*	*		*	*	*	*	*	*		*		
KeyDown	*		*	*	*	*		*	*	*	*	*	*		*		
KeyUp	*		*	*	*	*		*	*	*	*	*	*		*		
LinkClose	*	*	*												*		
LinkError	*	*	*												*		
LinkExcute	*																
LinkNotify		*	*														
LinkOpen	*	*	*												*		
Load	*																
LostFocus	*		*	*	*	*		*	*	*	*	*	*		*		
MouseDown	*	*	*	*	*	*	*		*			*	*		*	*	
MouseUp	*	*	*	*	*	*	*		*			*	*		*	*	
MouseMove	*	*	*	*	*	*	*		*			*	*		*	*	
Paint	*														*		
PathChange													*				
PatternChange													*				
QueryUnload	*																
Resize	*														*		
Scroll								*									
Timer														*			
Unload	*														*		

* 表示在属性窗口中具有的属性，可直接在设计阶段设置。

表示在属性窗口中没有的属性，只能通过程序代码设置、修改或读取。

附录 D　VB 常用对象的方法表

对象 / 方法	窗体	标签	文本框	命令按钮	检查框	单选钮	框架	滚动条	列表框	组合框	驱动器列表框	目录列表框	文件列表框	直线	形状	图片框	图像框	打印机
AddItem									*	*								
Circle	*															*		
Clear									*	*								*
Cls	*															*		
Drag		*	*	*	*	*	*	*	*	*	*	*	*			*	*	
EndDoc																		*
Hide	*																	
Line	*															*		*
LinkExecute		*	*													*		*
LinkPoke		*	*													*		
LinkRequest		*	*													*		
LinkSend																*		
Move	*	*	*	*	*	*	*	*	*	*	*	*	*	*	*	*	*	*
NewPage																	*	
Point	*															*		
Print	*															*		*
PrintForm	*																	
Pset	*															*		*
Refresh	*	*	*	*	*	*	*	*	*	*	*	*	*	*	*	*	*	
RemovItem									*	*								
Scale	*															*		
SetFocus	*	*	*	*	*	*		*	*	*	*	*	*			*		
Show	*																	
TextHeight	*															*		*
TextWidth	*															*		*

* 表示在属性窗口中具有的属性，可直接在设计阶段设置。

表示在属性窗口中没有的属性，只能通过程序代码设置、修改或读取。

Visual Basic 程序设计教程习题及习题解答（第4版）

本书为普通高等教育“十一五”国家级规划教材《Visual Basic程序设计教程（第4版）》的配套教材，对该教材中的习题做了详细解答。本书以强化学生实践能力为目的，详细讲解了等级考试中的选择题、填空题、思考题、编程题等各种题型的应试技巧和分析解答方法，既便于学生检测知识掌握程度，又符合各类VB考试题型。同时，提供的各种编程典型习题既方便教师掌握重点，也便于学生复习应试。本书的程序调试技术一章，对VB上机技巧和程序调试方法进行了详细讲解。由于VB控件及其属性内容繁多，为方便读者使用时查询，本书附录中列出了VB常用事件、属性和方法及其含义。

本书可作为高等学校、高职高专院校相关专业的教材，也可作为全国计算机等级考试二级Visual Basic语言的辅导教材，或者作为Visual Basic语言的“编程实例详解”单独使用。

ISBN 978-7-121-19781-9
9 787121 197819 >
定价： 元

责任编辑：冉 哲
封面设计：徐海燕